code:
;FileName_EVENTCOMEND.ERB -------------------------- Rev1. 00
;턴 종료의 처리
;CALL		SYSTEM
;ARG		VOID
;RETURN		VOID
;COMMENT
;-----------------------------------------------------------
@EVENTCOMEND
#DIM 장비부위
;일동작의 에러 체크
;FOR LOCAL, 0, CHARANUM
;	SIF CFLAG:LOCAL:326 == 0 && CFLAG:LOCAL:행동 == 5
;		PRINTFORMW ### COM_END 326ERROR %CALLNAME:LOCAL% ### CHARA_NO:{LOCAL}
;	SIF CFLAG:LOCAL:현재위치 == 0
;		PRINTFORMW ### COM_END LOCATION ERROR %CALLNAME:LOCAL% 의 현재위치가 이상합니다 ### CHARA_NO:{LOCAL}
;NEXT
;턴 종료 공통 처리
CALL TURN_RESET

SIF FLAG:연회개최플래그 > 0
	CALL ENKAI_SINKOU

SIF TFLAG:102 < 2 && TARGET && SHIRAHU(TARGET) && !FLAG:70
	CALL 봇슈트
IF BASE:MASTER:술기운 >= MAXBASE:MASTER:술기운
	;AT_HOME 보는 것만으로도 좋을지도 모른다
	IF !AT_HOME(MASTER) && GET_MAPID(CFLAG:MASTER:현재위치) != MAIN_MAP
		CALL 데이트종료타임업처리
	ELSE
		PRINTFORMW 한계까지 계속 마신 %마스터는%, 그대로 풀썩 쓰러져 버렸다…
	ENDIF
	GOTO OYASUMI
ENDIF
IF BASE:MASTER:체력 <= 500 && ITEM:리버스페더
	FONTSTYLE 1
	PRINTFORML 리버스 페더가 빛을 발하며　조용히 불타올랐다！
	FONTREGULAR
	CALL RECOVER_PERMIL(MASTER, 300, "체력", 0, "리버스페더에 의한 회복")
	CALL RECOVER_PERMIL(MASTER, 200, "기력", 0, "리버스페더에 의한 회복")
	ITEM:리버스페더 --
ELSEIF (BASE:MASTER:체력 <= 500 && !TCVAR:TARGET:마음대로해) || BASE:MASTER:체력 <= 0
	PRINTL (체력이 한계에 와 있습니다)
	;발정 없이 방에 들어갈 경우or 뉴 플랜 에는 선택지를 내놓지 않는다.
	IF CFLAG:MASTER:현재위치 == OMANEKIBEYA() && (!TALENT:TARGET:연모 || TFLAG:거품목욕)

	ELSE
		;밀어넘어뜨려지고 안이라면 분기
		IF TFLAG:102 == 3 && BASE:MASTER:체력 > 0 && !TCVAR:TARGET:마음대로해
			CALL ASK_YN("마음대로 하게 한다", "풀어달라고 한다")
			IF !RESULT
				TCVAR:TARGET:마음대로해 = 1
				BASE:MASTER:체력 -= 250
				GOTO SUKINISITE
			ENDIF
		ENDIF
	ENDIF
	PRINTW 　하루를 종료합니다
	IF BASE:MASTER:체력 <= 0 && !CHK_DATENOW(CFLAG:MASTER:데이트중)
		PRINTFORMW %마스터는% 체력을 다 써버려서 실신해 버렸다…
		FLAG:기절중단 = 100 + CFLAG:MASTER:현재위치 / 100
	ENDIF
	;밀어넘어뜨려지고 안이라면 TARGET에 기학 경험
	IF TFLAG:102 == 3
		IF BASE:MASTER:체력 <= 0
			EXP:가학쾌락경험 += 10
			PRINTL
			PRINTFORMW 가학쾌락경험 +10(%CALLNAME%)
			PRINTL
		ELSE
			EXP:가학쾌락경험 += 2
			PRINTL
			PRINTFORMW 가학쾌락경험 +2(%CALLNAME%)
			PRINTL
		ENDIF
	ENDIF
	$OYASUMI
	CALL SET_SLEEP(1, MASTER, 0)
	CFLAG:MASTER:취침시간 = DAY * 1440 + TIME
	;외박 할 수 있는 상태라면 그대로 잔다(연모 전제)
	IF CFLAG:MASTER:현재위치 == OMANEKIBEYA() && !TCVAR:(CFLAG:MASTER:초대):연속숙박금지
		IF CFLAG:(CFLAG:MASTER:초대):데이트중
			PRINTFORMW 오늘은 %조사처리(CALLNAME:(CFLAG:MASTER:초대),"가")% 묵게 해 주는 것 같다
			FLAG:기절중단 = 0
		;발정 없이 방에 들어설 경우
		ELSE
			PRINTFORML 슬슬 %CALLNAME:(CFLAG:MASTER:초대)%도 제정신으로 돌아올 시간이겠지, 적당한 때인 것 같다…
			IF TFLAG:102 >= 2
				PRINTFORMW 매달리는 %조사처리(CALLNAME:(CFLAG:MASTER:초대),"를")% 어떻게든 떼어내고, %조사처리(GET_MAPNAME(MAIN_MAP),"로")% 돌아왔다…
			ELSE
				PRINTFORMW 촉촉하게 열을 띠고 있는 %CALLNAME:(CFLAG:MASTER:초대)%의 시선에서 벗어나기 위해, %조사처리(GET_MAPNAME(MAIN_MAP),"로")% 돌아왔다…
			ENDIF
			;部屋の汚れを保存しておく
			IF YOGORE:(OMANEKIBEYA()) > 0
				TCVAR:(CFLAG:MASTER:초대):초대방의더러움 = YOGORE:(OMANEKIBEYA())
			ELSE
				;０以下にした場合は識別するために１残す
				TCVAR:(CFLAG:MASTER:초대):초대방의더러움 = 1
			ENDIF
			;초대フラグに0を代入
			CFLAG:TARGET:초대 = 0
			CFLAG:MASTER:초대 = 0
			FLAG:기절중단 = 0
			CFLAG:MASTER:현재위치 = ENTRANCE
			PRINTW
		ENDIF
	ELSEIF TFLAG:거품목욕 && CFLAG:우후후
		PRINTFORMW
		CALL 泡風呂終了判定("あなた限界")
		IF MAIN_MAP != 9
			PRINTFORMW %CALLNAME:MASTER%は最後の力を振り絞って、%GET_MAPNAME(MAIN_MAP)%まで戻ってきた
			FLAG:기절중단 = 100 + CFLAG:MASTER:현재위치 / 100
			CFLAG:MASTER:현재위치 = ENTRANCE
		ELSE
			PRINTFORMW %CALLNAME:MASTER%はふらつく足取りで、私室へと戻った
			CFLAG:MASTER:현재위치 = CFLAG:MASTER:311
			FLAG:기절중단 = 0
		ENDIF
	ELSEIF TARGET > 0 && TALENT:TARGET:연모 && BASE:TARGET:체력 >= 500 && !CFLAG:TARGET:수면 && !CFLAG:MASTER:동침중 && ((CFLAG:TARGET:동행 && CHK_DATENOW(CFLAG:MASTER:데이트중)) || !CHK_DATENOW(CFLAG:MASTER:데이트중))
		IF CHK_DATENOW(CFLAG:MASTER:데이트중)
			PRINTFORMW %CALLNAME:TARGET%에게 업혀서 %마스터는% %GET_MAPNAME(MAIN_MAP)%까지 돌아왔다
		ENDIF
		IF CFLAG:MASTER:현재위치 != CFLAG:MASTER:초기위치
			PRINTFORMW %타겟이% %마스터를% 사실로 옮겨 주었다…
			CFLAG:MASTER:현재위치 = CFLAG:MASTER:311
			;おやすみ口上,6-0,MASTERがダウン（部屋に運ばれる）
			CALL KOJO_MESSAGE_SEND("EVENT", 3, TARGET, 6, 0)
		ELSE
			PRINTFORMW %타겟이% %마스터를% \@ INROOM(CFLAG:MASTER:현재위치) == 2 ? 침대 # 이불 \@로 옮겨 주었다…
			;おやすみ口上,6-1,MASTERがダウン（自室）
			CALL KOJO_MESSAGE_SEND("EVENT", 3, TARGET, 6, 1)
		ENDIF
		CFLAG:MASTER:데이트중 = MAIN_MAP
		FLAG:기절중단 = 0
	ELSEIF CFLAG:MASTER:현재위치 == P54난쟁이벌레통
		CFLAG:MASTER:현재위치 = P10툇마루
		PRINTFORMW 어느새 %마스터는% 툇마루에 널브러저 있었다…
	ELSEIF CHK_DATENOW(CFLAG:MASTER:데이트중)
		PRINTFORMW %마스터는% 마지막 힘을 쥐어짜서, %GET_MAPNAME(MAIN_MAP)%까지 돌아왔다
		FLAG:기절중단 = 100 + CFLAG:MASTER:현재위치 / 100
		CFLAG:MASTER:현재위치 = ENTRANCE
		PRINTFORMW
	ELSE
		PRINTFORMW %마스터는% 휘청거리는 발걸음으로, 사실로 돌아왔다
		CFLAG:MASTER:현재위치 = CFLAG:MASTER:311
		FLAG:기절중단 = 0
	ENDIF
	FOR LOCAL, 0, CHARANUM
		SIF CFLAG:LOCAL:출금
			CONTINUE
		CALL ENDUFUFU(LOCAL)
		CALL SET_PRANK(0, LOCAL)
		CFLAG:LOCAL:데이트중 = MAIN_MAP
		CFLAG:LOCAL:동행 = 0
		FLAG:데이트상대 = 0
		TFLAG:데이트전호감도 = 0
		SIF !BATHCHECK(CFLAG:LOCAL:현재위치) && TEQUIP:LOCAL:목욕탕플레이
			TEQUIP:LOCAL:목욕탕플레이 = 0
	NEXT
;	수치 변수가 소수점 이하를 잘라 버리는 특성을 악용 한 것이 이쪽, 다른 변수를 사용하지 않는 분 이쪽 쪽이 좋은 것인지도?
	CFLAG:MASTER:데이트중 = CFLAG:MASTER:현재위치 / 100
	TFLAG:106 = 0
	BEGIN AFTERTRAIN
ENDIF
IF TARGET && TFLAG:키스마크 == TARGET && TFLAG:붕대 && !RAND:2
	PRINTFORMW %CALLNAME:MASTER%의 목덜미를 가리고 있는 것을 알아차린 %타겟은%, 붕대를 풀어버렸다
	TFLAG:붕대 = 0
	BASE:분노 += 500
ENDIF
$SUKINISITE
IF TARGET && CFLAG:언짢음 && !FLAG:70
	IF TALENT:TARGET:담력 < 0
		PRINTFORMW %타겟을% 울려 버렸다…
		CALL KOJO_MESSAGE_SEND("EVENT", 29, TARGET, 2,0)
	ELSE
		PRINTFORMW %타겟을% 화나게 해 버렸다…
		CALL KOJO_MESSAGE_SEND("EVENT", 29, TARGET, 1,0)
	ENDIF

	IF !TCVAR:억지로 && FLAG:모리야복권데이트상대 != TARGET
		FOR LOCAL, 0, CHARANUM
			CALL ENDUFUFU(LOCAL)
		NEXT
		TFLAG:106 = 0
	ELSEIF CFLAG:우후후 && TFLAG:거품목욕
		CALL 泡風呂終了判定("女の子フキゲン")
	ENDIF
ENDIF

IF FLAG:70 == 1
	IF BASE:MASTER:TSP <= 0
		PRINTFORML (TSP가 없어졌습니다)
		[SKIPSTART]
		FOR LOCAL, 1, CHARANUM
			IF CFLAG:LOCAL:현재위치 == CFLAG:PLAYER:현재위치
				FOR 장비부위, 0,23
					SIF (!수면시간(LOCAL) && EQUIP:LOCAL:장비부위 != CFLAG:LOCAL:(200 + 장비부위)) || (수면시간(LOCAL) && EQUIP:LOCAL:장비부위 != CFLAG:LOCAL:(240 + 장비부위))
						LOCAL:1 ++
					SIF 장비부위 == 6 && CFLAG:LOCAL:노팬티 && ((!수면시간(LOCAL) && EQUIP:LOCAL:장비부위 != CFLAG:LOCAL:(200 + 장비부위)) || (수면시간(LOCAL) && EQUIP:LOCAL:장비부위 != CFLAG:LOCAL:(240 + 장비부위)))
						LOCAL:1 --
				NEXT
				IF LOCAL:1
					EQUIP:LOCAL:0 = 1
					IF CFLAG:LOCAL:시간정지착의 == 1
						PRINTFORMW %마스터는% %CALLNAME:LOCAL%를\@ CFLAG:LOCAL:노팬티 ?  팬티 이외를 #  \@원래 모습에 되돌렸다!
						CFLAG:LOCAL:시간정지착의 = 0
					ELSE
						PRINTFORMW %마스터는% %CALLNAME:LOCAL%에\@ CFLAG:LOCAL:노팬티 ?  팬티 이외의#  \@옷을 입혔다
					ENDIF
				ENDIF
			ENDIF
		NEXT
		[SKIPEND]
		IF CFLAG:MASTER:현재위치 != OMANEKIBEYA()
			PRINTFORMW %마스터는% 당황해하면서 그 자리를 떴다
			TFLAG:운반 = 0
			CALL FISHER_YATES_SHAFFLE(CHARANUM - 1)
			FOR LOCAL:10, 0, CHARANUM - 1
				LOCAL = SHAFFLE_ARRAY:(LOCAL:10) + 1
				SIF CFLAG:LOCAL:현재위치 != CFLAG:MASTER:현재위치
					CONTINUE
				IF EX:LOCAL:C절정 + EX:LOCAL:V절정 + EX:LOCAL:A절정 + EX:LOCAL:B절정 + EX:LOCAL:M절정 > 30
					PRINTFORMW 갑자기 격렬한 절정에 휩싸인 %조사처리(CALLNAME:LOCAL,"는")%, 달콤한 교성을 지르며 그 자리에 주저앉아 버렸다
					CFLAG:LOCAL:333 += 20
				ELSEIF EX:LOCAL:C절정 + EX:LOCAL:V절정 + EX:LOCAL:A절정 + EX:LOCAL:B절정 + EX:LOCAL:M절정 > 0
					PRINTFORMW %조사처리(CALLNAME:LOCAL,"는")% 수수께끼의 쾌감에 움찔 몸을 굳었다
				ELSEIF CFLAG:LOCAL:291 == 1
					PRINTFORMW 모르는 사이에 팬티가 벗겨진 %조사처리(CALLNAME:LOCAL,"는")% 안절부절하고 있다
				ENDIF
				SIF EX:LOCAL:처녀상실체크 == 1
					PRINTFORMW %조사처리(CALLNAME:LOCAL,"는")% 사타구니의 둔통에 얼굴을 찌푸리고 있다…
			NEXT
			IF !AT_HOME(MASTER)
				PRINTFORML %GET_MAPNAME(MAIN_MAP)%까지 전력질주했다…
				BASE:MASTER:체력 -= TIME_REQUIRED(MAIN_MAP, CFLAG:MASTER:데이트중) * 5
				BASE:MASTER:기력 -= TIME_REQUIRED(MAIN_MAP, CFLAG:MASTER:데이트중) * 5
				BASE:MASTER:체력 = MAX(BASE:MASTER:체력, 0)
				BASE:MASTER:기력 = MAX(BASE:MASTER:기력, 0)
				TIME += TIME_REQUIRED(MAIN_MAP, CFLAG:MASTER:데이트중) / 5
			ENDIF
			IF CFLAG:MASTER:현재위치 != CFLAG:MASTER:초기위치
				CFLAG:MASTER:현재위치 = CFLAG:MASTER:초기위치
			ELSE
				CFLAG:MASTER:현재위치 = 피난처
			ENDIF
			BASE:MASTER:TSP = 0
			CALL AFTER_AFFAIR
			FOR LOCAL, 0, CHARANUM
				;MASTER가 없어진 뒤인데 표시되는 것이 이상하고 시간 정지 해제 시에 어긋나는 말이 호출될 위험성이 높고 말 기믹을 파괴할 가능성이 높다는 코멘트 아웃
				[SKIPSTART]
				IF CFLAG:LOCAL:우후후 == 1 && LOCAL != 0
					;우후후 종료 말호출
					CALL KOJO_MESSAGE_SEND("EVENT", 15, LOCAL, 2,0)
				ENDIF
				[SKIPEND]
				CALL ENDUFUFU(LOCAL)
				CFLAG:LOCAL:데이트중 = MAIN_MAP
			NEXT
			FLAG:데이트상대 = 0
			TFLAG:데이트전호감도 = 0
			TFLAG:106 = 0
		ELSE
			CALL FISHER_YATES_SHAFFLE(CHARANUM - 1)
			FOR LOCAL:10, 0, CHARANUM - 1
				LOCAL = SHAFFLE_ARRAY:(LOCAL:10) + 1
				IF CFLAG:LOCAL:현재위치 == CFLAG:MASTER:현재위치
					IF EX:LOCAL:C절정 + EX:LOCAL:V절정 + EX:LOCAL:A절정 + EX:LOCAL:B절정 + EX:LOCAL:M절정 > 30
						PRINTFORMW 갑자기 격렬한 절정에 휩싸인 %조사처리(CALLNAME:LOCAL,"는")%, 달콤한 교성을 지르며 그 자리에 주저앉아 버렸다
						CFLAG:LOCAL:333 += 20
					ELSEIF EX:LOCAL:C절정 + EX:LOCAL:V절정 + EX:LOCAL:A절정 + EX:LOCAL:B절정 + EX:LOCAL:M절정 > 0
						PRINTFORMW %조사처리(CALLNAME:LOCAL,"는")% 수수께끼의 쾌감에 움찔 몸을 굳었다
					ELSEIF CFLAG:LOCAL:291 == 1
						PRINTFORMW 모르는 사이에 팬티가 벗겨진 %조사처리(CALLNAME:LOCAL,"는")% 안절부절하고 있다
					ENDIF
					SIF EX:LOCAL:처녀상실체크 == 1
						PRINTFORMW %조사처리(CALLNAME:LOCAL,"는")%는 사타구니의 둔통에 얼굴을 찌푸리고 있다…
				ENDIF
				CALL ENDUFUFU(LOCAL)
			NEXT
			CFLAG:MASTER:우후후 = 0
		ENDIF
		FLAG:70 = 0
		RESETBGCOLOR
		BASE:MASTER:TSP = 0
	ENDIF
	SIF CFLAG:MASTER:우후후 && BASE:MASTER:발기 <= 900 && BASE:MASTER:정력 >= 500
		BASE:MASTER:발기 += 10
ENDIF
IF TARGET && BASE:체력 < 500
	IF FLAG:70 == 1 && CFLAG:MASTER:현재위치 != OMANEKIBEYA()
		IF !CFLAG:쇠약
			PRINTFORML (%CALLNAME:TARGET%의 체력이 한계에 다다랐습니다)
			PRINTFORMW %마스터는% 그 자리를 떠나, 시간 정지를 해제했다
			CFLAG:쇠약 = 360
			FLAG:70 = 0
			FLAG:데이트상대 = 0
			TFLAG:데이트전호감도 = 0
			RESETBGCOLOR
			IF AT_HOME(MASTER)
				CFLAG:MASTER:데이트중 = MAIN_MAP
				IF CFLAG:MASTER:현재위치 != CFLAG:MASTER:초기위치
					CFLAG:MASTER:현재위치 = CFLAG:MASTER:초기위치
				ELSE
					CFLAG:MASTER:현재위치 = 피난처
				ENDIF
			ELSE
				IF STR:(6000 + CFLAG:MASTER:현재위치 + 10) == ""
					CFLAG:MASTER:현재위치 -= 10
				ELSE
					CFLAG:MASTER:현재위치 += 10
				ENDIF
			ENDIF
		ENDIF
		CALL CHARA_SLEEP, TARGET, 0
		FOR LOCAL, 0, CHARANUM
			CALL ENDUFUFU(LOCAL)
			CALL SET_PRANK(0, LOCAL, 1)
			SIF LOCAL
				CFLAG:LOCAL:데이트중 = MAIN_MAP
			CFLAG:LOCAL:동행 = 0
		NEXT
		FLAG:데이트상대 = 0
		TFLAG:데이트전호감도 = 0
		TFLAG:106 = 0
	ELSEIF TFLAG:거품목욕
		CALL 泡風呂終了判定("体力限界")
	ELSE
		IF !CFLAG:쇠약
			PRINTFORMW (%CALLNAME:TARGET%의 체력이 한계에 다다랐습니다)
			CFLAG:쇠약 = 360
			FLAG:70 = 0
			RESETBGCOLOR
		ENDIF
		CALL CHARA_SLEEP, TARGET, 1
		FOR LOCAL, 0, CHARANUM
			CALL ENDUFUFU(LOCAL)
			CALL SET_PRANK(0, LOCAL, 1)
			SIF LOCAL
				CFLAG:LOCAL:데이트중 = MAIN_MAP
			CFLAG:LOCAL:동행 = 0
		NEXT
		FLAG:데이트상대 = 0
		TFLAG:데이트전호감도 = 0
		TFLAG:106 = 0
	ENDIF
ENDIF

IF GET_EXISTMOB(MAIN_MAP)
	FOR LOCAL, MINROOM(), MAXROOM
		IF ROOMDATA:(LOCAL % 100) & 장소_모브
			YOGORE:LOCAL += TIME_PROGRESS(TIME:1) * MOB_YOGORE(MAIN_MAP)
			SIF CFLAG:MASTER:현재위치 == LOCAL
				TRYCALLFORM MOB_TURNENDEVENT_{MAIN_MAP}
		ENDIF
	NEXT
ENDIF
IF !FLAG:70 && TARGET
	IF TALENT:TARGET:동물귀 && CFLAG:TARGET:900 == 7 && !TCVAR:TARGET:발정 && PALAM:TARGET:욕정 >= PALAMLV:5 && !TALENT:TARGET:임신
		PRINTFORML %타겟은% 뺨을 상기시키고, 숨을 거칠게 쉬고 있다…
		PRINTFORML 위험일에 극도로 흥분한 %타겟은% [발정]해 버린 것 같다
		TCVAR:TARGET:발정 = 1
	ENDIF
ENDIF

FOR LOCAL,0, CHARANUM
	SIF TIME > TCVAR:LOCAL:쫒겨난시각 + 30
		TCVAR:LOCAL:쫒겨난장소 = 0
	;IF !CFLAG:LOCAL:우후후
		FOR LOCAL:2, 101,109
			SIF TEQUIP:LOCAL:(LOCAL:2) != TEQUIP:LOCAL:(LOCAL:2 + 10)
				TEQUIP:LOCAL:(LOCAL:2) = 0
		NEXT
	;ENDIF
	;일기의 기입 판정
	CALL KOJO_MESSAGE_SEND("DIARY", 0, LOCAL, 0, 0, "BEFORE_CHECK")
	SIF CFLAG:LOCAL:현재위치 == SUKIMA()
		CONTINUE
[SKIPSTART]
	;오늘의 인사를 안 한 경우 조우시 인사 말을 내
	IF LOCAL > 0 && !TCVAR:LOCAL:오늘만났는지 && CFLAG:MASTER:현재위치 == CFLAG:LOCAL:현재위치 && !FLAG:70
		IF CFLAG:LOCAL:안면 == 0
			CALL SYOTAIMEN(LOCAL)
		ELSEIF CHK_DATENOW(CFLAG:MASTER:데이트중)
			CALL KOJO_MESSAGE_SEND("EVENT",1,LOCAL,9,FLAG:데이트상대)
		ELSE
			CALL KOJO_MESSAGE_SEND("EVENT",1,LOCAL,8,0)
		ENDIF
		CALL FirstMeetToday(LOCAL)
	ENDIF
[SKIPEND]
	;의상의 갈아입음 수선은@SHOW_STATUS에 이동
	[SKIPSTART]
	LOCAL:1 = 0
	FOR 장비부위, 0, 23
		SIF (!수면시간(LOCAL) && EQUIP:LOCAL:장비부위 != CFLAG:LOCAL:(200 + 장비부위)) || (수면시간(LOCAL) && EQUIP:LOCAL:장비부위 != CFLAG:LOCAL:(240 + 장비부위))
			LOCAL:1 ++
		SIF 장비부위 == 6 && CFLAG:LOCAL:노팬티 && ((!수면시간(LOCAL) && EQUIP:LOCAL:장비부위 != CFLAG:LOCAL:(200 + 장비부위)) || (수면시간(LOCAL) && EQUIP:LOCAL:장비부위 != CFLAG:LOCAL:(240 + 장비부위)))
			LOCAL:1 --
	NEXT
	IF !CFLAG:LOCAL:우후후 && CFLAG:LOCAL:현재위치 != BATHROOM(CFLAG:LOCAL:현재위치) && LOCAL:1
		IF 수면시간(LOCAL)
			;파자마 세트
			CALL CTRL_CLOTHES_SET(LOCAL, "현재 의상의 변경_파자마")
			CFLAG:LOCAL:파자마 = 1
		ELSE
			;평상복 세트
			CALL CTRL_CLOTHES_SET(LOCAL, "현재 의상의 변경_평상복")
			CFLAG:LOCAL:파자마 = 0
		ENDIF
		SIF !수면시간(LOCAL) && CFLAG:LOCAL:면간발각
			CFLAG:LOCAL:면간발각 = 0
		CALL CLOTHES_SETTING_TRAIN(LOCAL)
	ENDIF
	[SKIPEND]
	;일단 CLOTHES_SETTING_TRAIN는 해 손해는 없을 것이다
		SIF !수면시간(LOCAL) && CFLAG:LOCAL:면간발각
			CFLAG:LOCAL:면간발각 = 0
		CALL CLOTHES_SETTING_TRAIN(LOCAL)
NEXT
;강제적으로 WAIT가 추가되는 것을 회피하기 위해서 TWAIT를 가르친다
;이것으로 입력 회수가 1회줄어든다
IF CFLAG:MASTER:현재위치 == 99
	SETCOLOR C_RED
	PRINTFORMW ### COM_END LOCATION ERROR! %마스터는% 존재하지 않는 장소에 있다! ###
	PRINTW 강제로 하루를 종료합니다. 로그를 뽑아서 Gitgud에 버그 보고를 부탁드립니다.
	RESETCOLOR
	CFLAG:MASTER:현재위치 = CFLAG:MASTER:311
	CALL SET_SLEEP(1, MASTER, 0)
	TFLAG:106 = 0
	BEGIN AFTERTRAIN
ENDIF
	;CFLAG:MASTER:취침시간 = DAY * 1440 + TIME
SIF SELECTCOM == 400
	TWAIT 100,0

@ENKAI_SINKOU
SIF FLAG:연회개최플래그 == 0 || TFLAG:연회상태 == 101
	RETURN

;宴会の参加人数をセット
CALL 연회참가인원수재계산()

;天気が悪いと中止になる宴会もある

IF TIME:5 >= 4 && FLAG:36 > 0 && GROUPMATCH(CFLAG:[[텐시]]:연회참가, 1, 2)
	PRINTFORMW 악천후로 인해 연회가 중지될 줄 알았지만, 텐시의 힘으로 하늘은 맑아졌다
	TIME:5 = 0
	;(天候パッチ)風速を0にする
	WIND_VELOCITY = 0
	;(天候パッチ)1日が終わるまで天候変更を禁止する
	FORBIDDEN_CHANGE_WEATHER = 1
ENDIF
IF TIME:5 >= 4 && FLAG:36 > 0 && DAY >= FLAG:개시일 && TIME >= FLAG:33
	PRINTL
	PRINTFORMW 유감스럽게도, \@ CFLAG:MASTER:현재위치 == TFLAG:연회장 ? 날씨가 안 좋아서 연회는 취소되었습니다 # 악천후로 연회는 취소된 것 같습니다\@
	CALL ENKAI_STOP
;연회의 도중에 내리기 시작했을 경우
ELSEIF TFLAG:연회상태 >= 98 && TIME:5 >= 4 && FLAG:36 > 0
	PRINTL
	PRINTFORMW 유감스럽게도, \@ CFLAG:MASTER:현재위치 == TFLAG:연회장 ? 날씨가 안 좋아서 연회는 취소되었습니다 # 악천후로 연회는 취소된 것 같습니다\@
	CALL ENKAI_STOP
;연회 개시
ELSEIF 연회개시일판정() && TIME >= FLAG:33
	IF CFLAG:MASTER:현재위치 == TFLAG:연회장
		PRINTFORML
		IF FLAG:참가인수 < 2
			PRINTW 조용한 분위기 속에서, 연회가 시작되었습니다…
		ELSEIF FLAG:참가인수 >= 8
			PRINTW 떠들썩한 분위기 속에서, 연회가 시작되었습니다…
		ELSE
			PRINTW 화기애애한 분위기 속에서, 연회가 시작되었습니다…
		ENDIF
		PRINTFORML
		TFLAG:연회상태 = 99
	ELSE
		TFLAG:연회상태 = 98
	ENDIF
ELSEIF TFLAG:연회상태 == 98 && DAY >= FLAG:개시일 && TIME >= FLAG:33
	IF CFLAG:MASTER:현재위치 == TFLAG:연회장
		PRINTL
		PRINTW 이미 연회는 시작된 것 같습니다…
		PRINTL
		TFLAG:연회상태 = 99
	ENDIF
	;주최자측은 아직 연회준비 플래그 인 채의 모아 두어 플래그를 억지로 진행한다
	FOR LOCAL,0,CHARANUM
		SIF CFLAG:LOCAL:20 == 2
			CALL ENKAI_WORK(LOCAL)
	NEXT
ELSEIF 연회개최중판정() && FLAG:참가인수 == 0
	IF CFLAG:MASTER:현재위치 == TFLAG:연회장
		PRINTL
		PRINTW 연회가 열릴 것 같습니다…
		PRINTL
		TFLAG:연회상태 = 101
	ELSE
		TFLAG:연회상태 = 100
	ENDIF
ELSEIF TFLAG:연회상태 == 100 && CFLAG:MASTER:현재위치 == TFLAG:연회장
	PRINTL
	PRINTW 벌써 연회는 끝난 것 같습니다…
	PRINTL
	TFLAG:연회상태 = 101
ENDIF


code:
@EVENTTURNEND
#DIM 양육비

;연회 당일 하루가 끝나면 연회 플래그 OFF
;IF DAY >= FLAG:34 && FLAG:연회개최플래그
IF TIME + 1440 * DAY >= FLAG:개시시간 + 1440 * FLAG:개시일 && FLAG:연회개최플래그
	CALL ENKAI_END()
ENDIF
;령묘 플래그 OFF
FLAG:17 = 0
DRAWLINE
PRINTFORML
;SETCOLOR C_PINK
;FONTBOLD
FOR LOCAL,0,CHARANUM
	;이 3가지는 당신도 취득가능
	IF EXP:LOCAL:키스경험 > 300 && ABL:LOCAL:M감각 > 5 && !TALENT:LOCAL:키스마
		PRINTFORML 수많은 키스를 나누고 마음의 지주를 찾은 %조사처리(CALLNAME:LOCAL,"는")%、키스를 하지 않으면 불안해서 견딜 수 없게 되어 버렸다…
		;陥落素質取得口上,6,キス魔
		CALL KOJO_MESSAGE_SEND("EVENT", 21, LOCAL, 6)
		PRINTFORMW %조사처리(CALLNAME:LOCAL,"는")% [키스마]를 얻었다
		TALENT:LOCAL:키스마 = 1
	ENDIF
	IF EXP:LOCAL:조교자위경험 > 200 && EXP:LOCAL:자위경험 > 500 && !TALENT:LOCAL:자위중독
		PRINTFORML 욕망을 내뱉는 법을 알게 된 %조사처리(CALLNAME:LOCAL,"는")%、늘 자위를 생각하게 되었다…
		;陥落素質取得口上,7,自慰狂い
		CALL KOJO_MESSAGE_SEND("EVENT", 21, LOCAL, 7)
		PRINTFORMW %조사처리(CALLNAME:LOCAL,"는")% [자위중독]을 얻었다
		TALENT:LOCAL:자위중독 = 1
	ENDIF
	IF EXP:LOCAL:자궁구개발경험 > 50 && !TALENT:LOCAL:포르치오성감
		PRINTFORML 계속해서 자궁구를 공격당한 %조사처리(CALLNAME:LOCAL,"는")%, 고통보다는 쾌락을 느끼게 된 것 같다…
		;陥落素質取得口上,8,ポルチオ性感
		CALL KOJO_MESSAGE_SEND("EVENT", 21, LOCAL, 8)
		PRINTFORMW %조사처리(CALLNAME:LOCAL,"는")% [포르치오성감]을 얻었다
		TALENT:LOCAL:포르치오성감 = 1
	ENDIF
;	IF FLAG:おもらし有効 > 0 && EXP:LOCAL:放尿経験 > 1000 * (10 - 2 * TALENT:LOCAL:中毒しやすい) / 10 && !TALENT:LOCAL:おもらし癖
;		PRINTFORML 絶頂時に放尿し続けてきた%CALLNAME:LOCAL%の身体は、今や放尿するのが癖になってしまう程に緩くなってしまった…
;		;陥落素質取得口上,9,おもらし癖
;		CALL KOJO_MESSAGE_SEND("EVENT", 21, LOCAL, 9)
;		PRINTFORMW %CALLNAME:LOCAL%は[おもらし癖]を得た
;		TALENT:LOCAL:おもらし癖 = 1
;	ENDIF
	;이하는 아나타는 건너뛰기
	SIF LOCAL == MASTER
		CONTINUE
	;연모 취득
	IF 연모취득조건(LOCAL)
		;멘레이키는 특별 취급
		IF LOCAL == 57
			PRINTFORMW %조사처리(CALLNAME:LOCAL,"는")% %CALLNAME:MASTER%에 대한 마음을 자각함과 동시에, 입꼬리가 느슨해진다
			PRINTFORMW 자연스러운 미소를 짓는 %조사처리(CALLNAME:LOCAL,"는")%, 표정을 얻은 것 같다
			;陥落素質取得口上,1,恋慕
			CALL KOJO_MESSAGE_SEND("EVENT", 21, LOCAL, 1)
			PRINTFORMW %CALLNAME:LOCAL%의 \@ TALENT:LOCAL:사모 ? [사모]는 # [애욕]은 \@ [연모]가 되었다
			FLAG:10 ++
		;애욕의 유무로 분기
		ELSEIF TALENT:LOCAL:애욕
			PRINTFORMW %조사처리(CALLNAME:LOCAL,"는")% 몸도 마음도 %마스터로% 채워지는 것에 기쁨을 느끼고 있다…
			PRINTL 문란한 관계의 끝에 맺어진 연심, 그것 또한 하나의 사랑의 형태…
			;陥落素質取得口上,1,恋慕
			CALL KOJO_MESSAGE_SEND("EVENT", 21, LOCAL, 1)
			PRINTFORMW %CALLNAME:LOCAL%의 [애욕]은 [연모]가 되었다
		ELSE
			PRINTFORML %조사처리(CALLNAME:LOCAL,"는")% %마스터와% 보내는 시간에 행복을 느끼는 것 같다…
			;陥落素質取得口上,1,恋慕
			CALL KOJO_MESSAGE_SEND("EVENT", 21, LOCAL, 1)
			PRINTFORMW %CALLNAME:LOCAL%의 [사모]는 [연모]가 되었다
			FLAG:10 ++
		ENDIF
		SIF !MARK:LOCAL:시간각인 && !TALENT:LOCAL:애욕 && !CFLAG:LOCAL:수면간 && !CFLAG:LOCAL:만취간
			CFLAG:LOCAL:깨끗한교제 = 1
		TALENT:LOCAL:연모 = 1
		TALENT:LOCAL:사모 = 0
		TALENT:LOCAL:애욕 = 0
		TALENT:LOCAL:섹프 = 0

		;악보 입수 처리
		PRINTL
		RESULT = 악보입수_함락(LOCAL)
		SIF RESULT != -1
			PRINTL
	;사모 취득
	;섹프/정부의 취득으로 분기
	ELSEIF 사모취득조건(LOCAL)
		IF TALENT:LOCAL:섹프
			PRINTFORML %조사처리(CALLNAME:LOCAL,"는")% %CALLNAME:MASTER%의 신체 이외에도 흥미가 있는 것 같다
			;陥落素質取得口上,2,思慕
			CALL KOJO_MESSAGE_SEND("EVENT", 21, LOCAL, 2)
			PRINTFORMW %조사처리(CALLNAME:LOCAL,"는")% [사모]를 얻었다
		ELSE
			PRINTFORML %조사처리(CALLNAME:LOCAL,"는")% %마스터가% 신경쓰이는 것 같다…
			;陥落素質取得口上,2,思慕
			CALL KOJO_MESSAGE_SEND("EVENT", 21, LOCAL, 2)
			PRINTFORMW %조사처리(CALLNAME:LOCAL,"는")% [사모]를 얻었다
		ENDIF
		TALENT:LOCAL:사모 = 1
	;애욕 취득
	;봉사쾌락의 분기가 아직 없으므로, 취득조건에 병합
	ELSEIF 애욕취득조건(LOCAL)
		;사모하며, 진심으로 신체를 선택한
		IF TALENT:LOCAL:사모
			PRINTFORML %조사처리(CALLNAME:LOCAL,"는")% %마스터와% 마음 이상으로 몸의 연결에 만족하고 있는 것 같다…
			PRINTFORML 눈에 보이지 않는 정보다도 눈에 보이는 증표를 원하며 앞으로도 육체를 겹칠 것이다
			;陥落素質取得口上,3,愛欲
			CALL KOJO_MESSAGE_SEND("EVENT", 21, LOCAL, 3)
			PRINTFORMW %조사처리(CALLNAME:LOCAL,"는")% [애욕]을 얻었다
		ELSE
			PRINTFORML %조사처리(CALLNAME:LOCAL,"는")% %마스터와%의 정욕에 빠져있는것 같다…
			;陥落素質取得口上,3,愛欲
			CALL KOJO_MESSAGE_SEND("EVENT", 21, LOCAL, 3)
			PRINTFORMW %조사처리(CALLNAME:LOCAL,"는")% [애욕]을 얻었다
		ENDIF
			FLAG:10 ++
			TALENT:LOCAL:애욕 = 1
			TALENT:LOCAL:사모 = 0

			;악보 입수 처리
			PRINTL
			RESULT = 악보입수_함락(LOCAL)
			SIF RESULT != -1
				PRINTL
	;섹프/정부 취득
	;섹프와 정부의 차이는 사모의 유무
	ELSEIF 섹프취득조건(LOCAL)
		;사모 취득 완료＋섹프 미취득
		IF TALENT:LOCAL:사모 && !TALENT:LOCAL:섹프
			PRINTFORML %조사처리(CALLNAME:LOCAL,"는")% %마스터라%면 쾌락에 빠져도 괜찮다고 생각하기 시작한 것 같다…
			;陥落素質取得口上,4,愛人
			CALL KOJO_MESSAGE_SEND("EVENT", 21, LOCAL, 4)
			PRINTFORMW %조사처리(CALLNAME:LOCAL,"는")% %CALLNAME:MASTER%의 [정부]가 되었다
			TALENT:LOCAL:섹프 = 2
		;사모 취득 완료＋섹프  취득 완료
		ELSEIF TALENT:LOCAL:사모 && TALENT:LOCAL:섹프 == 1
			PRINTFORML %조사처리(CALLNAME:LOCAL,"는")% %CALLNAME:MASTER%에게 단순한 육체 관계 이상의 것을 바라고 있는 것 같다…
			;陥落素質取得口上,4,愛人
			CALL KOJO_MESSAGE_SEND("EVENT", 21, LOCAL, 4)
			PRINTFORMW %조사처리(CALLNAME:LOCAL,"는")% [정부]가 되었다
			TALENT:LOCAL:섹프 = 2
		;사모 미취득＋섹프 미취득이라면 우선 섹프 취득
		ELSEIF !TALENT:LOCAL:섹프
			PRINTFORML %조사처리(CALLNAME:LOCAL,"는")% %마스터가% 주는 쾌락에 빠져들고 있다…
			;陥落素質取得口上,5,セフレ
			CALL KOJO_MESSAGE_SEND("EVENT", 21, LOCAL, 5)
			PRINTFORMW %조사처리(CALLNAME:LOCAL,"는")% %CALLNAME:MASTER%의 [섹프]가 되어버렸다
			TALENT:LOCAL:섹프 = 1
		ENDIF
	ENDIF
	IF EXP:LOCAL:연주경험 >= 1000 && ABL:LOCAL:기교 > 3 && !TALENT:LOCAL:음악지식
		PRINTFORMW %조사처리(CALLNAME:LOCAL,"는")% 악기 연주에 조예가 깊어졌다
		PRINTFORMW %조사처리(CALLNAME:LOCAL,"는")% [악기지식]을 얻었다
		;PRINTFORMW %조사처리(CALLNAME:LOCAL,"는")% 음악에 조예가 깊어졌다
		;PRINTFORMW %조사처리(CALLNAME:LOCAL,"는")% [음악지식]을 얻었다
		TALENT:LOCAL:음악지식 = 1
		;技能レベルアップ口上,7-1,音楽知識
		CALL KOJO_MESSAGE_SEND("EVENT", 34, LOCAL, 7, 1)
	ELSEIF EXP:LOCAL:연주경험 >=5000 && !TALENT:LOCAL:음감
		PRINTFORMW %CALLNAME:LOCAL%의 음감은 상당히 단련된 것 같다
		PRINTFORMW %조사처리(CALLNAME:LOCAL,"는")% [상대음감]을 얻었다
		TALENT:LOCAL:음감 = 1
		;技能レベルアップ口上,7-2,相対音感
		CALL KOJO_MESSAGE_SEND("EVENT", 34, LOCAL, 7, 2)
	ENDIF
	IF CFLAG:LOCAL:언짢음
		CFLAG:LOCAL:언짢음 --
		IF !CFLAG:LOCAL:언짢음
			SIF MARK:LOCAL:반발각인
				CFLAG:LOCAL:언짢음 ++
		ENDIF
	ENDIF
	;EX캐릭터 연모 시에는 멤버 추가 이벤트 플래그를 세운다
	IF !CFLAG:LOCAL:신사거주
		SIF TALENT:LOCAL:연모 && EXP:LOCAL:애정경험 >= 200
			FLAG:1 = 3
	ENDIF
NEXT

IF EXP:MASTER:이상청소경험 > (500 + 100 * FLAG:850) && ABL:MASTER:청소기능 > 3 && !TALENT:MASTER:청소중독
	PRINTFORMW %마스터는% 무서운 것을 깨닫고 말았다
	PRINTFORMW 자나 깨나, 시간을 멈춘 사이나, 소녀들과 살을 겹치고 있을 때조차도,
	PRINTFORMW 방이 어질러져 있지는 않은지 신경이 쓰여서 견딜 수 없다
	PRINTFORMW 벌써 %CALLNAME:MASTER%의 마음이 편안해지는 것은 이렇게 청소를 하는 동안 뿐
	PRINTFORMW %CALLNAME:MASTER%의 청소에 대한 집착은, 돌이킬 수 없는 경지에 도달해 있었다
	PRINTFORMW %마스터는% [청소중독]을 얻었다
	TALENT:MASTER:청소중독 = 1
ENDIF
;MASTER을 제외한 전원의 프리셋 처리
FOR LOCAL,1,CHARANUM
	;기력의 회복
	BASE:LOCAL:기력 = MAXBASE:LOCAL:기력

	LOCAL:1 = MAXBASE:LOCAL:체력 / 3

	;회복 빠름, 회복 느림
	LOCAL:1 = LOCAL:1 * 3 / (3 - TALENT:LOCAL:회복속도)

	;동침하고 있으면 회복속도 향상
	SIF CFLAG:LOCAL:동침중
		LOCAL:1 *= 2
	SIF CFLAG:LOCAL:철야
		LOCAL:1 /= 2
	CFLAG:LOCAL:철야 = 0
	BASE:LOCAL:체력 = MIN(MAXBASE:LOCAL:체력, BASE:LOCAL:체력 + LOCAL:1)

	;나무삼 카드 발행자가 있으면 발행 플래그 OFF
	CFLAG:LOCAL:328 = 0
NEXT
;전원의 프리셋 처리
FOR LOCAL,1,CHARANUM
	BASE:LOCAL:정력 = MIN(MAXBASE:LOCAL:정력, BASE:LOCAL:정력 + MAXBASE:LOCAL:정력 * BASE:LOCAL:체력 / MAXBASE:LOCAL:체력)

	;은밀 중인 캐릭터가 있으면 플래그 OFF
	CALL SET_SNEAK(0,LOCAL,1)
	;CFLAG:LOCAL:은밀중 = 0

	;장난 중인 캐릭터가 있으면 플래그 OFF
	CALL SET_PRANK(0,LOCAL,1)
	;CFLAG:LOCAL:장난 = 0

NEXT

;연회의 개최 판정
SIF FLAG:3 && FLAG:연회개최플래그 == 0
	CALL ENKAI_SETTING

IF CFLAG:MASTER:육아인원수
	양육비 = CFLAG:MASTER:육아인원수 * 300 / (1 + FLAG:생활능력무) ;원판은 인당 1000원
	PRINTFORML %마스터는% 자녀 양육비로 \\{양육비} 지불했다.
	MONEY -= 양육비
ENDIF

CALL RANK_POINT
PRINTW

FLAG:0 = 0

;모브코의 세이브
IF TALENT:RANDOM_CHARANUM:스침 == 1 && GETBIT(TFLAG:일일일회, 9)
	IF FORGET_MOBGIRL
		FORGET_MOBGIRL = 0
		GOTO FORGET_MOB2
	ENDIF
	PRINTFORML %마스터는% 오늘 만난 %조사처리(CALLNAME:RANDOM_CHARANUM,"를")%……
	CALL ASK_YN("잊는다","기억해 둔다")
	IF RESULT
		VARSET TCVAR:RANDOM_CHARANUM:0, 0
		VARSET EX:RANDOM_CHARANUM:0, 0
		VARSET TEQUIP:RANDOM_CHARANUM:0, 0
		VARSET PALAM:RANDOM_CHARANUM:0, 0
		CALL InitializeBeforeTrain(RANDOM_CHARANUM)
		CFLAG:RANDOM_CHARANUM:900 ++
		CALL SAVE_MOBGIRL
		SIF RESULT == -1
			GOTO FORGET_MOB
		PRINTFORMW %마스터는% %조사처리(CALLNAME:RANDOM_CHARANUM,"를")% \@ TALENT:RANDOM_CHARANUM:풍속양 ? 다시 즐기고 싶다고 # 다시 만났으면 좋겠다고 \@ 생각했다
	ELSE
		$FORGET_MOB
		PRINTFORMW %마스터는% %조사처리(CALLNAME:RANDOM_CHARANUM,"를")% 깨끗히 잊었다
	ENDIF
ENDIF
$FORGET_MOB2
IF TALENT:RANDOM_CHARANUM:스침 == 2 && GETBIT(TFLAG:일일일회, 9)
	CALL MOBGIRL_PROFILE2()
	VARSET TCVAR:RANDOM_CHARANUM:0, 0
	VARSET EX:RANDOM_CHARANUM:0, 0
	VARSET TEQUIP:RANDOM_CHARANUM:0, 0
	VARSET PALAM:RANDOM_CHARANUM:0, 0
	CALL InitializeBeforeTrain(RANDOM_CHARANUM)
	SAVECHARA TOSTR(CFLAG:RANDOM_CHARANUM:모브코세이브슬롯), CSTR:RANDOM_CHARANUM:모브코프로필1, RANDOM_CHARANUM
	[IF DEBUG]
		PRINTFORMW %CALLNAME:RANDOM_CHARANUM%의 정보를 갱신했습니다.
	[ENDIF]
ENDIF

BEGIN SHOP

@RANK_POINT
VARSET LOCAL
;청결도
LOCAL:2 = EX:MASTER:오늘의청소회수 / 10
PRINTFORML 오늘 청소한 쓰레기의 양 {EX:MASTER:오늘의청소회수}
PRINTFORML 청소로 얻은 보너스 \\{LOCAL:2}
IF DAY:2 == 4 && DAY:3 == 31
	PRINT (x2)  【대청소 보너스】
	LOCAL:2 *= 2
ENDIF
;거점 주인의 호감도와 신뢰도가 상승
LOCAL:3 = MIN(10, EX:MASTER:오늘의청소회수 / 2000)
IF LOCAL:3
	PRINTFORML 청소를 열심히 한 것으로, %CALLNAME:거점_주인%의 평가가 올라갔다
	CFLAG:거점_주인:호감도 += LOCAL:3 * 2
	CFLAG:거점_주인:신뢰도 += LOCAL:3
	IF LOCAL:3 / 3 > 0
		PRINTFORML 청소를 열심히 하니、추가로 싸구려식재를 {LOCAL:3 / 3}개 받았다
		ITEM:싸구려식재 += LOCAL:3 / 3
	ENDIF
ENDIF
PRINTL
;심부름 보수
LOCAL:4 = TCVAR:MASTER:도운양

SIF LOCAL:4 > 0
	PRINTFORML 업무 도우미로 얻은 보너스 \\{LOCAL:4}

MONEY += LOCAL:2 + LOCAL:4

;호감도 상승
IF FLAG:12 && MONEY:2 < CM_LIMIT
	LOCAL:5 = MAX(1,FLAG:12 / 100)
	LOCAL:5 = MIN(LOCAL:5, CM_LIMIT - MONEY:2)
	PRINTFORML 호감도 상승 보너스 ＋{LOCAL:5} 카리스마
	MONEY:2 += LOCAL:5
	PRINTFORML 현재 카리스마：%세자리구분(MONEY:2)%　카리스마 보유 상한：%세자리구분(CM_LIMIT)%
ENDIF
FLAG:12 = 0

PRINTW

@연모취득조건(ARG)
#FUNCTION
SIF TALENT:ARG:연모
	RETURNF 0
SIF CFLAG:ARG:호감도 < REQUIRED_FAVOR_연모
	RETURNF 0
SIF CFLAG:ARG:신뢰도 < REQUIRED_TRUST_연모
	RETURNF 0
SIF ABL:ARG:친밀 < REQUIRED_AFFECTION_연모
	RETURNF 0
SIF MARK:ARG:반발각인
	RETURNF 0
SIF TALENT:ARG:애욕 && !FLAG:함락노선변화
	RETURNF 0
RETURNF 1

@사모취득조건(ARG)
#FUNCTION
SIF TALENT:ARG:애욕
	RETURNF 0
SIF TALENT:ARG:연모
	RETURNF 0
SIF TALENT:ARG:사모
	RETURNF 0
SIF CFLAG:ARG:호감도 < REQUIRED_FAVOR_사모
	RETURNF 0
SIF CFLAG:ARG:신뢰도 < REQUIRED_TRUST_사모
	RETURNF 0
SIF ABL:ARG:친밀 < REQUIRED_AFFECTION_사모
	RETURNF 0
SIF TALENT:ARG:섹프 && !FLAG:함락노선변화
	RETURNF 0
RETURNF 1

@애욕취득조건(ARG)
#FUNCTION
SIF TALENT:ARG:연모
	RETURNF 0
SIF TALENT:ARG:애욕
	RETURNF 0
SIF !TALENT:ARG:섹프
	RETURNF 0
SIF CFLAG:ARG:호감도 < REQUIRED_FAVOR_애욕
	RETURNF 0
SIF ABL:ARG:욕망 < REQUIRED_DESIRE_애욕
	RETURNF 0
SIF (EXP:ARG:절정경험 - EXP:ARG:무자각절정경험) < REQUIRED_ECSTACY_애욕
	RETURNF 0
SIF (ABL:ARG:욕망 - ABL:ARG:친밀) < 1
	RETURNF 0
RETURNF 1

@섹프취득조건(ARG)
#FUNCTION
SIF TALENT:ARG:연인
	RETURNF 0
SIF TALENT:ARG:연모
	RETURNF 0
SIF TALENT:ARG:섹프 == 2
	RETURNF 0
SIF CFLAG:ARG:호감도 <= REQUIRED_FAVOR_섹프
	RETURNF 0
SIF ABL:ARG:욕망 < REQUIRED_DESIRE_섹프
	RETURNF 0
SIF (EXP:ARG:절정경험 - EXP:ARG:무자각절정경험) < REQUIRED_ECSTACY_섹프
	RETURNF 0
SIF (ABL:ARG:욕망 - ABL:ARG:친밀) < 0
	RETURNF 0
RETURNF 1

license: mit


code:
;신사라면, 역시 운세 뽑기죠
;그런 이유로 운세 뽑기 시간입니다

;LOCAL 0:전체운, 1:금전운, 2:건강운, 3:연애운, 4:여행운, 99:덤으로 한마디
;의 각 항목의 선택 번호.
;무엇을 뽑았는지 참조해서 나중에 처리할 수 있는 구조로 해두었기에, 추가하고 싶을 경우는 각 항목의 마지막에 덧붙여 씁시다
@FORTUNE
;우선은 제비의 결과를 보여드리죠
;전체운, 소재는 조만간 늘리겠습니다
PRINT 　전체운：
PRINTDATAW LOCAL:0
	DATAFORM 대길
	DATAFORM 중길
	DATAFORM 소길
	DATAFORM 길
	DATAFORM 반길
	DATAFORM 말길
	DATAFORM 말소길
	DATAFORM 흉
	DATAFORM 소흉
	DATAFORM 반흉
	DATAFORM 말흉
	DATAFORM 대흉
	DATAFORM
	DATAFORM 토길 (兎吉)
	DATAFORM 견길 (犬吉)
	DATAFORM 묘길 (猫吉)
	DATAFORM 조길 (釣吉)
	DATAFORM 당첨
	DATAFORM 꽝
	DATAFORM 오늘의 말린 오징어
	DATAFORM 광 (狂)
	DATAFORM 향 (鄕)
	DATAFORM 모후길
ENDDATA

PRINT 　금전운：
PRINTDATAW LOCAL:1
	DATAFORM 없는 지갑을 털지 말 것
	DATAFORM 낭비할 것 같다. 주의해야 할 것
	DATAFORM 임시수입을 들어오 예감 있음
	DATAFORM 견실이 제일
	DATAFORM 투자한다면 먼저 자기 자신에게 할 것
	DATAFORM 지갑끈을 조여라
	DATAFORM 돌고 도는 것, 사용하면 돌아온다
	DATAFORM 길을 잘못 들지 않으면 낭비는 길
	DATAFORM 낭비는 흉, 지금은 참을 때
	DATAFORM 잃어버린 것이 발견된다
	DATAFORM 좋은 이야기일수록 의심할 것
	DATAFORM 하면 이루어진다
	DATAFORM 해도 이루어지지 않는다
	DATAFORM 한 대로 요구해라
	DATAFORM 보답을 요구하지 말지어다
	DATAFORM 포기해라
	DATAFORM 마음 가는 대로 하라
	DATAFORM 왼쪽 매매 또한 재미
	DATAFORM 지식에 쏟아부을 것
	DATAFORM 습득물에 불행 있다
	DATAFORM 습득물에 행운 있다
	DATAFORM 흘린 물건에 불행 있다
	DATAFORM 흘린 물건에 행운 있다
	DATAFORM 1회 휴식
	DATAFORM 지불에 주의
	DATAFORM 수상쩍은 이야기에야말로 활로 있다
	DATAFORM 수상쩍은 이야기는 흉하다
	DATAFORM 뭐…… 힘내라
	DATAFORM 사치에 교만하지 말지어다
	DATAFORM 고간의 보물을 조심해라
ENDDATA

PRINT 　건강운：
PRINTDATAW LOCAL:2
	DATAFORM 오늘 밤은 일찍 자
	DATAFORM 밤샘도 때로는 보탬이 된다
	DATAFORM 일찍 일어나는 새가 벌레를 잡는다
	DATAFORM 일찍 일어나면 나쁜 소식을 듣는다
	DATAFORM 새끼발가락에 주의
	DATAFORM 책상 모서리를 경계하라
	DATAFORM 머리 위 주의
	DATAFORM 발 밑 주의
	DATAFORM 전방 주의
	DATAFORM 후방 주의
	DATAFORM 좌우 주의
	DATAFORM 동방 주의
	DATAFORM 서방 주의
	DATAFORM 북방 주의
	DATAFORM 남방 주의
	DATAFORM 중앙 주의
	DATAFORM 난입 주의
	DATAFORM 식사에 어려움 있다
	DATAFORM 식사는 모든 건강의 근원
	DATAFORM 술은 백약지장
	DATAFORM 술은 백독지장
	DATAFORM 마시고 삼키고 신나게 마실 것
	DATAFORM 마셔도 술에 먹히지 말지어다
	DATAFORM 될 대로 된다
	DATAFORM 될 대로 안 된다
	DATAFORM 될 대로밖에 안 된다
	DATAFORM 될 대로도 안 된다
	DATAFORM 문제 없음
	DATAFORM 문제뿐. 경계를 게을리하지 말도록
	DATAFORM 칼부림 사태를 주의하라
	DATAFORM 사소한 계기가 신세를 망친다
	DATAFORM 사소한 계기가 신세를 부지한다
	DATAFORM 희망은 버리지 말지어다
	DATAFORM 무모한 짓은 새로운 문을 연다
	DATAFORM 무모한 짓은 하지 않고 안정을
	DATAFORM 주워먹는 짓은 삼가라
	DATAFORM 먹는 것에 불안 없다
	DATAFORM 차린 밥상은 소녀의 고집
	DATAFORM 치린 밥상을 먹지 않는 건 고자뿐
	DATAFORM 매일의 훈련을 빠뜨리지 말 것
	DATAFORM 과도한 훈련은 몸을 망친다
	DATAFORM 쿵후가 부족합니다
	DATAFORM 쿵후가 충분합니다
ENDDATA

PRINT 　연애운：
PRINTDATAW LOCAL:3
	DATAFORM 자신이 믿은 길을 뚫고 나아가라
	DATAFORM 믿어 돌진한 미래가 완전하게 예상외로 사랑의 소망을 끊기고 단절 상태가 보내져 오는 것은
	DATAFORM 지금은 그때가 아니다
	DATAFORM 지금이지
	DATAFORM 언제냐니 지금이지
	DATAFORM 동성에게 인기 있을 조짐
	DATAFORM 이성에게 끌릴 조짐
	DATAFORM 다만 진행해 단지 진행하고
	DATAFORM 급돌아라
	DATAFORM 부딪쳐서 깨져라
	DATAFORM 부딪치면 깨진다
	DATAFORM 지금은 뽑아
	DATAFORM 지금은 눌러
	DATAFORM 이루어지지 않고, 신식연을 찾을 것
	DATAFORM 이별은 만남의 시작이라
	DATAFORM 만남은 이별의 시작이라
	DATAFORM 깨닫지 못한 새에 사랑은 여문다
	DATAFORM 포기해
	DATAFORM 원점으로 되돌아 간다
	DATAFORM 뭐…… 화이팅, 이야
	DATAFORM 봄이 오다
	DATAFORM 여름이 오다
	DATAFORM 질려오다
	DATAFORM 겨울이 오다
	DATAFORM 새 봄이 왔다
	DATAFORM 호의의 뒤를 의심하지 말 것
	DATAFORM 호의의 뒤를 의심할 것
	DATAFORM 본심을 털어놓아, 말하지만 좋아
	DATAFORM 배의 나를 찾지 말 것
	DATAFORM 본심을 털어놓으면 배 할애해진다
	DATAFORM 비 내리지 않고와도 땅 굳어진다
	DATAFORM 비라고 땅 굳어지지 않고
	DATAFORM 그 사랑은 환상에 지나지 않는, 현실을 다시 봐라
	DATAFORM 그 사랑은 이차원에 지나지 않는, 삼차원으로 돌아간다
	DATAFORM 지금은 알 때
	DATAFORM 지금은 히나 때
	DATAFORM 자립 때야말로 가장 경계할 때
	DATAFORM 자립한 새는 둥지에 있지 않고
	DATAFORM 자립한 새는 둥지로 돌아간다
	DATAFORM 마음은 전해진다
	DATAFORM 마음은 비뚤어진다
	DATAFORM 마음은 전해지지 않고
	DATAFORM 마음의 일그러짐은 인연도 비뚤어지게 한다
	DATAFORM 사랑은 종족조차 넘어선다
	DATAFORM 종족의 벽 두텁도다
	DATAFORM 욕망은 이성을 넘는다
	DATAFORM 이성은 욕망을 억누른다
	DATAFORM 사랑은 이성을 녹인다
	DATAFORM 이성은 사랑을 억누른다
	DATAFORM 욕망은 사랑을 변질시킨다
	DATAFORM 사랑은 욕망을 맑게 개이게 한다
	DATAFORM 사랑과 욕망이 손을 잡았다. 바라는 대로 마음을 성취하라
	DATAFORM 상대를 돌보는 형태가 좋다
	DATAFORM 상대를 억제하는 형태가 좋다
	DATAFORM 당신을 억제하는 형태가 좋다
	DATAFORM 자신의 생각하는 대로 가는 것이 길하다
	DATAFORM 상대에게 맡기는 것이 좋아
	DATAFORM 마음은 몸에 딸려간다
	DATAFORM 몸은 마음에 딸려간다
	DATAFORM 육욕이 가는 대로
	DATAFORM 육욕에 내맡기는 것은 재앙의 화근
ENDDATA

PRINT 　여행운：
PRINTDATAW LOCAL:4
	DATAFORM 견문을 넓히는 여행을 떠나라
	DATAFORM 집에 있는 것이 길하다
	DATAFORM 인간마을에 흉조
	DATAFORM 인간마을에 길조
	DATAFORM 오랜 유적이 좋다
	DATAFORM 오랜 유적에 다가가지 말라
	DATAFORM 절은 좋은 기로 가득하다
	DATAFORM 절에 침전물 서로 봐
	DATAFORM 안개 강하게 흐린다
	DATAFORM 안개 개인다
	DATAFORM 붉은 저택에 좋은 꿈 있음
	DATAFORM 붉은 저택에 악몽 있음
	DATAFORM 대나무 끝에 빛 있음
	DATAFORM 대나무 속에 미혹 있음
	DATAFORM 숲에 가고는 양식
	DATAFORM 숲에 행 재앙 있어
	DATAFORM 저승에 좋은 서로 불러
	DATAFORM 저승에 초대되고 바구니점괘 있어
	DATAFORM 산으로 향하는 게 좋다
	DATAFORM 산으로 향하지 않는 게 좋다
	DATAFORM 산 꼭대기는 쾌청하다
	DATAFORM 산 꼭대기는 황천하다
	DATAFORM 땅밑은 고요하다
	DATAFORM 땅밑이 소란스럽다
	DATAFORM 예정을 세우면 걱정 없음
	DATAFORM 하늘을 가는 것이 길하다
	DATAFORM 생각하지만 진에 가지만 좋다
	DATAFORM 조난하지만 살아남는다
	DATAFORM 예정을 생각하지 않고 가는 것도 좋다
ENDDATA

PRINT 　한마디：
PRINTDATAW LOCAL:99
	DATAFORM 당첨, 한 장 더
	DATAFORM 네, 한 번 더
	DATAFORM 1주간 뒤에 한 번 더 뽑으러 오세요. 진정한 운세 뽑기를 보여드리겠습니다
	DATAFORM 그대는 이 결과를 믿어도 되고, 내일도 한 번 더 뽑아도 된다
	DATAFORM 믿는 것도 믿지 않는 것도 당신하기 나름
	DATAFORM 믿는 자는 구원받는다
	DATAFORM 결과에 불만이라면 내일 또 와주십시오
	DATAFORM ……뒤에 (문장은 여기서 끊겨 있다)
	DATAFORM 액을 털어내고 싶으면 지금 당장 새전을!
	DATAFORM 고민을 털어놓는 것도 때로는 필요합니다
	DATAFORM 마음은 그때가 올 때까지 숨겨두자
	DATAFORM 환상향　아아 환상향　환상향
	DATAFORM 이 운세를 읽었을 때, 과연 나는 이 세상에 아직 남아있을까요?
	DATAFORM 소재 내놔, 얼렁
	DATAFORM 덧붙여 당 신사의 새전함에는 아직 약간의 여유가 있습니다
	DATAFORM b의 f말할 수 있는Vhr, c2m0응fh4；l(이해를 거부하게 하는 뭔가가 적혀 있었다)
	DATAFORM 맞는 것도 팔괘, 안 맞는 것도 팔괘
	DATAFORM ↑↑↓↓←→←→봄샤ㅅ(피츙―)
	DATAFORM 진척 어때요?
	DATAFORM 진척 안 돼요
ENDDATA

;확인용. 각각 몇 번째의 것이 놓쳤는지 콘솔에 기록
;DEBUGPRINTFORML FORTUNE:{LOCAL:0, 2}, {LOCAL:1, 2}, {LOCAL:2, 2}, {LOCAL:3, 2}, {LOCAL:4, 2}, {LOCAL:99, 2}

;현상은 표시만
;운세 결과에 응해 어떠한 액션을 일으킨다면 이쪽을 묻어 주세요
;각 항목의 n-1행째가 각 변수에 격납되고 있습니다

SELECTCASE LOCAL:0
	CASE 0,19
		TFLAG:행운보정 = 5
	CASE 13 TO 15
		TFLAG:행운보정 = 2
		TFLAG:귀보정 = LOCAL:0 - 11
	CASE 1 TO 3, 16, 17
		TFLAG:행운보정 = 3
	CASE 4 TO 6
		TFLAG:행운보정 = 1
	CASE 7 TO 10
		TFLAG:행운보정 = -3
	CASE 11,18
		TFLAG:행운보정 = -5
	CASEELSE
ENDSELECT

SELECTCASE LOCAL:1
CASEELSE
ENDSELECT

SELECTCASE LOCAL:2
CASEELSE
ENDSELECT

SELECTCASE LOCAL:3
CASEELSE
ENDSELECT

SELECTCASE LOCAL:4
CASEELSE
ENDSELECT

SELECTCASE LOCAL:99
	;맞았으므로 이제(벌써) 한 장(다음의 제비를 뽑았을 때에 캐쉬 백)
	;다만, 정기 계약중 이라면 캐쉬 백 이야기(처리가 귀찮아서)
	CASE 0 TO 2
		SIF !FLAG:운세뽑기계약
			SETBIT FLAG:999, 10
CASEELSE
ENDSELECT

RETURN 0


code:
;	GIFT.ERB
;		Version 0.03-0. 5
;		원래 proto판의 바리안트에 더욱 proto판의 패치를 만든다든가 무엇 생각하는거야 이녀석, 대용품
;		변수 A, B를 고쳐 쓸 뿐만 아니라 CFLAG:990을 마음대로 사용하고 있기 (위해)때문에, 본체의 버전 업에
;		따라 수정 필수
;		그것을 밟은 위에 당신은 이 패치를 도입해도 되고 도입하지 않아도 괜찮다
;		일단, 얻은 팬티의 소유자와 종류는 기록하고 있기 (위해)때문에, 후에 무언가에 사용할 수 있는 것 같은 생각이 들지 않는 것도 아니다
;		현상, 관리가 귀찮아서 당신이 취침중 때에 취침중이 아니었다 노예 마셔 실행 가능, 이라고 한다
;		묶음새가 미실장
;
;		패치 작성이라든지 처음에 어째서 소스가 더러운데는 용서를(·ω·｀
;
;		사용 변수
;			A, B
;
;		사용 CFLAG
;			990	선물 된 팬티의 취득 판정에 이용한다. 노예 옆으로부터는 「선물 한 팬티의 종류」라고 말하는 취급
;				본체의 (분)편으로 이용하게 되면(자) 확실히 번호 변경
;
;		향후 실장 예정
;			어제 신고 있던 것을 선물, 그게 아니라, 지금 신고 있는 것을 벗어 그 자리에 남겼다는 판도 만들고 싶다
;
;		중요한 말
;			예정은 미정
;
;		-- 이력 --
;			2010/07/18	ProtoVer0. 03용 0.05 공개
;
@GET_GIFT
VARSET LOCAL
LOCAL = RAND:(CHARANUM-1) + 1
;호감도 1000이상, 트렁크스 외의 속옷을 입고 있다
SIF CFLAG:LOCAL:2 < 1000 || EQUIP:LOCAL:6 < 2
	RETURN 0
SIF RAND:10 > 0
	RETURN 0

;묘사
DRAWLINE
PRINTFORMW %마스터가% 눈을 뜨니 머리맡에 뭔가 눈에 익지 않은 것이 놓여 있었다
IF RAND:3 == 2
	PRINTFORM 난잡하게
ELSEIF RAND:2 == 1
	PRINTFORM 정중하게
ELSE
	PRINTFORM 대충
ENDIF
PRINTFORM 개어둔 그것을 펼쳐보니, 그것은 아무래도 %CALLNAME:LOCAL%가 소유한
PRINTFORMW %PANTSNAME(EQUIP:LOCAL:6, LOCAL)% 같다
PRINTFORMW 미묘하게 주름이 남은 그것은, 아무래도 %CALLNAME:LOCAL%가 하루 신은 뒤의 것임이 사
PRINTFORMW %마스터는% 그것을 책상의 서랍에 넣고는, 아무 일도 없었던 것처럼 몸치장을 개시했다

PRINTFORML
PRINTFORMW %CALLNAME:LOCAL%의 %PANTSNAME(EQUIP:LOCAL:6, LOCAL)%를 입수했습니다
PRINTFORML

[SKIPSTART]
LOCALS = %CALLNAME:LOCAL%{EQUIP:LOCAL:6}/
SIF SAVESTR:10 == ""
	SAVESTR:10 = /
SAVESTR:10 += LOCALS
[SKIPEND]
CFLAG:LOCAL:(EQUIP:LOCAL:6) += 1


code:
#DIM UFUFU_START_TIME
#DIM UFUFU_START_DAY


code:
;-------------------------------------------------
;함수명 :FESTIVAL
;개요:행사를 나타내는 캐릭터라인을 취득하는 함수
;인수:없음
;-------------------------------------------------
@FESTIVAL
#FUNCTIONS
IF DAY:2 == 1
	SELECTCASE DAY:3
		CASE 1
			RETURNF "설날" ;元日
		CASE 3
			RETURNF "히나마츠리" ;雛祭り
		CASE 11
			RETURNF "어버이날" ;母の日
		CASE 13
			RETURNF "화이트 데이 전날" ;ホワイトデー前日
		CASE 14
			RETURNF "화이트 데이" ;ホワイトデー
		CASE 25
			RETURNF "예대제" ;例大祭
		CASE 28
			RETURNF "바니의날" ;バニーの日
	ENDSELECT
ELSEIF DAY:2 == 2
	SELECTCASE DAY:3
		CASE 7
			RETURNF "칠석" ;七夕
		CASE 14
			RETURNF "추석" ;お盆
		CASE 28
			RETURNF "바니의날" ;バニーの日
	ENDSELECT
ELSEIF DAY:2 == 3
	SELECTCASE DAY:3
		CASE 1
			RETURNF "풍신 「입춘으로부터 210일째 되는 날」" ;"風神「二百十日」
		CASE 15
			RETURNF "보름" ;十五夜
		CASE 27
			RETURNF "수확제" ;収穫祭
		CASE 28
			RETURNF "바니의날" ;バニーの日
		CASE 30
			RETURNF "할로윈" ;ハロウィン
	ENDSELECT
ELSEIF DAY:2 == 4
	SELECTCASE DAY:3
		CASE 3
			RETURNF "절분" ;節分
		CASE 14
			RETURNF "발렌타인 데이" ;バレンタインデー
		CASE 24
			RETURNF "크리스마스 이브" ;クリスマスイヴ
		CASE 25
			RETURNF "크리스마스" ;クリスマス
		CASE 28
			RETURNF "바니의날" ;バニーの日
		CASE 31
			RETURNF "그믐날" ;大晦日
	ENDSELECT
ENDIF

;특정의 축제의 기상시 메시지
@SEASONAL_MORNING
#DIM 생활시간보정, 5
SELECTCASE FESTIVAL()
	CASE "설날"
		SIF DAY != 1
			CALL NEWYEAR
	CASE "발렌타인 데이"
		PRINTL 초콜릿을 받을 수 있을까…
	CASE "화이트 데이 전날"
		PRINTFORMW 답례품의 준비를 해두자
	CASE "바니의날"
		PRINTFORMW \@ DAY:2 == 4? 엄동설한에 바니걸 복장을 입고 보내야 하는 혹독한 날이다… # 모두 바니걸 복장을 입고 있을 것이다\@
	CASE "화이트 데이"
		FOR LOCAL, 1, CHARANUM
			SELECTCASE CFLAG:LOCAL:발렌타인
				CASE 1
					PRINTFORML %조사처리(CALLNAME:LOCAL,"는")% 블랙썬더를 주었군……
					PRINTFORMW 답례는 적당히 해도 될까
				CASE 2
					PRINTFORML %조사처리(CALLNAME:LOCAL,"는")% 보통 초콜릿을 주었군……
					PRINTFORMW 그만한 답례를 하자
				CASE 3
					PRINTFORML %조사처리(CALLNAME:LOCAL,"는")% 비싸 보이는 초콜릿을 주었군……
					PRINTFORMW 제대로 된 답례를 하자
				CASE 4
					PRINTFORML %조사처리(CALLNAME:LOCAL,"는")% 손수 만든 초콜릿을 주었군……
					PRINTFORMW 정성어린 답례를 하자
			ENDSELECT
		NEXT
	CASE "그믐날"
		PRINTL 오늘은 모든 캐릭터가 밤샘합니다
ENDSELECT

@SEASONAL(ARG)
VARSET LOCAL
;만났을 때
SELECTCASE FESTIVAL()
	CASE "설날"
		PRINTL
		PRINTFORML ●%마스터는% %CALLNAME:ARG%에게 새해 인사를 했다
		CALL OTOSHIDAMA(ARG)
	CASE "발렌타인 데이"
		SELECTCASE CFLAG:ARG:호감도
			CASE 1000 TO 3000
				PRINTFORMW %ARG로%부터 초콜릿을 받았다
				PRINTW ……블랙썬더다
				ITEM:120 ++
				CFLAG:ARG:발렌타인 = 1
			CASE 3001 TO 5000
				PRINTFORMW %ARG로%부터 초콜릿을 받았다
				ITEM:121 ++
				CFLAG:ARG:발렌타인 = 2
			CASE IS > 5000
				IF EXP:ARG:애정경험 < 100
					PRINTFORMW %ARG로%부터 비싸보이는 초콜릿을 받았다
					ITEM:122 ++
					CFLAG:ARG:발렌타인 = 3
				ELSE
					PRINTFORML %ARG로%부터 초콜릿을 받았다
					PRINTW 손수 만든 것이다! 기쁘다!
					ITEM:123 ++
					CFLAG:ARG:발렌타인 = 4
				ENDIF
		ENDSELECT
ENDSELECT

@FESTIVAL_MENU()
#FUNCTIONS
SELECTCASE DAY:2
	;춘
	CASE 1
		SELECTCASE DAY:3
			CASE 1,2
				RETURNF "오세치요리"
		ENDSELECT
	;하
	CASE 2
	;추
	CASE 3
	;동
	CASE 4
		SELECTCASE DAY:3
			CASE 3
				RETURNF "에호마키"
			CASE 31
				RETURNF "토시코시 소바"
		ENDSELECT
ENDSELECT

@NEWYEAR
#DIM YOME
#DIM ChildCount
#DIM CardCount
PRINTW 신년 새해복 많이 받으세요!
CardCount = 0
FOR LOCAL,1,CHARANUM
	IF ABL:YOME:친밀 >= 5 && !CFLAG:YOME:신사거주
		PRINTFORML %CALLNAME:YOME%から年賀状が届いた
		CardCount ++
	ENDIF
	SIF CFLAG:YOME:아이인원수 - CFLAG:YOME:육아인원수 <= 0
		CONTINUE
	FOR ChildCount,1,11
		IF DAY - CHILD_BIRTHDAY:YOME:ChildCount < 120
			PRINTFORML %CALLNAME:YOME%의 \@ CHILD_SEX:YOME:ChildCount == 1 ? 娘 # 息子 \@의 %CHILDNAME:YOME:ChildCount%から年賀状が届いた
			CardCount ++
		ENDIF
	NEXT
NEXT
IF CardCount > 0
	PRINTFORMW 합계 {CardCount}장！
ENDIF
PRINTFORML 자, %CALLNAME:MASTER%의 올해의 포부는…
PRINTL
PRINTLC [1] 자위 장려
PRINTLC [2] 동성애 추진
PRINTLC [3] 암컷 구멍 개척
PRINTL
PRINTL
PRINTLC [4] 항문 개발
PRINTLC [5] 유협진흥
PRINTLC [6] 자손 번영
$LOOP
INPUT
SELECTCASE RESULT
	CASE 1
		LOCALS = 자위
	CASE 2
		LOCALS = 동성애
	CASE 3
		LOCALS = 섹스
	CASE 4
		LOCALS = 애널 섹스
	CASE 5
		LOCALS = 파이즈리
	CASE 6
		LOCALS = 아이 만들기
	CASEELSE
		GOTO LOOP
ENDSELECT

PRINTFORML %마스터는% 올해 %LOCALS%에 전념하기로 굳게 마음먹었다
FLAG:포부 = RESULT

;연중 플래그를 리셋
FOR LOCAL, 0, CHARANUM
	CFLAG:LOCAL:연중 = 0
NEXT

@OTOSHIDAMA(ARG)
SIF DAY <= 1
	RETURN
SIF !TALENT:ARG:유치 && !TALENT:ARG:육아중
	RETURN
IF TALENT:ARG:유치
	PRINTFORMW %CALLNAME:ARG%에게 세배돈을 졸라졌다
ELSE
	PRINTFORMW %CALLNAME:ARG%에게 %SHOW_CHILDNAME(ARG)%へのお年玉を要求された
ENDIF

SELECTCASE MONEY
	CASE IS > 500000
		LOCAL = 10000
	CASE IS > 100000
		LOCAL = 5000
	CASE IS > 30000
		LOCAL = 3000
	CASE IS > 3000
		LOCAL = 1000
	CASEELSE
		LOCAL = 0
ENDSELECT
IF !LOCAL
	PRINTFORML %마스터가% 지갑의 내용물을 보여주자, %ARG는% \@ TALENT:ARG:유치 ? 연민이 담긴 눈빛을 향했다 # 노골적으로 실망한 표정을 지었다\@
	PRINTFORMW %CALLNAME:MASTER%의 마음이 상했다
ELSE
	PRINTFORML %CALLNAME:ARG%에게 \\{LOCAL}뺏겼다…
	PRINTFORML %ARG는% 기분이 좋다
	MONEY -= LOCAL
	TALENT:ARG:기분 = 1
	CFLAG:ARG:호감도 += LOCAL/200
	CFLAG:ARG:신뢰도 += LOCAL/200
ENDIF

path: ERB/이벤트관련/기념일이벤트_t.ERB


code:
@내방플래그
#DIM 내방자수

#DIM 내방확률
#DIM 기본내방확률
#DIM 내방인원수상한
#DIM 내방연쇄

;LOCAL = 각 캐릭터의 NO
내방자수 = 0
내방확률 = 0
기본내방확률 = 0
CVARSET CFLAG, GETNUM(CFLAG, "거점내방")
;출금인원수에 응해 기본 내방율의 조정. 실수가 아니고 비율 계산으로 변경하고 있습니다
SELECTCASE FLAG:출금인원수 * 100 / CHARANUM
CASE IS >= 60
	기본내방확률 = 15
CASE IS >= 40
	기본내방확률 = 16
CASE IS >= 20
	기본내방확률 = 17
CASEELSE
	기본내방확률 = 18
ENDSELECT

IF TALENT:MASTER:천객만래
	내방인원수상한 = 12
ELSE
	내방인원수상한 = 10
ENDIF
;우선 확정으로 오는 캐릭터를 카운트
FOR LOCAL,1,CHARANUM
	TRYCCALLFORM K{LOCAL}_BEFORETRAIN
		;PRINTFORML K{LOCAL}があったよ
	CATCH
		;PRINTFORML K{LOCAL}はないよ
	ENDCATCH
	IF CFLAG:LOCAL:연회참가
		IF IN_HOME(FLAG:32)
			CFLAG:LOCAL:거점내방 = 1
			CFLAG:LOCAL:마지막내방일 = DAY
			내방자수 ++
		;다만 신사외라면 확실히 오지 않는(술>>(넘을 수 없는 벽)>>연인의 당신)
		ELSE
			CFLAG:LOCAL:거점내방 = 0
		ENDIF
	;발원함 대상/의뢰실행중 및 달성 사람은 내방한다
	; // 자택 멤버는 발원함 플래그가 ON에서도 내방하지 않게 from 발원함 패치. ERB//
	ELSEIF 내방가능(LOCAL, "강제") && (CFLAG:LOCAL:다음날내방) || CFLAG:LOCAL:의뢰상황 == 3
		CFLAG:LOCAL:거점내방 = 1
		CFLAG:LOCAL:마지막내방일 = DAY
		내방자수 ++
	ELSE
		CFLAG:LOCAL:거점내방 = 0
	ENDIF
NEXT

CALL FISHER_YATES_SHAFFLE(CHARANUM - 1)
FOR LOCAL:10 , 0 , CHARANUM - 1
	LOCAL = SHAFFLE_ARRAY:(LOCAL:10) + 1
	;출금인원수＋각 소질에 의한 수정으로부터 내방율을 산출해, 판정
	;일부 캐릭터에게는 추가로 조건이나 확률 변동이 존재
	SIF FLAG:교류내방모드 == 3
		BREAK
	;내방자수가 내방인원수상한을 넘으면(자) 종료
	SIF 내방자수 >= 내방인원수상한
		BREAK
	;이미 내방 플래그가 서 있으면 판정하지 않음
	SIF CFLAG:LOCAL:거점내방
		CONTINUE
	SIF !내방가능(LOCAL)
		CONTINUE
	내방확률 = 기본내방확률 - TALENT:LOCAL:연모 * 2 - TALENT:LOCAL:사모 - CFLAG:LOCAL:(420 + MAIN_MAP)
	SIF CFLAG:LOCAL:딸딸이들킴
		내방확률 += 5
	;最初の一ヶ月は補正なし
	IF DAY > 30
		;마지막내방일が近いほど来訪しにくい
		IF DAY - CFLAG:LOCAL:마지막내방일 < 4
			내방확률 += 100
		ELSEIF DAY - CFLAG:LOCAL:마지막내방일 < 8
			내방확률 += 10
		ELSEIF DAY - CFLAG:LOCAL:마지막내방일 > 20
			내방확률 = 내방확률 * 3 / 4
		ELSEIF DAY - CFLAG:LOCAL:마지막내방일 > 30
			내방확률 = 내방확률 * 2 / 3
		ENDIF
	ENDIF
	IF !RAND:(내방확률 + CFLAG:LOCAL:내방억제치)
		CFLAG:LOCAL:거점내방 = 1
		CFLAG:LOCAL:마지막내방일 = DAY
		내방자수 ++
	ELSE
		CFLAG:LOCAL:거점내방 = 0
	ENDIF
	;相性の良いキャラが一緒に来訪してくるシステム(最大2人まで)
	내방연쇄 = 0
	IF CFLAG:LOCAL:거점내방
		FOR LOCAL:11, 1, CHARANUM - 1
			LOCAL:1 = SHAFFLE_ARRAY:(LOCAL:11) + 1
			IF RELATION:LOCAL:(LOCAL:1) > 100 && 내방가능(LOCAL:1) && 내방자수 < 내방인원수상한 && 내방연쇄 < 2
				;相性200%で50%,120%で30%
				IF RAND:100 < RELATION:LOCAL:(LOCAL:1) / 4
					CFLAG:(LOCAL:1):거점내방 = 1
					CFLAG:(LOCAL:1):마지막내방일 = DAY
					내방자수 ++
					내방연쇄 ++
				ENDIF
			ENDIF
		NEXT
	ENDIF
NEXT

;외박중은 비표시
IF 내방자수 > 0 && AT_HOME(MASTER)
	PRINTFORMW %GET_MAPNAME(MAIN_MAP)%에 내방자가 있던 것 같다.
	PRINTL
ENDIF

@내방가능(ARG, ARGS)
#FUNCTION
;절대 내방할 수 없는 판정
SIF CFLAG:ARG:신사거주 > 0
	RETURNF 0
SIF CFLAG:ARG:출금 || TALENT:ARG:스침
	RETURNF 0
SIF CFLAG:ARG:내방불능
	RETURNF 0
IF GROUPMATCH(ARG, [[와카사기히메]])
	SIF !GETBIT(FLAG:풀장, 3)
		RETURNF 0
ENDIF

;강제라면 여기서 판정
SIF ARGS == "강제"
	RETURNF 1

;모드에 따른 설정
IF FLAG:교류내방모드 == 1
	SIF CFLAG:ARG:호감도 <= 50
		RETURNF 0
ELSEIF FLAG:교류내방모드 == 2
	SIF !TALENT:ARG:사모 && !IS_LOVER(ARG) && !TALENT:ARG:연인 && !TALENT:ARG:섹프
		RETURNF 0
ELSEIF FLAG:교류내방모드 == 3
	RETURNF 0
ENDIF
;캐릭터별 설정
SELECTCASE ARG
CASE [[모코우]], [[카구야]]
	SIF CFLAG:ARG:호감도 <= 500
		RETURNF 0
;와타츠키、사구메は穢れた地上には行きたがらない
CASE [[요리히메]], [[토요히메]], [[사구메]]
	SIF !TALENT:ARG:연모 && !TALENT:ARG:사모
		RETURNF 0
ENDSELECT

RETURNF 1


code:
@데이트종료타임업처리
VARSET LOCAL
;기력 제로, 쇠약시 데이트 중단
FOR LOCAL, 0, CHARANUM
	SIF CHK_DATENOW(CFLAG:LOCAL:데이트중) && (CFLAG:LOCAL:쇠약 || (BASE:LOCAL:기력 == 0 && CFLAG:LOCAL:현재위치 != 혼숙여관 && CFLAG:LOCAL:현재위치 != 러브호텔 && CFLAG:LOCAL:현재위치 != OMANEKIBEYA()))
		LOCAL:1 ++
	SIF CHK_DATENOW(CFLAG:LOCAL:데이트중) && CFLAG:LOCAL:쇠약
		CALL ENDUFUFU(LOCAL)
NEXT
SIF NEMUKE() >= 720 || BASE:MASTER:술기운 >= MAXBASE:MASTER:술기운
	LOCAL:2 ++
;초대받았으며 연속숙박이 아니라면 묵을 수 있다
IF CFLAG:MASTER:초대 && (LOCAL:1 || LOCAL:2) && !FLAG:70
	IF !TALENT:TARGET:연모
		PRINTFORML 슬슬 %CALLNAME:(CFLAG:MASTER:초대)%도 정신을 차릴 때가 되었겠지、적당한 때인가…
		IF CFLAG:TARGET:우후후
			PRINTFORMW 매달리는 %조사처리(CALLNAME:(CFLAG:MASTER:초대), "를")% 어떻게든 풀어내어、%GET_MAPNAME(MAIN_MAP)%(으)로 되돌아갔다…
		ELSE
			PRINTFORMW 흥건히 열띤 %CALLNAME:(CFLAG:MASTER:초대)%의 시선에서 벗어나듯이、%GET_MAPNAME(MAIN_MAP)%(으)로 되돌아갔다…
		ENDIF
		GOTO ESCAPE
	ELSEIF TFLAG:거품목욕
		CALL 泡風呂終了判定("あなた限界")
		GOTO ESCAPE
	ENDIF
	IF LOCAL:1
		PRINT 기력이 없어지기 시작했으나
	ELSEIF LOCAL:2
		PRINT 이만 돌아갈 시간이나
	ENDIF
	SIF !TCVAR:(CFLAG:MASTER:초대):연속숙박금지
		PRINTFORMW 오늘은 %CALLNAME:(CFLAG:MASTER:초대)%가 묵게 해준다고 합니다
	SIF TCVAR:(CFLAG:MASTER:초대):연속숙박금지
		PRINTFORMW 오늘도 %CALLNAME:(CFLAG:MASTER:초대)%에 묵고 가겠습니까?
	PRINTL [0]묵고 간다 [1]돌아간다
	WHILE 1
	INPUT
	SIF RESULT == 0 || RESULT == 1
		BREAK
	WEND
	IF !RESULT && !TCVAR:(CFLAG:MASTER:초대):연속숙박금지
		PRINTFORMW 호의를 받아들여 묵고 가기로 했다
		;데이트는 종료
		CFLAG:MASTER:데이트중 = MAIN_MAP
		CFLAG:TARGET:데이트중 = MAIN_MAP
		FLAG:데이트상대 = 0
		TFLAG:데이트전호감도 = 0
		TFLAG:데이트도중 = 0
		CALL 모리야복권데이트상대의리셋()
		;이후의 처리를 컷
		GOTO OTOMARI
	ELSEIF !RESULT && TCVAR:(CFLAG:MASTER:초대):연속숙박금지
		PRINTFORMW 아무리 그래도 연달은 숙박은 거절당했다…
	ENDIF
ENDIF
$ESCAPE
IF LOCAL:1 && CFLAG:MASTER:현재위치
	PRINTL
	PRINTW ***** 기력이 0이 되었기에, 귀로에 접어들었습니다 *****
	CALL 귀가의시간경과
	FLAG:시간정지 = 0
	RESETBGCOLOR
ELSEIF LOCAL:2
	PRINTL
	IF BASE:MASTER:술기운 >= MAXBASE:MASTER:술기운
		PRINTW ***** 만취했기 때문에, 갈짓자 걸음으로 귀로에 접어들었습니다 *****
		PRINTFORMW %GET_MAPNAME(MAIN_MAP)%에 돌아온 %마스터는% 그 자리에서 쓰러졌습니다
	ELSE
		PRINTW ***** 졸음이 엄습했기에, 귀로에 접어들었습니다 *****
	ENDIF
	CALL 귀가의시간경과
ENDIF
IF LOCAL:1 || LOCAL:2
	FOR LOCAL, 0, CHARANUM
		IF LOCAL && CFLAG:LOCAL:쇠약
			IF !CFLAG:LOCAL:신사거주
				CFLAG:LOCAL:초기위치 = SUKIMA()
				CFLAG:LOCAL:현재위치 = SUKIMA()
				CFLAG:LOCAL:거점내방 = 0
				CALL KOJO_MESSAGE_SEND("EVENT", 20, LOCAL, 2,0)
				PRINTFORMW 녹초가 된 %CALLNAME:LOCAL%는 돌아갔습니다
			ELSE
				PRINTFORM %CALLNAME:LOCAL%는 %NAME_FROM_PLACE(CFLAG:LOCAL:초기위치)%
				PRINTW 에 돌아왔습니다
				SIF CFLAG:MASTER:초기위치 != CFLAG:LOCAL:초기위치
					LOCK:(CFLAG:LOCAL:초기위치) = 1
				CFLAG:LOCAL:현재위치 = CFLAG:LOCAL:초기위치
				CALL KOJO_MESSAGE_SEND("EVENT", 3,1, LOCAL, 0)
			ENDIF
		ENDIF
		CALL SET_DATE(99, LOCAL)
		IF CFLAG:LOCAL:초대
			CFLAG:LOCAL:초대 = 0
			IF LOCAL == 0
			ELSEIF CFLAG:LOCAL:신사거주
				CFLAG:LOCAL:초기위치 = CFLAG:LOCAL:신사거주
			ELSE
				CFLAG:LOCAL:초기위치 = SUKIMA()
			ENDIF
		ENDIF
		SIF CFLAG:LOCAL:동침중 > 0
			CFLAG:LOCAL:동침중 = 0
		SIF CFLAG:LOCAL:우후후
			CALL ENDUFUFU(LOCAL)
		;장난 OFF
		CALL SET_PRANK(0,LOCAL,1)

	NEXT
	CALL 모리야복권데이트상대의리셋()
ENDIF
IF TFLAG:데이트도중 && !FLAG:데이트상대
	TIME += TFLAG:데이트도중
	TFLAG:데이트도중 = 0
	CFLAG:MASTER:현재위치 = GET_ENTRANCE(CFLAG:MASTER:데이트중)
ENDIF
$OTOMARI

path: ERB/이벤트관련/데이트종료타입업처리_t.ERB


code:
@DAILY_CONDITION_EV23
SIF GETBIT(FLAG_DAILY_EV23, 1)
	RETURN -1
SIF ITEM:미라다의돌 > 0
	RETURN -1
SIF RAND:15 > 0
	RETURN -1
SIF MONEY < DAILY_EV23_미라다_바닥시세 && !GETBIT(FLAG_DAILY_EV23, 4)
	RETURN -1
CALL DAILY_EV23
RETURN 1

@DAILY_EV23
#DIMS リソース名
IF !GETBIT(FLAG_DAILY_EV23, 0)
	SETBIT FLAG_DAILY_EV23, 0
	PRINTFORML 하루를 마치고 이제 잘까하는 무렵에 현관 쪽에서 목소리가 들렸다
	PRINTFORMW 「실례하겠습니다～」
	PRINTFORML 이런 시간에 누구일까?
	PRINTFORMW %마스터는% 졸린 눈을 비비면서 응대에 나섰다
	PRINTFORML 그곳에 있던 것은…
	CALL ASK_YN("머리를 뒤로 빗어 넘긴 토끼 방문 판매원","보이쉬한 단발머리 토끼 방문 판매원")
	FLAG_DAILY_EV23_판매원 = RESULT + 1
	CALL PRINT_IMAGE_EV23("服_通常")
	PRINTFORML 「처음 뵙겠습니다～」
	PRINTFORML 「훌륭한 상품을 소개하러 왔습니다～」
	PRINTFORML 그곳에 있던 사람은 팔짝팔짝 움직이는 토끼귀가 사랑스러운 방문 판매원이었다
	CALL ASK_M("되쫓아 버린다", 1, "설명을 들어본다", 1)
	IF !RESULT
		SETBIT FLAG_DAILY_EV23, 1
		PRINTFORML 방문 판매는 사절이다
		PRINTFORML 「그런ー, 적어도 이야기만이라도ー」
		PRINTFORMW 더욱더 물고 늘어지는 판매원을 무시하고 문단속을 확인하고 잤다
		RETURN
	ENDIF
	SETBIT FLAG_DAILY_EV23, 2
	PRINTFORMW 흥미가 끌린 %마스터는% 상품 설명을 들어보기로 했다
	CALL DAILY_EV23_SALESTALK(0)
	FONTBOLD
	PRINTFORMW 「이럴 수가!」
	PRINTFORM 「이번에 가격
	CALL COLORMESSAGE(@"\\{DAILY_EV23_미라다_저가}", C_RED, 0, 1)
	PRINTFORMW 에 제공이 실현했습니다!」
	FONTREGULAR
	PRINTFORM 「상식을 벗어난 환상향만의
	CALL COLORMESSAGE("특별 가격", C_RED, 0, 1)
	PRINTFORML 입니다!!」
	PRINTFORM 「이런
	CALL COLORMESSAGE("파격적인 가격", C_RED, 0, 1)
	PRINTFORMW 상식을 벗어난 환상향만의 특별가격입니다!!」
	FONTBOLD
	PRINTFORML 「자, 고객님!」
	PRINTFORM 「충실한 성생활을 약속하는
	CALL COLORMESSAGE("애욕을 독점하는 미라다의 돌 밀레니엄", C_D_PINK, 0, 1)
	PRINTFORMW !!!!」
	PRINTFORML 「하나, 어떠세요?!」
	FONTREGULAR
	PRINTFORML 　현재의 소지금:%금액표시(MONEY)%
	CALL ASK_M("흥미없다!", 1, "비싸!", 1, "샀다!", MONEY >= DAILY_EV23_미라다_저가)
	SELECTCASE RESULT
	CASE 0
		$IRANAI
		PRINTFORML 「……그런가요」
		PRINTFORMW 「시간을 내주셔서 감사합니다!」
		PRINTFORML 「여기, 저희 회사 상품의 샘플입니다」
		PRINTFORMW 「괜찮으시면 사용해 주세요」
		;우콘, 로션, 콘돔 세트
		CALL GET_ITEM("우콘드링크", 1, 1)
		CALL GET_ITEM("로션", 1, 1)
		CALL GET_ITEM("콘돔", 1, 1)
		PRINTFORMW 「그럼 실례하겠습니다」
		PRINTFORMW 그 판매원은 마지막으로 정중히 절을 하고 떠났다
		RETURN
	CASE 1
		PRINTFORMW 「비, 비싸!?」
		PRINTFORML 「뭐ー, 그렇죠」
		PRINTFORM 「구매자들도
		CALL COLORMESSAGE("효과는 훌륭하지만 가격이 비싸다", C_YELLOW, 0, 1)
		PRINTFORMW 는 의견이 다수 있었습니다」
		PRINTFORM 「저희 회사 쪽에서도 이
		CALL COLORMESSAGE("훌륭한 상품", C_YELLOW, 0, 1)
		PRINTFORML 을 한 분이라도 더 보내드리고 싶다고 분발해서」
		PRINTFORM 「이번에,
		CALL COLORMESSAGE("저렴하게 제공", C_RED, 0, 1)
		PRINTFORML 된 기획이 완성되었습니다만. 으음～」
		PRINTFORMW 「내 재량으로 너무 제멋대로 해버리면 혼날까……」
		PRINTFORMW 「…………」
		PRINTFORMW 「……알았습니다」
		FONTBOLD
		PRINTFORM 「
		CALL COLORMESSAGE(@"통상 가격\\{DAILY_EV23_미라다_정가}", C_RED, 0, 1)
		PRINTFORM 의
		CALL COLORMESSAGE("애욕을 독점하는 미라다의 돌 밀레니엄", C_D_PINK, 0, 1)
		PRINTFORML !!!」
		PRINTFORM 「여기에,
		CALL COLORMESSAGE("로션 3개 세트", C_YELLOW, 0, 1)
		PRINTFORMW 를 붙여!」
		PRINTFORMW 「게다가!」
		PRINTFORM 「이쪽의
		CALL COLORMESSAGE("콘돔 10개 세트", C_YELLOW, 0, 1)
		PRINTFORML 도 많이 주어야겠어요!」
		PRINTFORM 「이
		CALL COLORMESSAGE("미라다 안심 세트", C_D_PINK, 0, 1)
		PRINTFORM 있으면, 만약 내일
		CALL COLORMESSAGE("섹프", C_RED, 0, 1)
		PRINTFORMW 가 생겨도 괜찮아!」
		PRINTFORM 「가격은……
		CALL COLORMESSAGE(@"\\{DAILY_EV23_미라다_바닥시세}", C_RED, 0, 1)
		PRINTFORML 에 제공이 됩니다! 이제 더 이상 싸지지 않을 거예요!」
		PRINTFORM 「자, 빨리
		CALL COLORMESSAGE("애욕을 독점하는 미라다의 돌 밀레니엄", C_D_PINK, 0, 1)
		PRINTFORM 을 손에 넣어,
		CALL COLORMESSAGE("팡팡", C_RED, 0, 1)
		CALL HPH_PRINT(@"HPH")
		CALL COLORMESSAGE("앙앙", C_RED, 0, 1)
		CALL HPH_PRINT(@"HPH")
		PRINTFORMW  가버립시다!!」
		FONTREGULAR
		PRINTFORML 　현재의 소지금:%금액표시(MONEY)%
		CALL ASK_M("필요없어!", 1, "샀다!", 1)
		SIF !RESULT
			GOTO IRANAI
		MONEY -= DAILY_EV23_미라다_바닥시세
		;러브 로션 3개 세트, 콘돔 10개 세트
		CALL GET_ITEM("로션", 3, 1)
		CALL GET_ITEM("콘돔", 10, 1)
	CASE 2
		MONEY -= DAILY_EV23_미라다_저가
	ENDSELECT
	SETBIT FLAG_DAILY_EV23, 3
	;여기서 미라다 입수
	CALL GET_ITEM("미라다의돌", 1, 1)
	FONTBOLD
	PRINTFORMW 「구매해 주셔서 감사합니다!」
	FONTREGULAR
	CALL PRINT_IMAGE_EV23("服_笑顔")
	CALL ASK_M("졸리고 이제 잘까?", 1, "시험삼아 써볼까?", 1)
	IF !RESULT
		$START_SLEEP
		PRINTFORML 「그럼 실례하겠습니다. 또 사주신다면 기쁘겠습니다」
		PRINTFORMW 당신은 판매원을 배웅하자 기대에 부풀며 잠이 들었다
		RETURN
	ENDIF
	PRINTFORMW ……근데 이거 어떻게 써?
	PRINTFORMW 「이런 깜빡하고 있었네요.실례가 많았습니다」
	PRINTFORML 「사용할 상대방의 얼굴을 떠올리며, 손에 쥐고 힘을 주면……」
	PRINTFORMW 「희미하게 빛을 내기 시작하는 겁니다!」
	PRINTFORML 눈앞에 있는 판매원의 얼굴을 떠올리며 손에 쥐고 힘을 주자……돌은 희미하게 빛나기 시작했다
	PRINTFORMW 과연 이것으로 되는가?
	$USE_NOW
	ITEM:미라다의돌 --
	SETBIT FLAG_DAILY_EV23, 4
	PRINTFORML 「오! 빠른 이용 감사합니다!」
	PRINTFORMW 「목적의 여성은 어떤 분입니까? 역시 대물 노린 건가요? 하쿠레이 무녀라든지! 스키마 요괴라든지!」
	PRINTFORMW 「……에? 나! ……아, 안됩니다～ 그런 일을 하시면 곤란합니다～」
	CALL PRINT_IMAGE_EV23("服_赤面")
	PRINTFORMW 「앗! 안된다고요! 젖가슴 만지면 안돼!」
	PRINTFORML 「고객님?! 치마에 얼굴을 들이대지 마세요!?」
	CALL HPH_PRINT("「싫어……더러운데 혀를 집어넣다니……아앙HPH……마시는 것도 안돼HPH」", "W")
	PRINTFORML 　·
	PRINTFORML 　·
	PRINTFORML 　·
	IF GETBIT(TALENT:MASTER:성별, 1) && !TALENT:MASTER:1
		PRINTFORMW 동정상실(%CALLNAME:MASTER%)
		TALENT:MASTER:1 = 1
	ENDIF
	CALL SET_FIRSTKISS_MASTER("심야 자기 방에서 미라다의 돌 방문 판매원을 돌로 발정 시킨 혼탁함에 바쳤다")
	PRINTFORMW
	PRINTFORML 「고객님도 참, 사실 이런 거 안 되니까요～?」
	PRINTFORMW 「저 화났거든요? 듣고 있어요?」
	CALL PRINT_IMAGE_EV23("裸_LOVE")
	PRINTFORMW 「정말이지! ……다른 고객님 댁도 돌아야 하는데 떠나고 싶지 않아지잖아요」
	PRINTFORMW 「……굉장히, 기분 좋았으니까요. 그, 용서해주지 않아도… …역시 안돼요」
	PRINTFORML 「한번 더 꼬셔주세요. 그러면 용서해 줄테니까요」
	IF GETBIT(TALENT:MASTER:성별, 1)
		PRINTFORML 「또 샤워도 빌려주세요. 정액 투성이고 냄새가 지독해요」
	ELSE
		PRINTFORML 「또 샤워도 빌려주세요. 애액 투성이고 냄새가 지독해요」
	ENDIF
	WAIT
	CALL PRINT_IMAGE_EV23("服_赤面")
	CALL HPH_PRINT("「그럼 실례하겠습니다. 또 사주신다면 기쁘겠습니다HPH」", "W")
	PRINTFORMW 방문 판매원 토낭과 섹프가 되었습니다
	RETURN
ELSE
	PRINTFORML 하루를 마치고 이제 잘까하는 무렵에 현관 쪽에서 목소리가 들렸다
	PRINTFORMW 「실례하겠습니다～」
	PRINTFORML 이런 시간에 누구일까?
	PRINTFORMW 당신은 졸린 눈을 비비며 응대에 나섰다
	IF GETBIT(FLAG_DAILY_EV23, 4)
		CALL PRINT_IMAGE_EV23("服_LOVE")
		CALL HPH_PRINT("「에헤헤 또 와버렸어요HPH」", "L")
	ELSE
		CALL PRINT_IMAGE_EV23("服_笑顔")
		PRINTFORM 「매번 친숙한
		CALL COLORMESSAGE("애욕을 독점하는 미라다의 돌 밀레니엄", C_D_PINK, 0, 1)
		PRINTFORML  방문판매에 왔습니다～」
	ENDIF
	PRINTFORMW 그곳에 있던 사람은 팔짝팔짝 움직이는 토끼귀가 사랑스러운 방문판매원이었다
	PRINTFORML 「그 후 상태는 어떠세요～?」

	IF !GETBIT(FLAG_DAILY_EV23, 3)
		PRINTFORM 「이번에야 말로
		CALL COLORMESSAGE("애욕을 독점하는 미라다의 돌 밀레니엄", C_D_PINK, 0, 1)
		PRINTFORML 의 훌륭함을 이해해주셨으면 해서, 실례했습니다!」
	ELSE
		PRINTFORM 「
		CALL COLORMESSAGE("애욕을 독점하는 미라다의 돌 밀레니엄", C_D_PINK, 0, 1)
		PRINTFORML 의 위력은 실감하실 수 있으실까요?」
		SIF GETBIT(FLAG_DAILY_EV23, 4)
			CALL HPH_PRINT("「……제가 더 실감하고 있을지도 모르겠네요～ 에헤헤 HPH」", "L")
	ENDIF
	WAIT

	$MIRADA_2ND
	PRINTFORML 　현재의 소지금:%금액표시(MONEY)%
	CALL ASK_M("더 이상 오지 않아도 된다", 1, "필요 없다", 1, "사고 싶다", MONEY >= DAILY_EV23_미라다_바닥시세, "밀어넘어뜨린다!", GETBIT(FLAG_DAILY_EV23, 4))
	SELECTCASE RESULT
	CASE 0
		IF GETBIT(FLAG_DAILY_EV23, 4)
			PRINTFORML 「……네?」
			PRINTFORML 「네!? 그건 좀 심하지 않아요!? 버려진 건가요! ?」
			CALL ASK_M("더 이상 오지 말아라", 1, "농담이야", 1)
			SIF !RESULT
				GOTO MOUKONAI
			PRINTFORML 「뭐야 놀래키지 마세요～」
			PRINTFORML 「사실은 어떻게 하고 싶은 거예요?」
			GOTO MIRADA_2ND
		ELSE
			$MOUKONAI
			SETBIT FLAG_DAILY_EV23, 1
			PRINTFORML 「하아, 그렇습니까?」
			PRINTFORMW 방문 판매원은 풀이 죽은 얼굴을 하고 떠났다
		ENDIF
	CASE 1
		GOTO IRANAI
	CASE 2
		CALL GET_ITEM("미라다의돌", 1, 1)
		CALL PRINT_IMAGE_EV23("服_笑顔")
		FONTBOLD
		PRINTFORMW 「구매해 주셔서 감사합니다!」
		FONTREGULAR
		IF MONEY < DAILY_EV23_미라다_저가
			PRINTFORML 「……라니, 어라～？」
			PRINTFORM 「저～, 고객님? 이 상품 가격이
			CALL COLORMESSAGE(@"\\{DAILY_EV23_미라다_저가}", C_RED, 0, 1)
			PRINTFORMW 이 되고 있지만……」
			PRINTFORML 「음, 어쩔 수 없네요」
			PRINTFORM 「이번에는
			CALL COLORMESSAGE(@"\\{DAILY_EV23_미라다_바닥시세}", C_RED, 0, 1)
			PRINTFORML 에 팔아 드릴까요」
			PRINTFORMW 「저와 고객님 사이니까, 특별할텐데요?」
			MONEY -= DAILY_EV23_미라다_바닥시세
		ELSE
			MONEY -= DAILY_EV23_미라다_저가
		ENDIF
		IF GETBIT(FLAG_DAILY_EV23, 4)
			CALL PRINT_IMAGE_EV23("服_赤面")
			PRINTFORMW 「고객님은 쇼핑만으로 만족하나요～？」
			CALL ASK_M("하지 않는다", 1, "한다", 1)
			IF !RESULT
				PRINTFORML 「어라 피곤 하셨어요?」
				PRINTFORMW 「그럼 오늘은 이만 실례할게요」
				PRINTFORMW 판매원 토낭은 아쉬운 듯 떠났다
				RETURN
			ELSE
				CALL HPH_PRINT(@"「후훗 그렇죠～HPH」", "L")
				PRINTFORMW 「……상냥하게 부탁드릴게요?」
				GOTO DO_H
			ENDIF
		ENDIF
		CALL ASK_M("졸리고 이제 잘까?", 1, "……이 돌을 판매원씨에게 써보자", 1)
		SIF !RESULT
			GOTO START_SLEEP
		PRINTFORML 눈앞에 있는 판매원의 얼굴을 떠올리며 손에 쥐고 힘을 주자……돌은 희미하게 빛나기 시작했다
		GOTO USE_NOW
	CASE 3
		CALL PRINT_IMAGE_EV23("服_LOVE")
		PRINTDATAL
			DATA 「시ー러 범해진다～」
			;やーん、おかされる～
			DATA 「정말ー 고객님은 바로 그런일을 해요～」
			DATA 「우와!? 너무 달라붙는군요!」
			DATA 「일 때문에 온 거니까 상품을 사 주세요～」
			DATA 「네, 네. 엣찌는 이불 깔때까지 자제해요～」
		ENDDATA
		CALL ASK_M("입맞춤으로 입을 맞추게 한다", 1, "팁을 쥐어준다(\\3000)", MONEY >= 3000, "팁을 쥐어준다(10CM)", MONEY:2 >= 10)
		SELECTCASE RESULT
		CASE 0
			PRINTFORMW 「응ー！？」
			PRINTDATAW
				DATALIST
					DATA 「…………」
					DATA 「앗! 안돼요……아직 전혀 부족하니까요. 이봐요, 좀 더 쭈욱 해요」
				ENDLIST
				DATA 「또 그렇게 속이고……아, 갑자기 젖꼭지 만지기는 없다니까요. 약한거 알면서」
				DATA 「……능숙하시군요. 왠지 화가 나네요. 그 여유」
				;手慣れてるなぁ。なーんか腹が立ちますね。その余裕
				DATA 「……고객님, 토끼의 성욕을 우숩게 보고 있네요. 어떻게 되든 모르니까요!」
			ENDDATA
		CASE 1, 2
			SIF RESULT == 1
				MONEY -= 3000
			SIF RESULT == 2
				MONEY:2 -= 10
			SELECTCASE RAND:3
			CASE 0
				PRINTFORMW 「……후와아, 돈입니까. 뭐ー 뭐 받을건 받아두겠습니다만」
			CASE 1
				PRINTFORML 「우햐ー. 돈을 노리고 안겨있는게 아닌데요?」
				PRINTFORMW 「고객님의 테크닉이 너무 굉장한 게 나쁘니까요!」
			CASE 2
				PRINTFORML 「으응～. 직접 돈을 받는 것보다는 그」
				PRINTFORMW 「다음에 같이 점심 먹으면서……」
				PRINTFORMW 「……"소리가 작아서 안 들렸다"? ……오ー케이, 알겠습니다.얼른 합시다」
			ENDSELECT
		ENDSELECT
		$DO_H
		PRINTFORML 　·
		PRINTFORML 　·
		PRINTFORML 　·
		PRINTFORMW
		[SKIPSTART]
		;생각 중
		SELECTCASE RAND:3
		CASE 0
			SIF !GETBIT(TALENT:MASTER:성별, 0)
				GOTO ERO_SCENE_BASIC
			SELECTCASE RAND:4
			CASE 0
				PRINTFORML ……후우
				CALL HPH_PRINT("「……후우 HPH」", "W")
				PRINTFORML 「아ー, 위응으로 가버립니다」
				PRINTFORMW 「다 냈으면 흐늘흐늘이군요?」
				PRINTFORMW 「후후. 에헤헤. 이 감촉도 독특하네요～. 말랑말랑 하고 부드럽고 재미있습니다」
				PRINTFORMW 인후 여기서 놀지 말아라
			CASE 1
				PRINTFORMW 「……꺄아악!」
				PRINTFORML 「어째서 고간으로부터 얼굴의 (분)편까지 정액 날아 옵니까～. 건강이 너무 좋습니다!이 에로 마신!」
				CALL HPH_PRINT("「\"굉장히 기분 좋았다\"가 아니어요, 바보 HPH」", "W")
				PRINTFORML 「이나다, 머리카락에 걸리고 있고～. 냄새 잡히지 않아요? 압니까?」
				PRINTFORMW 「아아, 이제(벌써)!이상한 일 말하지 말아 주세요!조금은 반성해라ー！！」
			CASE 2
				PRINTFORML 「이봐이봐」
				PRINTFORMW 「몇 번 말해야 알고 있습니까? 싱싱한은 안 되는 것입니까는……」
				PRINTFORML 「이봐, 고무 붙여드리겠습니다로부터」
				CALL HPH_PRINT("「……네, 좋아요. 와 주세요 HPH」", "W")
			CASE 3
				PRINTFORML 「언제나 생각합니다만, 고객님은 대단한 양을 내는군요」
				PRINTFORMW 「무엇인가 비결이라든지 있습니까?」
				PRINTFORMW 「나날의 청소와 팬티 모양? ……무엇입니까 그것? 의미 몰라요」
			ENDSELECT
		CASE 1
			SIF !GETBIT(TALENT:MASTER:성별, 1)
				GOTO ERO_SCENE_BASIC
			SELECTCASE GET_TALENTNAME(GETNUM(TALENT, "가슴사이즈"), TALENT:MASTER:가슴사이즈)
			CASE "거유", "폭유"
				PRINTFORML ……포근포근
				PRINTFORMW 「으응～, 고객님은 젖가슴 크지요」
				PRINTFORML ……포근포근
				CALL HPH_PRINT("「좋겠다, 계속 손대고 있었던 구 되어버리는 HPH」", "W")
			CASEELSE
				GOTO ERO_SCENE_BASIC
			ENDSELECT
		CASE 2
			GOTO ERO_SCENE_BASIC
			;어떻게든(어쩐지) 만들었지만 판정을 별로(호빵) 채우기라고 없다
			;판매원씨는 몸을 당긴다
			PRINTFORML 판매원과 같이 가 당신의 방까지 이동하려고 했지만, 거기에는 먼저 온 고객님이 있었다
			PRINTFORML
			PRINTFORML 「조금ー, 저것 누구입니까인가～？」
			PRINTFORML
			PRINTFORML 「……네? 자녀분?」
			PRINTFORML 「……네? 사모님?」
			PRINTFORML 「……네? 연인?」
			PRINTFORML
			PRINTFORML 「우와, 최저!바람둥이!여자의 적!」
			PRINTFORML 「나와 놀 때가 아니지 않습니까!」
			PRINTFORML
			PRINTFORML 「자자, 상관해 주지 않으면 외로워서 울음있어요?」
			PRINTFORML 「나에 대한 것은 좋으니까. 분명하게 귀여워해 주세요」
			PRINTFORML
			PRINTFORML 「나는 들키지 않는 동안에 실례합니다」
			PRINTFORML 「다음은 천천히」
			WAIT
			RETURN
		CASEELSE
			;특히 조건이 없는 패턴은 여기
			$ERO_SCENE_BASIC
			SELECTCASE RAND:5
			CASE 0
				PRINTFORML 「……그러니까요」
				PRINTFORML 「오늘은 급한 용무가 들어와 버려서. 곧바로 돌아가지 않으면 안된다  입니다」
				PRINTFORML 「이므로, 한번만 밖에 할 수 없습니다. 미안해요」
				PRINTFORML 「예, 이 장소에서. 선채로 합시다. 시간 없으니까」
				PRINTFORML 「1회 하면(자) 끝이에요?」
				PRINTFORML 「다음에 왔을 때는 충분히 상대 해 드릴테니까. 에에에, 빨리 해 주세요」
			CASE 1
				;미실장 분기
				;진한 에로 역시 쓸 수 없었다
				PRINTFORML 「오늘은이군요. 여기가 마지막 영업처입니다」
				PRINTFORML 「시간 만들려고 생각해 노력해 버렸습니다. 칭찬해 주세요. 에헤헤」
				PRINTFORML 「……많이 해요?」
				WAIT
				PRINTFORML 　·
				PRINTFORML 　·
				PRINTFORML 　·
				PRINTFORMW
				RETURN
			CASE 2
				PRINTFORML ……쪽
				PRINTFORMW 「고객님은 즐 있고입니다. 이 타이밍에 쪽을 하는 것은 반칙입니다」
				PRINTFORML ……쪽
				PRINTFORMW 「에헤헤. 답례♪」
				PRINTFORML ……쪽
				PRINTFORMW 「싫어!답례의 답례는 안 돼요」
				PRINTFORMW ……쭙, 쭙, 쭙, 쭙
			CASE 3
				PRINTFORML 「……아……아……싫어」
				PRINTFORMW 「유두, 약하니까……그런, 계속 스윽스윽 되면(자) ……」
				PRINTFORML 「으응 ……이상하게, 로부터……아」
				PRINTFORMW 「도, 부탁이니까……!……용서해!……괴롭히지 말고 창고 차이야 」
			CASE 4
				PRINTFORML 「!기다려!조금 기다려!」
				PRINTFORMW 「스톱!스～톱!」
				PRINTFORML 「어디 손대어 입니까!거기는 엉덩이의 구멍이에요!」
				PRINTFORMW 「그것은 안 돼요. 그런것은 안 돼요. 그런 일의 허가는 낼 수 없습니다」
				PRINTFORML 「보통으로 합시다. 보통이 좋아요」
				PRINTFORMW 「정말이지. 방심도 틈도 없네요, 고객님도 참」
			ENDSELECT
		ENDSELECT
		WAIT
		PRINTFORML 　·
		PRINTFORML 　·
		PRINTFORML 　·
		PRINTFORMW
		[SKIPEND]
		CALL PRINT_IMAGE_EV23("裸_LOVE")
		PRINTFORML 「정신없는게 좀 아쉽네요」
		CALL HPH_PRINT(@"「고마웠습니다. 아주 좋았어요HPH」", "W")
		PRINTFORML 「언제가 될지는 모르겠지만 다시 올테니까요. 기다려 주세요」
		PRINTFORMW 「그럼 실례하겠습니다」
		PRINTFORMW 방문 판매원은 서운해하면서도 떠났다
	ENDSELECT
ENDIF

@PRINT_IMAGE_EV23(ARGS)
#DIMS リソース名
リソース名 = 立ち絵_%ARGS%_MOB_イナバ_{FLAG_DAILY_EV23_판매원}
CALL 画像表示(リソース名, 0, 0, 0, 100)
CALL ピクセル自動改行(デフォルトキャラ画像横幅)
CALL 画像描画終了

;다른 함수에서 부르기 위해 분리.내용은 바꾸지 않았습니다.
@DAILY_EV23_SALESTALK(ARG)
PRINTFORM 「갑작스럽지만 고객님,
CALL COLORMESSAGE("엣찌", C_H_PINK, 0, 1)
PRINTFORML 는 좋아하시나요?」
PRINTFORMW 「좋아하시는군요～ 이 상품, 틀림없이 만족하실 수 있습니다」
PRINTFORMW 「이번에 소개드리는 것은 이 상품!」
PRINTFORM 「짜잔!
CALL COLORMESSAGE("애욕을 독점하는 미라다의 돌 밀레니엄", C_D_PINK, 0, 1)
PRINTFORML 입니다!」
PRINTFORMDW 그렇게 말하면서 판매원이 \@ !ARG ? 꺼낸 # 손에 든 \@ 것은 의심스러운 반짝이는 결정과도 같았다
PRINTFORML 「환상향에는 매력적인 여성분들이 정말 많이 계시네요」
PRINTFORML 「절벽위의 꽃무녀님에, 아름답고 경외한 여신님」
PRINTFORMW 「건방요염한 대요괴에서부터, 순진무구한 요정님에 이르기까지, 더 힘차게 나아갑니다」
PRINTFORMW 「……편안하고 친근한 요괴토끼 같은 것도 있지만요～ 에헤헤」
PRINTFORM 「고객님도 신경이 쓰이는 그 여자아이와
CALL COLORMESSAGE("그런～일", C_H_PINK, 0, 1)
PRINTFORM 이나
CALL COLORMESSAGE("이런～일", C_H_PINK, 0, 1)
PRINTFORMW , 해보고 싶다고 생각한적이 있으시죠?」
FONTBOLD
PRINTFORM 「
CALL COLORMESSAGE("연모", C_H_PINK, 0, 1)
PRINTFORML 에 빠지게 하는 것도 좋지만, 일\@ GETBIT(TALENT:MASTER:성별, 1) ? 부 # 처\@다처는 힘들어요」
PRINTFORM 「당신을 원해! 독차지하고 싶다! 그렇게 바라는 것은
CALL COLORMESSAGE("사랑하는 소녀", C_H_PINK, 0, 1)
PRINTFORMW 의 숙명인 것입니다!」
FONTREGULAR
PRINTFORM 「그 소녀도 연모, 이 소녀도 연모… …삼각, 사각, 사랑의 다각형이 닿는 끝은 질퍽질퍽한
CALL COLORMESSAGE("정욕의 아수라장", C_D_RED, 0, 1)
PRINTFORMW 입니다」
PRINTFORM 「
CALL COLORMESSAGE("초록 눈동자 그 사람", C_GREEN, 0, 1)
PRINTFORML 이 아주 좋아하는 음식이네요～ 무섭다 무서워」
PRINTFORM 「
CALL COLORMESSAGE("질투로 얼룩진 서스펜스 극장", C_D_RED, 0, 1)
PRINTFORM 은
CALL COLORMESSAGE("꿈의 환상 하렘", C_H_PINK, 0, 1)
PRINTFORMW 과는 거리가 멀어요」
FONTBOLD
PRINTFORML 「좀더 부담없이 섹스하고 싶어!」
PRINTFORM 「이
CALL COLORMESSAGE("애욕을 독점하는 미라다의 돌 밀레니엄", C_D_PINK, 0, 1)
PRINTFORMW 이 그 꿈을 확실하게 이뤄줍니다!!」
FONTREGULAR
PRINTFORM 「
CALL COLORMESSAGE("연모", C_H_PINK, 0, 1)
PRINTFORM 는 귀찮아, 하지만 노리고
CALL COLORMESSAGE("섹프", C_RED, 0, 1)
PRINTFORML 를 만드는 것은 어려운 일이죠～」
PRINTFORML 「애초에 미움을 받고 있으면 화제에 오르지도 않습니다」
;箸にも棒にもかからないとは、例えば何か、企画書や、提案、アイデアなどがあったときに、それらが、候補にすら上がらないというときに使われます。つまり、話題にもあがらない全く持って問題外であるという意味になります
;대충 화제에도 올라오지(걸리지) 않는 다는 뜻
PRINTFORM 「그렇다고, 잘 대해보면 이번엔 상대방이
CALL COLORMESSAGE("사랑", C_H_PINK, 0, 1)
PRINTFORMW 에 빠져버린다」
PRINTFORML 「압니다, 알아요, 그 고민!」
FONTBOLD
PRINTFORM 「뚜렷한 신체만의 관계를 원하는데, 아무래도
CALL COLORMESSAGE("신뢰도", C_YELLOW, 0, 1)
PRINTFORML 가 너무 높아버린다!」
PRINTFORM 「이
CALL COLORMESSAGE("애욕을 독점하는 미라다의 돌 밀레니엄", C_D_PINK, 0, 1)
PRINTFORMW 이 그 고민을 확실하게 해결 해버립니다!!」
PRINTFORM 「그런데, 이쪽의
CALL COLORMESSAGE("애욕을 독점하는 미라다의 돌 밀레니엄", C_D_PINK, 0, 1)
PRINTFORML !」
FONTREGULAR
PRINTFORM 「한번 사용하면 성욕중추에 압력을 가해 여자아이 뇌속에 고객님의
CALL COLORMESSAGE("성적인 매력", C_RED, 0, 1)
PRINTFORMW 이 박힙니다!」
PRINTFORM 「모습을 보면
CALL COLORMESSAGE("두근!", C_RED, 0, 1)
PRINTFORM 시선이 마주치면
CALL COLORMESSAGE("두근!", C_RED, 0, 1)
PRINTFORM 손이 닿으면
CALL COLORMESSAGE("두근!", C_RED, 0, 1)
PRINTFORM 더이상
CALL COLORMESSAGE("두근 두근", C_RED, 0, 1)
PRINTFORML 이 멈추지 않는다!」
FONTBOLD
PRINTFORM 「한층 더
CALL COLORMESSAGE("신뢰도", C_YELLOW, 0, 1)
PRINTFORM 의 상승을 억제하고, 단단한 그 소녀도
CALL COLORMESSAGE("섹프", C_RED, 0, 1)
PRINTFORMW 로 일직선!」
PRINTFORMW 「이런 멋진 아이템, 이제 구할 수 밖에 없죠!?」
FONTREGULAR
PRINTFORML 「그렇게 되면 신경쓰이는 것은 가격 쪽입니다」
PRINTFORM 「
CALL COLORMESSAGE("좋은 것", C_YELLOW, 0, 1)
PRINTFORM 은
CALL COLORMESSAGE("좋은 값", C_YELLOW, 0, 1)
PRINTFORML 이라는 것이, 세상의 상식, 사람의 상식, 환상향의 상식」
PRINTFORM 「
CALL COLORMESSAGE("가격 이상", C_M_GREEN, 0, 1)
PRINTFORMW 이란 맛있는 이야기는 좀처럼 있을 수 없습니다」
PRINTFORM 「실은 이
CALL COLORMESSAGE("애욕을 독점하는 미라다의 돌 밀레니엄", C_D_PINK, 0, 1)
PRINTFORML 은 순도가 높은 매우 희소한 소재로 만들어졌고」
PRINTFORML 「완벽한 마법사가 몇 명이나 힘을 합쳐 술을 건 매우 수고 많은 상품인 것입니다」
PRINTFORM 「그래서, 통상의 판매 루트라면 가격은
CALL COLORMESSAGE(@"\\{DAILY_EV23_미라다_정가}", C_RED, 0, 1)
PRINTFORMW 로 되어 있습니다만……」
RETURN

path: ERB/이벤트관련/데일리이벤트/DAILY_EV23 방문판매_t.ERB


code:
@DAILY_CONDITION_EV28
SIF !TALENT:[[카게로]]:연모 && !TALENT:[[카게로]]:사모
	RETURN -1
SIF TALENT:[[카게로]]:육아중 || TALENT:[[카게로]]:임신
	RETURN -1
SIF MARK:[[카게로]]:반발각인
	RETURN -1
;誰かと同居中だと駄目
SIF PRIVATEROOM:(CFLAG:MASTER:초기위치 % 100)
	RETURN -1
;일시체재キャラがいると駄目
SIF FLAG:일시체재
	RETURN -1

IF !GETBIT(FLAG_DAILY_EV28, 0)
	;初回は必ず発生
ELSE
	LOCAL = 25 - CFLAG:[[카게로]]:1004
	SIF GETBIT(FLAG_DAILY_EV28, 1)
		LOCAL -= 2
	LOCAL = MAX(LOCAL, 5)
	SIF RAND:LOCAL > 0
		RETURN -1
ENDIF
;DAILY_EV28_発生回数 ++
CALL DAILY_EV28
RETURN 1

@DAILY_EV28
CFLAG:[[카게로]]:마지막내방일 = DAY
CFLAG:[[카게로]]:1004 = 0
IF !GETBIT(FLAG_DAILY_EV28, 0)
	CALL DAILY_EV28_0
ELSE
	CALL DAILY_EV28_1
ENDIF

;初回用
@DAILY_EV28_0
CALL SPTALK, [[카게로]], "通常", 0, @"「こんばんわー」", 1
PRINTFORMW 문을 열자 훌륭한 구리꿩을 들고있는 카게로가 서 있었다
PRINTFORML 「ね、これすごいでしょ？さっき獲ってきたの」
PRINTFORMW 「%CALLNAME:MASTER%料理得意でしょ？おすそ分けするからなにか作ってくれない？」
PRINTFORML
CALL ASK_YN("어쩔 수 없다", "대접해 준다면")
CALL EV28_MEAL(300)
CALL GET_ITEM("구리꿩", 5, 1)
PRINTFORML
PRINTFORML 「いまさらこんなことを聞くのもなんだけど…」
PRINTFORML 「いきなり夜中に押しかけて迷惑だった？」
PRINTFORML
CALL ASK_YN("언제라도 환영","또 선물 부탁해")
CALL PRINT_FACE, [[카게로]], "笑顔", "服", "", ""
IF !RESULT
	PRINTFORML 「ほんと？ありがとう！」
ELSE
	PRINTFORML 「今回みたいな大物はなかなか獲れないと思うけどね」
ENDIF
PRINTFORMW 「それじゃ、またお邪魔させてもらうわね」
CALL CHANGE_CFLAG(2,[[카게로]],150)
CALL CHANGE_CFLAG(4,[[카게로]],50)
SETBIT FLAG_DAILY_EV28, 0

;二回目以降
@DAILY_EV28_1
#DIM CAN_WOLFHUNT
#DIM MANHUNT
#DIM 변동치
VARSET LOCALS
CAN_WOLFHUNT = 0
MANHUNT = 0
IF CFLAG:[[카게로]]:생해금 == 2 && BASE:MASTER:정력 >= 500 && BASE:MASTER:체력 >= 1500
	CAN_WOLFHUNT = 1
ENDIF

CALL SPTALK, [[카게로]], "笑顔", 0, @"「こんばんわー」", 1
PRINTFORMW 또 카게로가 왔다
PRINTFORMW 「上がっていい？」
SELECTCASE RAND:12
	CASE 0
		PRINTFORML 「見て見て、大きいのが獲れたの！また捌いてくれない？」
		CALL EV28_MEAL(300)
		CALL GET_ITEM("구리꿩", 5, 1)
	CASE 1
		PRINTFORML 「珍しい石を見つけたから持ってきたわ」
		PRINTFORMW 「あ、姫には内緒よ？」
		CALL GET_ITEM("진사", 1, 1)
	CASE 2, 3
		IF !RAND:2
			PRINTFORML 「お魚を獲ってきたわ」
			PRINTFORMW 「ウサギ狩りしてよかったらもっとお肉が食べられるんだけどね」
		ELSE
			PRINTFORMW 「自慢できる程のじゃないけど、獲れたからお裾分けね」
		ENDIF
		CALL EV28_MEAL(100)
	CASE 4
		CALL EV28_MAMUSHI
	CASEELSE
		PRINTFORML 「はいお土産！」
		LOCALS = %TEXTR("죽순/고려인삼/산나물/쓴뿌리")%
		CALL GET_ITEM(LOCALS, 3, 1)
ENDSELECT
IF CFLAG:[[카게로]]:1006
	PRINTFORMW 「それとこれは姫からね」
	IF !RAND:10
		LOCALS = 용의손톱
		CALL GET_ITEM(LOCALS, 1, 1)
	ELSE
		LOCALS = %TEXTR("상약초/어박/약초/물고기/맛있는물고기/활석")%
		CALL GET_ITEM(LOCALS, 5, 1)
	ENDIF
	CFLAG:[[카게로]]:1006 = 0
ENDIF

PRINTFORMW
PRINTFORML 「何をしましょうか」
IF CAN_WOLFHUNT
	CALL ASK_M("잡담",1,"달맞이 술(필요：청주)",ITEM:청주,"늑대 사냥",1)
ELSE
	CALL ASK_M("잡담",1,"달맞이 술(필요：청주)",ITEM:청주)
ENDIF
SELECTCASE RESULT
	CASE 0
		변동치 = 100
	CASE 1
		ITEM:청주 --
		변동치 = 300
	CASE 2
		CALL WOLF_HUNTING
		RETURN 1
ENDSELECT
{
LOCALS = 짖는 연습을 하게 되었다/풀뿌리 요괴 네트워크 이야기를 했다/TRPG 시나리오 작성을 도왔다
/수렵 채집 이야기를 했다/마을 이야기를 했다/향림당 이야기를 했다/옷과 악세서리 이야기를 했다/새로운 게임을 했다
}
SIF CFLAG:[[카게로]]:생해금 == 2
	LOCALS += "/장래에 대해 이야기를 했다"
SIF CFLAG:[[카게로]]:아이인원수 - CFLAG:[[카게로]]:육아인원수
	LOCALS += "/독립해 나간 아이 이야기를 했다"

;晴れじゃないと月見酒の効果低下
IF !GROUPMATCH(TIME:5, 0, 1) && RESULT
	PRINTFORML 「今夜は月は見れないわね…」
	변동치 = 200
ENDIF
PRINTFORML \@ RESULT == 1? 술을 즐기면서# \@ %TEXTR(@"%LOCALS%")%

PRINTFORMW ………
PRINTFORMW ……
PRINTFORMW …
CALL BUFF_BASE(MASTER,BASE_기력,변동치,1)
CALL CHANGE_CFLAG(2,[[카게로]], 변동치)
CALL CHANGE_CFLAG(4,[[카게로]], 변동치/2)
IF TALENT:[[카게로]]:연모 && RESULT == 1 && !RAND:2
	IF !RAND:2
		CALL DAILY_EV28_2
	ELSE
		CALL DAILY_EV28_3
	ENDIF
	RETURN 1
ENDIF
PRINTFORMW 「そろそろお暇するわね」
PRINTFORMW 「それじゃ、おやすみなさい」
RETURN 1

@WOLF_HUNTING
#DIM 주입량
#DIM 기본주입량
#DIM 호감도상승량
#DIM 횟수
#DIM 경험증가수,58
VARSET 경험증가수
횟수 = MAX(MIN(ABL:[[카게로]]:욕망 / 2, 10) + CFLAG:[[카게로]]:욕구불만도 / 200,1)
주입량 = 0

경험증가수:0 += 1 + RAND:(횟수 * 3)
경험증가수:1 += 1 + RAND:(횟수 * 5)
경험증가수:20 += 횟수
경험증가수:4 += 3 + ABL:[[카게로]]:M감각 * 2
경험증가수:3 += RAND:(횟수 * 2)
경험증가수:5 += 경험증가수:0 / 2
경험증가수:6 = 횟수 * RAND(3, 5)
경험증가수:8 += 경험증가수:3 / 2
경험증가수:10 += 경험증가수:0 / 2
경험증가수:25 = 경험증가수:4
경험증가수:27 = 경험증가수:4
EXP:MASTER:M경험 += 경험증가수:4
EXP:MASTER:키스경험 += 경험증가수:4
EXP:MASTER:펠라티오경험 += 경험증가수:4
EXP:MASTER:삽입경험 += 경험증가수:20

CALL PRINT_FACE, [[카게로]], "発情", "服", "", ""
IF !GETBIT(FLAG_DAILY_EV28, 1)
	;初回
	PRINTFORMW 「へっ？えっと、それって……」
	PRINTFORMW 「あっ……」
	PRINTFORML ………
	PRINTFORML ……
	PRINTFORMW …
	PRINTFORMW 「んっ……ちゅっ……」
	PRINTFORMW 「もっと……」
	PRINTFORML ………
	PRINTFORML ……
	PRINTFORMW …
	CALL HPH_PRINT,@"「あっ……HPH　はあっ……んっ……HPH」","W"
	PRINTFORML ………
	PRINTFORML ……
	PRINTFORMW …
	{
	CALL HPH_PRINT,@"「%TEXTR("んっ/あっ/あんっ/フーーッ")%HPH %TEXTR("んっ/あっ/あんっ/フーーッ")%HPH
	ひんっ %TEXTR("んっ/あっ/あんっ/フーーッ")%HPH んはぁっHPH」","W"
	}
	PRINTFORML ………
	PRINTFORML ……
	PRINTFORMW …
	CALL HPH_PRINT,@"「ひーッHPH ひーッHPH ひぃーーッHPH ひぃいーーッHPH 」","W"
	PRINTFORML ………
	PRINTFORML ……
	PRINTFORMW …
	FONTBOLD
	CALL HPH_PRINT,@"「オ゛ッHPH オ゛ッHPH オ゛ッHPH オ゛オ゛オッHPH 」","W"
	CALL HPH_PRINT,@"「ア゛オ゛オ゛～～～～～～～～ッ ホオ゛～～～～～～～～～～～～～～～～HPH 」","W"
	FONTREGULAR
	PRINTFORML ………
	PRINTFORML ……
	PRINTFORMW …
	CALL PRINT_FACE, [[카게로]], "オナニー", "裸", "", "吐息"
	CALL HPH_PRINT,@"「はーーッ……HPH %CALLNAME:MASTER%に狩られちゃった……HPH」","W"
	PRINTFORMW 「狩った獲物は手放しちゃ駄目よ？」
	PRINTFORMW 카게로는 %CALLNAME:MASTER%에게 달려들어 안겼다
ELSE
	;二回目以降
	IF CFLAG:[[카게로]]:욕구불만도 < 500
		PRINTFORMW 「あっ…その……」
		PRINTFORMW 「お手柔らかに……」
	ELSE
		PRINTFORMW 「%TEXTR("うん…期待、してた……/あは♪　やっぱりしたかったのね/今夜も、いっぱい哭かされちゃうのね……/素敵な狩人さんに狙われちゃったわね")%」
		CALL HPH_PRINT,@"「んっHPH」","W"
	ENDIF
	PRINTFORML ………
	PRINTFORML ……
	PRINTFORMW …
	SELECTCASE RAND:3
		CASE 0
			CALL HPH_PRINT,@"「すきっHPH すきっHPH すきっHPH 」","W"
		CASE 1
			CALL HPH_PRINT,@"「%TEXTR("んっ/あっ/あんっ/フーーッ")%HPH %TEXTR("んっ/あっ/あんっ/フーーッ")%HPH% TEXTR("んっ/あっ/あんっ/フーーッ")%HPH」","W"
		CASE 2
			CALL HPH_PRINT,@"「はあぁ～～～～HPH HPH 」","W"
	ENDSELECT
	PRINTFORML ………
	PRINTFORML ……
	PRINTFORMW …
	FONTBOLD
	SELECTCASE RAND:3
		CASE 0
			CALL HPH_PRINT,@"「ほお゛おお゛～～～～HPH お゛お゛お゛お゛ーーHPH 」","W"
		CASE 1
			CALL HPH_PRINT,@"「%TEXTR("ああ゛/ほお゛おお゛/あ゛～/はへえ")%～～～～～～～HPH」","W"
		CASE 2
			CALL HPH_PRINT,@"「オ゛オ゛ーッHPH オ゛オ゛ーッHPH オ゛オ゛ーッHPH オ゛オ゛オオ゛ーッHPH 」","W"
	ENDSELECT
	FONTREGULAR
	PRINTFORML ………
	PRINTFORML ……
	PRINTFORMW …
	CALL PRINT_FACE, [[카게로]], "オナニー", "裸", "", "吐息"
	CALL HPH_PRINT,@"「もう、いっつも激しいんだから……HPH」","W"
ENDIF

기본주입량 = (ABL:[[카게로]]:질 + 1) * 20 + BASE:MASTER:정력 / 20
FOR LOCAL,1,횟수
	주입량 += 기본주입량 + RAND:(기본주입량)
NEXT

주입량 = MIN(주입량, BASE:MASTER:정력)

CALL ADD_SAMEN_V([[카게로]], 주입량)
SETCOLOR C_YELLOW
FOR LOCAL,0,58
	IF 경험증가수:LOCAL
		PRINTFORML %EXPNAME:LOCAL% ＋{경험증가수:LOCAL}
		EXP:[[카게로]]:LOCAL += 경험증가수:LOCAL
	ENDIF
NEXT
RESETCOLOR
CALL JUEL_UP_SPECIAL([[카게로]],"욕정","욕망",20,1)
CALL JUEL_UP_SPECIAL([[카게로]],"쾌V","V감각",15,1)
CALL ADD_EXP_LESNIAN(횟수 * 5, [[카게로]], "PRINT")
호감도상승량 = 횟수 * ABL:[[카게로]]:친밀 * 10
CALL CHANGE_CFLAG(2,[[카게로]],호감도상승량)
;竹林以外が拠点なら泊まっていく
IF MAIN_MAP != 4 && !FLAG:일시체재
	FLAG:일시체재 = [[카게로]]
	CFLAG:[[카게로]]:본래의초기위치 = CFLAG:[[카게로]]:초기위치
	CFLAG:[[카게로]]:초기위치 = CFLAG:MASTER:초기위치
	CFLAG:[[카게로]]:신사거주 = CFLAG:[[카게로]]:초기위치
ENDIF
CFLAG:[[카게로]]:욕구불만도 = 0
CALL RECOVER(MASTER,횟수 * -300,"체력", @"{횟수}회에 걸친 밤샘교미")
CALL RECOVER(MASTER,횟수 * -300,"기력", @"{횟수}회에 걸친 밤샘교미")

CFLAG:MASTER:철야 = 1
SETBIT FLAG_DAILY_EV28, 1

;人里長屋限定
IF CFLAG:MASTER:초기위치 == 230
	CALL PRINT_FACE, [[카게로]], "通常", "裸", "", ""
	IF !DAILY_EV28_벽쿵
		PRINTFORML 「ところでシてる最中、お隣の壁がすごいドンドンいってたんだけど……」
		PRINTFORML
		CALL ASK_YN("흔히 있는 일","인간 마을의 풍습")
		IF !RESULT
			PRINTFORMW 「変わったお隣さんなのね」
		ELSE
			PRINTFORMW 「人間はよくわからないことするわね」
		ENDIF
	ELSEIF DAILY_EV28_벽쿵 == 5
		PRINTFORML 「最近は隣のドンドン聞かないわね、馴れたのかしら？」
	ELSEIF DAILY_EV28_벽쿵 < 3
		PRINTFORML 「お隣、またドンドンいってたわね」
		PRINTFORML
		CALL ASK_YN("신경쓰지마","세례받고 있어")
		IF !RESULT
			PRINTFORMW 「%CALLNAME:MASTER%がそう言うならそうなのかな」
		ELSE
			PRINTFORMW 「流石にそれはないと思うけど」
		ENDIF
	ENDIF
	IF DAILY_EV28_벽쿵 >= 5
		LOCAL = 횟수 / 3 + 1
		CALL COLORMESSAGE(@"자위경험＋{LOCAL}（반키）",C_PINK,1)
		EXP:[[반키]]:자위경험 += LOCAL
		CALL JUEL_UP_SPECIAL([[반키]],"치정","자위중독", LOCAL)
		CALL JUEL_UP_SPECIAL([[반키]],"욕정","욕망",5 + LOCAL,1)
		CFLAG:[[반키]]:욕구불만도 = 0
	ELSE
		CALL JUEL_UP_SPECIAL([[반키]],"욕정","욕망",5,1)
		CFLAG:[[반키]]:욕구불만도 += 500
		CALL COLORMESSAGE(@"반키의 욕구 불만이 높아졌다",C_YELLOW,1)
		CALL COLORMESSAGE(@"반키의 기분이 나빠졌다",C_RED,1)
		TALENT:[[반키]]:기분 = -1
	ENDIF
	DAILY_EV28_벽쿵 ++
ENDIF

@MAN_HUNTING
;相手から襲ってくるパターン（未作成）
IF !RAND:2
	PRINTFORMW 「ねえ」
	PRINTFORMW 「%CALLNAME:MASTER%が欲しくなっちゃった」
	PRINTFORMW 「ふふ、人間狩りの時間ね」
ELSE
	PRINTFORMW 「こうやって二人で過ごすのは楽しいけど、家族が増えたらもっと楽しいわ」
	PRINTFORMW 「ね、増やしましょう？」
ENDIF

PRINTFORML 「%CALLNAME:MASTER%のカラダって好きよ」
PRINTFORML 「綺麗で、引き締まってて、美味しそうな牡鹿みたい」
PRINTFORML 「今夜はぜんぶ、私のものだから」

PRINTFORML 「あら、お疲れ？」
PRINTFORML 「それじゃあ続きは帰ってからにしましょうね」
PRINTFORML %CALLNAME:MASTER%はオオカミの巣穴にお持ち帰りされてしまいました

@EV28_MEAL(ARG)
PRINTFORMW ………
PRINTFORMW ……
PRINTFORMW …
PRINTFORMW 「御馳走様！おいしかったわ」
CALL BUFF_BASE(MASTER,BASE_체력,ARG,1)
CALL BUFF_BASE(MASTER,BASE_기력,ARG,1)
CALL BUFF_BASE(MASTER,BASE_정력,ARG,1)
CALL BUFF_BASE([[카게로]],BASE_체력,ARG,1)
CALL BUFF_BASE([[카게로]],BASE_기력,ARG,1)
EXP:MASTER:요리경험 ++

@DAILY_EV28_2
PRINTFORML 「なんだかいい気分……」
PRINTFORML 「ね、踊りましょう？」
LOCALS = %TEXTR("ぐるぐる/ゆらゆら/ふらふら/バタバタ/ピョコピョコ")%
$choice
CALL ASK_M("踊る",1,"どうやって？",1,"音楽を奏でる",1)
SELECTCASE RESULT
	CASE 0
	CASE 1
		PRINTFORMW 「私もよく知らないけど……なんか%LOCALS%動けばいいと思うわ！」
	CASE 2
		PRINTFORML 「音楽とかいいから、私は%CALLNAME:MASTER%と踊りたいの！」
		PRINTFORMW 怒られた
		GOTO choice
ENDSELECT
PRINTFORMW 夜更けまで二人で%LOCALS%と踊った
TALENT:[[카게로]]:기분 = 1
CALL CHANGE_CFLAG(2,[[카게로]], 300)

@DAILY_EV28_3
#DIM 周回数
#DIM cost
#DIM finish
finish = 0
周回数 = 1
PRINTFORML 「ふふ、酔ってきちゃった」
PRINTFORML 「ね、抱っこして」
CALL ASK_M("抱っこ",1,"お姫様抱っこ",1,"ブライダルお姫様抱っこ",1)
PRINTFORML %CALLNAME:MASTER%は카게로を抱え上げた
PRINTFORML 結構重い……
PRINTFORML 「それじゃこのままこの辺りを一周して」
$LOOP
cost = 周回数 * 100
CALL COLORMESSAGE(@"残り체력{BASE:MASTER:체력}",C_YELLOW,1)
CALL ASK_M("一周する", BASE:MASTER:체력 > cost ,"時間停止でサボりながらやる", BASE:MASTER:체력 > cost/ 2 && BASE:MASTER:TSP > cost, "疲れた", 周回数 > 1)
SELECTCASE RESULT
	CASE 0
		PRINTFORMW ………
		PRINTFORMW ……
		PRINTFORMW …
		PRINTFORMW 「んふー♪」
		CALL RECOVER(MASTER,cost * -1,"체력")
		CALL BUFF_BASE([[카게로]],BASE_기력,50,1)
		CALL CHANGE_CFLAG(2,[[카게로]], 100)
		周回数 ++
	CASE 1
		PRINTFORMW ………
		PRINTFORMW ……
		PRINTFORMW …
		PRINTFORMW 「なんか短くない？　酔ってるせいかしら……」
		CALL RECOVER(MASTER,cost * -1 / 2,"체력")
		CALL RECOVER(MASTER,cost * -1,"TSP")
		CALL BUFF_BASE([[카게로]],BASE_기력,30,1)
		CALL CHANGE_CFLAG(2,[[카게로]], 30)
		周回数 ++
	CASE 2
		finish = 1
ENDSELECT
IF !finish
	SIF RAND:周回数 > RAND:5 || 周回数 >= 5
		finish = 2
ENDIF
IF finish == 1
	IF 周回数 < 3
		PRINTFORML 「もう？　しょうがないわね……」
		PRINTFORML 「あふ……　それじゃあ、おやすみなさい」
	ELSE
		PRINTFORML 「お疲れ様！楽しかったわ」
		PRINTFORML 「おやすみなさい、また来るわね」
		PRINTFORMW 카게로는 기분 좋게 돌아갔다
		CALL CHANGE_CFLAG(2,[[카게로]], 300)
		TALENT:[[카게로]]:기분 = 1
		CFLAG:[[카게로]]:다음날내방 = 1
	ENDIF
	RETURN
ELSEIF finish == 2
	PRINTFORML 「………ぐぅ」
	PRINTFORML 寝落ちしたようだ
	CALL CHANGE_CFLAG(2,[[카게로]], 300)
	;竹林以外が拠点なら泊まっていく
	TALENT:[[카게로]]:기분 = 1
	IF MAIN_MAP != 4
		FLAG:일시체재 = [[카게로]]
		CFLAG:[[카게로]]:본래의초기위치 = CFLAG:[[카게로]]:초기위치
		CFLAG:[[카게로]]:초기위치 = CFLAG:MASTER:초기위치
		CFLAG:[[카게로]]:신사거주 = CFLAG:[[카게로]]:초기위치
	ENDIF
	RETURN
ENDIF
PRINTFORMW 「もう一周して」
GOTO LOOP

@EV28_MAMUSHI
IF FIRSTTIME("土産マムシ", 0, [[카게로]])
	PRINTFORMW 「見てー！マムシー！」
	PRINTFORMW まだ動いてる蛇の死体を見せつけてきた……
	PRINTFORML 「食べたことないの？塩焼きにすると美味しいのよ」
	PRINTFORMW 「いつも作ってもらってるから、今日は私がやるわね！」
	PRINTFORMW ……
	PRINTFORMW 意外と美味しかった
ELSE
	PRINTFORMW またマムシ捕ってきたわよ
	{
	PRINTFORMW %TEXTR("心臓いる？じゃあ私が貰うわね/皮をビーッって剥くのが楽しいのよね/
	姫たちは食べてくれないのよね、美味しいのに/
	レアでも美味しいんだけど、%CALLNAME:MASTER%がお腹壊すといけないからちゃんと焼かないとね")%
	}
	PRINTFORMW ……
ENDIF
CALL BUFF_BASE(MASTER,BASE_체력,50,1)
CALL BUFF_BASE(MASTER,BASE_기력,50,1)
CALL BUFF_BASE(MASTER,BASE_정력,300,1)
CALL BUFF_BASE([[카게로]],BASE_체력,50,1)
CALL BUFF_BASE([[카게로]],BASE_기력,50,1)
EXP:[[카게로]]:요리경험 ++

;PRINTFORMW 「所詮私はただの妖獣だものねー……」
;CALL ASK_YN("そう言うな","気にしてる？")
;IF !RESULT
;
;ELSE
;	PRINTFORMW 「そりゃあね」
;	PRINTFORMW 「近所には月のお姫様が住んでいて」
;	PRINTFORMW 「山に狩りに行けば河童や天狗がのさばってて」
;	PRINTFORMW 「友達のところに遊びに行けば紅い悪魔のお屋敷がでーんと建ってるのが目に入ってくるし」
;	PRINTFORMW 「気にしない方が無理でしょ」
;	IF TALENT:恋慕
;		PRINTFORMW 「まあ、最近は昔ほどは気にならないわ」
;		PRINTFORMW 「%CALLNAME:MASTER%にちやほやされると、自分が何かひとかどの妖怪にでもなったような気分にされちゃうからね」
;		CALL ASK_YN("ならもっとちやほやしよう","すごい存在になりたい？")
;		IF !RESULT
;			PRINTFORMW 「ふふ、ありがと」
;		ELSE
;			{
;			PRINTFORMW 「う～～ん……はっきりと何になりたいとか何がしたいとかあるわけじゃないし、
;			人狼なのが嫌ってわけでもないんだけど……」
;			}
;			PRINTFORMW 「嫌なことがあったとき、自分が別の存在だったらこんな思いしないで済むのにって感じることがあるのよ」
;			CALL ASK_YN("ない物ねだりするな","人間でもそういうことはある")
;			IF !RESULT
;				PRINTFORMW 「そう言われたらそれまでだけどね」
;			ELSE
;				PRINTFORMW 「ほんと？人間も人間で、それなりに面倒があるのね」
;			ENDIF
;		ENDIF
;	ENDIF
;ENDIF

path: ERB/이벤트관련/데일리이벤트/DAILY_EV28 왕래하는늑대.ERB


code:
@DAILY_CONDITION_EV29
SIF !K63_KENGAKU
	RETURN -1
K63_KENGAKU = 0
RETURN 1

@DAILY_EV29
CFLAG:[[카게로]]:1005 = 0
FLAG_DAILY_EV29_LASTTIME = DAY
CFLAG:[[카게로]]:마지막내방일 = DAY
CFLAG:[[와카사기히메]]:마지막내방일 = DAY
CFLAG:[[반키]]:마지막내방일 = DAY
IF !GETBIT(FLAG_DAILY_EV29, 0)
	CALL DAILY_EV29_0
ELSE
	CALL DAILY_EV29_1
ENDIF
RETURN 1

;初回用
@DAILY_EV29_0
CALL PRINT_FACE, [[카게로]], "通常", "服", "", ""
PRINTFORMW 「来たわねー」
CALL PRINT_FACE, [[와카사기히메]], "通常", "服", "", ""
PRINTFORMW 「오늘은 よろしくね」
CALL PRINT_FACE, [[카게로]], "通常", "服", "", ""
PRINTFORML 「どうこの恰好？かなり人間っぽくない？」
CALL ASK_YN("割と","多分バレない")
CALL PRINT_FACE, [[카게로]], "笑顔", "服", "", ""
PRINTFORMW 「でしょー？それじゃ、반키のところへしゅっぱーつ！」
PRINTFORMW ………
PRINTFORMW ……
PRINTFORMW …
CALL PRINT_FACE, [[와카사기히메]], "通常", "服", "", ""
PRINTFORMW 「いいお店ね」
CALL PRINT_FACE, [[카게로]], "通常", "服", "", ""
PRINTFORMW 「給仕服似合ってるじゃない」
CALL PRINT_FACE, [[반키]], "通常", "服", "", ""
PRINTFORMW 「ゲッ……なんでこんなところに……」
PRINTFORML 「%CALLNAME:MASTER%の仕業か」
PRINTFORML 「なんでこいつら連れてきた」
IF TALENT:[[카게로]]:연모
	CALL ASK_M("반키に会いたくて",1,"카게로の頼みとあらば断れない",1,"この煮込み雑炊一つください",1)
ELSE
	CALL ASK_M("반키に会いたくて",1,"카게로に頼まれたから",1,"この煮込み雑炊一つください",1)
ENDIF
SELECTCASE RESULT
	CASE 0
		CALL PRINT_FACE, [[반키]], "通常", "服", "", ""
		PRINTFORML 「連れてきた理由になってないだろ」
		PRINTFORML 「まあいい、さっさと食ってさっさと帰れ」
		CALL CHANGE_CFLAG(2,[[반키]],50)
		CALL CHANGE_CFLAG(4,[[반키]],10)
	CASE 1
		IF TALENT:[[카게로]]:연모
			CALL PRINT_FACE, [[카게로]], "発情", "服", "", ""
			PRINTFORML 「もう、すぐそういうこと言うんだから、もう……」
			CALL CHANGE_CFLAG(2,[[카게로]],100)
			CALL PRINT_FACE, [[반키]], "通常", "服", "", ""
			PRINTFORMW 「카게로うざい」
		ELSE
			CALL PRINT_FACE, [[반키]], "通常", "服", "", ""
			PRINTFORML 「こっちの都合も考えろよ」
			PRINTFORML 「まあいい、さっさと食ってさっさと帰れ」
		ENDIF
	CASE 2
		CALL PRINT_FACE, [[반키]], "通常", "服", "", ""
		PRINTFORMW 「それは今やってない」
		PRINTFORMW 「この世の終わりみたいな顔してないでさっさと他のもん注文しろ」
		DAILY_EV29_NIKOMI ++
ENDSELECT
PRINTFORMW ………
PRINTFORMW ……
PRINTFORMW …
CALL BUFF_BASE(MASTER,BASE_체력,150,1)
CALL BUFF_BASE(MASTER,BASE_기력,150,1)
CALL BUFF_BASE([[카게로]],BASE_체력,150,1)
CALL BUFF_BASE([[카게로]],BASE_기력,150,1)
CALL BUFF_BASE([[와카사기히메]],BASE_체력,150,1)
CALL BUFF_BASE([[와카사기히메]],BASE_기력,150,1)
CALL PRINT_FACE, [[와카사기히메]], "通常", "服", "", ""
PRINTFORMW 「美味しかったね」
PRINTFORMW 「あーそうだお金要るんだった」
CALL PRINT_FACE, [[와카사기히메]], "笑顔", "服", "", ""
PRINTFORMW 「私は持ってないけどね！」
CALL PRINT_FACE, [[카게로]], "笑顔", "服", "", ""
PRINTFORMW 「私も！」
CALL 画像セット(@"顔絵_服_通常_{[[카게로]]}", 0, 0, 100, 0, 0, "", "デフォルト顔絵")
CALL 画像セット(@"顔絵_服_通常_{[[와카사기히메]]}", 0, 0, 100, 0, 0, "", "デフォルト顔絵")
CALL 画像一斉表示("左", 0, 1)
PRINTFORMW 二人は揃って%CALLNAME:MASTER%の方を見た
CALL COLORMESSAGE(@"소지금: %금액표시(MONEY)%",C_YELLOW,1)
CALL ASK_M("立て替える（\10000）", MONEY >= 10000, "반키の方を見る",1)
SELECTCASE RESULT
	CASE 0
		MONEY -= 10000
	CASE 1
		CALL PRINT_FACE, [[반키]], "通常", "服", "", ""
		PRINTFORMW 「なんで私の方を見るんだよ」
		CALL PRINT_FACE, [[반키]], "怒り", "服", "", ""
		IF MONEY >= 10000
			PRINTFORMW 「ちゃんと持ってるじゃねーか！払えよ！」
			MONEY -= 10000
		ELSE
			CALL EV29_無銭飲食
		ENDIF
ENDSELECT

IF !FLAG_DAILY_EV29_DEBT
	CALL PRINT_FACE, [[와카사기히메]], "笑顔", "服", "", ""
	PRINTFORMW 「ありがとー！」
	CALL PRINT_FACE, [[카게로]], "笑顔", "服", "", ""
	PRINTFORMW 「今度埋め合わせするわね」
	CALL PRINT_FACE, [[반키]], "通常", "服", "", ""
	PRINTFORMW 「もう来るなよ」
	;通い狼に影響
	CFLAG:[[카게로]]:1004 += 3
	CFLAG:[[카게로]]:1006 ++
	FLAG_DAILY_EV29_PAID ++
	CALL CHANGE_CFLAG(2,[[카게로]],100)
	CALL CHANGE_CFLAG(4,[[카게로]],50)
	CALL CHANGE_CFLAG(2,[[와카사기히메]],100)
	CALL CHANGE_CFLAG(4,[[와카사기히메]],50)
	CALL CHANGE_CFLAG(2,[[반키]],100)
	CALL CHANGE_CFLAG(4,[[반키]],50)
ELSE
	CALL CHANGE_CFLAG(2,[[카게로]],50)
	CALL CHANGE_CFLAG(2,[[와카사기히메]],50)
ENDIF
SETBIT FLAG_DAILY_EV29, 0

;二回目以降
@DAILY_EV29_1
#DIM PAYMENT
#DIM C_ID
IF GROUPMATCH(TIME:5, 4, 5, 6)
	CALL PRINT_FACE, [[와카사기히메]], "笑顔", "服", "", ""
	PRINTFORMW 「오늘은 降ってるから泳いで来れたの」
	CALL PRINT_FACE, [[카게로]], "笑顔", "服", "", ""
	PRINTFORMW 「いつもこうなら楽なんだけどね」
ELSEIF !RAND:2
	CALL PRINT_FACE, [[카게로]], "通常", "服", "", ""
	PRINTFORMW 「あら、お待たせしちゃった？」
	CALL PRINT_FACE, [[와카사기히메]], "通常", "服", "", ""
	PRINTFORMW 「오늘은 早いのね」
ELSEIF !RAND:4
	CALL PRINT_FACE, [[와카사기히메]], "通常", "服", "", ""
	PRINTFORMW 「오늘은 %CALLNAME:MASTER%が押してくれる？」
	CALL PRINT_FACE, [[카게로]], "笑顔", "服", "", ""
	PRINTFORMW 「濡れタオル巻いてあるからちょっと重いわよー？」
ELSE
	CALL PRINT_FACE, [[와카사기히메]], "通常", "服", "", ""
	PRINTFORMW 「こんばんわ」
	CALL PRINT_FACE, [[카게로]], "通常", "服", "", ""
	PRINTFORMW 「さ、行きましょ」
ENDIF
PRINTFORMW ………
PRINTFORMW ……
PRINTFORMW …
IF !RAND:5
	CALL SPTALK, [[카게로]], "通常", 0, @"「ｸﾝｸﾝ…… 오늘은 %TEXTR("고기/물고기/야채/산나물")%が美味しいと思うわ」", 1
	CALL SPTALK, [[와카사기히메]], "通常", 0, @"「카게로ちゃん匂い嗅ぐのやめなよ」", 1
ELSE
	IF !RAND:2
		CALL SPTALK, [[와카사기히메]], "通常", 0, @"「こんばんわー」", 1
	ELSE
		CALL SPTALK, [[와카사기히메]], "通常", 0, @"「오늘은 割と%TEXTR("混んでる/空いてる")%ね」", 1
	ENDIF
	CALL SPTALK, [[카게로]], "通常", 0, @"「%TEXTR("また来たわよー/その髪型いいわね/ねえ今日の姫のファッション可愛くない？")%」", 1
ENDIF

CALL PRINT_FACE, [[반키]], "通常", "服", "", ""
IF FLAG_DAILY_EV29_DEBT > 0
	PRINTFORML 「오늘은 金あるんだろうな」
ELSEIF !RAND:10
	PRINTFORML 「狙いすましたように来るなお前ら、シフト変えたのに……」
ELSEIF !RAND:5
	PRINTFORML 「ぶぶ漬け出してやりたい……」
ELSE
	PRINTFORML 「またお前らか」
ENDIF
PRINTFORML 「ご注文は？」
C_ID = 0
CALL ASK_M("本日のおすすめ",1,"카게로と同じやつ",1,"와카사기히메と同じやつ",1,"この煮込み雑炊一つください",1)
;CALL ASK_M("本日のおすすめ",1,"카게로と同じやつ",1,"와카사기히메と同じやつ",1,"この煮込み雑炊一つください",1, "パーッといく(\30000)",MONEY >= 30000)
SELECTCASE RESULT
	CASE 0
		C_ID = [[반키]]
	CASE 1
		C_ID = [[카게로]]
	CASE 2
		C_ID = [[와카사기히메]]
	CASE 3
		DAILY_EV29_NIKOMI ++
		IF DAY:2 == 2
			PRINTFORMW 「この暑いのに物好きなことで」
		ELSEIF DAILY_EV29_NIKOMI >= 5 && !RAND:5
			PRINTFORML 「%CALLNAME:MASTER%はそればっかだな」
			PRINTFORML 「そんなに旨いの、それ」
			CALL ASK_YN("旨い","別に")
			IF !RESULT
				PRINTFORML 「ふーん」
			ELSE
				PRINTFORML 「お前はそういうやつだよ」
			ENDIF
		ENDIF
ENDSELECT
IF C_ID
	CALL CHANGE_CFLAG(2,C_ID,100)
	CALL CHANGE_CFLAG(4,C_ID,50)
ENDIF
PRINTFORMW ………
PRINTFORMW ……
PRINTFORMW …
CALL BUFF_BASE(MASTER,BASE_체력,150,1)
CALL BUFF_BASE(MASTER,BASE_기력,150,1)
CALL BUFF_BASE([[카게로]],BASE_체력,150,1)
CALL BUFF_BASE([[카게로]],BASE_기력,150,1)
CALL BUFF_BASE([[와카사기히메]],BASE_체력,150,1)
CALL BUFF_BASE([[와카사기히메]],BASE_기력,150,1)
PAYMENT = 0
IF FLAG_DAILY_EV29_DEBT > 0
	CALL PRINT_FACE, [[반키]], "通常", "服", "", ""
	PRINTFORML 「この間のもまとめて払えよ」
	CALL COLORMESSAGE(@"소지금: %금액표시(MONEY)%　勘定:%금액표시(10000)%　借金:%금액표시(FLAG_DAILY_EV29_DEBT)%",C_YELLOW,1)
	CALL ASK_M("耳を揃えて返す", MONEY >= 10000 + FLAG_DAILY_EV29_DEBT, "今日のお代だけで勘弁して貰う", MONEY >= 10000,"ツケで",1)
	SELECTCASE RESULT
		CASE 0
			PRINTFORMW 「よろしい」
			MONEY -= (10000 + FLAG_DAILY_EV29_DEBT)
			FLAG_DAILY_EV29_DEBT = 0
			FLAG_DAILY_EV29_PAID ++
			CALL CHANGE_CFLAG(2,[[반키]],100)
			CALL CHANGE_CFLAG(4,[[반키]],50)
			PAYMENT = 1
		CASE 1
			PRINTFORMW 「さっさと返せ」
			MONEY -= 10000
			FLAG_DAILY_EV29_PAID ++
			PAYMENT = 1
		CASE 2
			CALL PRINT_FACE, [[반키]], "怒り", "服", "", ""
			PRINTFORML 「ナメてんのか」
			FLAG_DAILY_EV29_DEBT += 10000
	ENDSELECT
ELSE
	CALL SPTALK, [[반키]], "通常", 0, @"「お勘定」", 1
	CALL COLORMESSAGE(@"소지금: %금액표시(MONEY)%　勘定:%금액표시(10000)%",C_YELLOW,1)
	IF !GETBIT(FLAG_DAILY_EV29, 1)
		CALL ASK_M("払う", MONEY >= 10000, "払えない",1)
	ELSE
		CALL ASK_M("払う", MONEY >= 10000, "ツケで",1)
	ENDIF
	SELECTCASE RESULT
		CASE 0
			MONEY -= 10000
			PAYMENT = 1
		CASE 1
			CALL EV29_無銭飲食
	ENDSELECT
ENDIF
;ちゃんと払った
IF PAYMENT
	CALL PRINT_FACE, [[카게로]], "笑顔", "服", "", ""
	PRINTFORML 「御馳走様！」
	CALL PRINT_FACE, [[와카사기히메]], "笑顔", "服", "", ""
	PRINTFORML 「ありがとー！」
	CALL PRINT_FACE, [[반키]], "通常", "服", "", ""
	PRINTFORMW 「お前らも当然のようにタカるな」
	;通い狼に影響
	CFLAG:[[카게로]]:1004 += 3
	CFLAG:[[카게로]]:1006 ++
	FLAG_DAILY_EV29_PAID ++
	CALL CHANGE_CFLAG(2,[[카게로]],100)
	CALL CHANGE_CFLAG(4,[[카게로]],50)
	CALL CHANGE_CFLAG(2,[[와카사기히메]],100)
	CALL CHANGE_CFLAG(4,[[와카사기히메]],50)
	CALL CHANGE_CFLAG(2,[[반키]],100)
	CALL CHANGE_CFLAG(4,[[반키]],50)
	IF !RAND:10
		CALL PRINT_FACE, [[카게로]], "通常", "服", "", ""
		PRINTFORML 「今日のはちょっと塩気が強かったかな」
		CALL PRINT_FACE, [[반키]], "通常", "服", "", ""
		PRINTFORMW 「やかましい」
	ELSEIF !RAND:5
		CALL PRINT_FACE, [[카게로]], "通常", "服", "", ""
		PRINTFORML 「またお肉持っていくわね」
		CALL PRINT_FACE, [[반키]], "通常", "服", "", ""
		PRINTFORMW 「生じゃないやつで頼む」
	ELSEIF !RAND:6
		CALL PRINT_FACE, [[카게로]], "通常", "服", "", ""
		PRINTFORML 「ところで次の会合なんだけど」
		CALL PRINT_FACE, [[반키]], "通常", "服", "", ""
		PRINTFORMW 「ここでその話はするな」
	ELSEIF !RAND:10
		CALL PRINT_FACE, [[와카사기히메]], "通常", "服", "", ""
		PRINTFORML 「今度の川下り応援に来てよ」
		CALL PRINT_FACE, [[반키]], "通常", "服", "", ""
		PRINTFORMW 「行けたらね」
	ELSE
		CALL PRINT_FACE, [[와카사기히메]], "笑顔", "服", "", ""
		PRINTFORML 「また来るね」
		CALL PRINT_FACE, [[반키]], "通常", "服", "", ""
		PRINTFORMW 「来ないでいい」
	ENDIF
;払ってない
ELSE
	CALL SPTALK, [[반키]], "怒り", 0, @"「さっさと返せよ」", 1
	CALL CHANGE_CFLAG(2,[[카게로]],50)
	CALL CHANGE_CFLAG(2,[[와카사기히메]],50)
ENDIF

@EV29_無銭飲食
IF !GETBIT(FLAG_DAILY_EV29, 1)
	CALL PRINT_FACE, [[반키]], "怒り", "服", "", ""
	PRINTFORMW 「マジで金ないのかよ！どーすんだよ！」
	PRINTFORMW 不審に思ったのか厨房から声が飛んだ
	CALL PRINT_FACE, [[반키]], "通常", "服", "", ""
	PRINTFORMW 「なんでもないでーす！」
	PRINTFORMW 「仕方ない……不足分は私が立て替えてやる」
	CALL PRINT_FACE, [[카게로]], "笑顔", "服", "", ""
	PRINTFORMW 「助かる！」
	CALL PRINT_FACE, [[와카사기히메]], "笑顔", "服", "", ""
	PRINTFORMW 「반키ちゃん大好き！」
	PRINTFORML
	CALL ASK_YN("すぐ返します","반키ちゃん大好き！")
	IF !RESULT
		CALL PRINT_FACE, [[반키]], "怒り", "服", "", ""
		PRINTFORMW 「当たり前だ」
	ELSE
		CALL PRINT_FACE, [[반키]], "怒り", "服", "", ""
		PRINTFORMW 「黙れアホ」
	ENDIF
ELSE
	CALL SPTALK, [[반키]], "怒り", 0, @"「%TEXTR("調子乗るな/ナメてんのか/貧乏人め")%」", 1
ENDIF
SETBIT FLAG_DAILY_EV29, 1
FLAG_DAILY_EV29_DEBT += 10000
CALL COLORMESSAGE(@"반키に借金しました　類型\\{FLAG_DAILY_EV29_DEBT}",C_YELLOW,2)

path: ERB/이벤트관련/데일리이벤트/DAILY_EV29 직장견학.ERB


code:
@DAILY_CONDITION_EV30

IF !RAND:15
	CALL DAILY_EV30
	RETURN 1
ELSE
	RETURN -1
ENDIF

@DAILY_EV30
PRINTFORML 오늘은 술벌레의 상태가 좋은 것 같다
CALL COLORMESSAGE(@"평소보다 양조가 진행되었다",C_YELLOW,1)
DAY:10 += 2
RETURN 1

path: ERB/이벤트관련/데일리이벤트/DAILY_EV30 양조쾌조.ERB


code:
@DAILY_CONDITION_EV33

IF FLAG:시장의신 == 1
	CALL DAILY_EV33
	RETURN 1
ELSE
	RETURN -1
ENDIF

@DAILY_EV33
;이 후에 플래그가 재설정된다
FLAG:시장의신 = 2
PRINTFORML 시장의 신의 힘이 토산품을 활성화시켰다！
PRINTFORMW 기념품 가게의 라인업이 대폭 변화했다！
RETURN 1

path: ERB/이벤트관련/데일리이벤트/DAILY_EV33 시장의신.ERB


code:
;DAILY_EV23의 작성해 광역 변수를 갖고 싶어졌다
;다른 데일리용으로도 부디

;데일리 이벤트 최대 수（40であれば EV0 ～ EV39 まで）
#DEFINE DAILY_EV_MAX 40

;----------------------------
;EV23：방문판매
;----------------------------
#DIM SAVEDATA FLAG_DAILY_EV23
;BIT0 이벤트가 한 번이라도 발생했다
;BIT1 맥 이루어, 이제 두 번 다시 판매원은 오지 않는다
;BIT2 설명을 끝까지 (들)물었다
;BIT3 미라다의 돌을 구입했다
;BIT4 판매원이 섹프가 되었다

#DIM SAVEDATA FLAG_DAILY_EV23_판매원

#DIM CONST DAILY_EV23_미라다_정가 = 980000
#DIM CONST DAILY_EV23_미라다_저가 = 49800
#DIM CONST DAILY_EV23_미라다_바닥시세 = 29800

;----------------------------
;EV28：通い狼
;----------------------------
#DIM SAVEDATA FLAG_DAILY_EV28
;BIT0 イベントが一度でも発生した
;BIT1 オオカミ狩り

#DIM SAVEDATA DAILY_EV28_벽쿵

;----------------------------
;EV29：職場見学
;----------------------------
#DIM SAVEDATA FLAG_DAILY_EV29
;BIT0 イベントが一度でも発生した
;BIT1 ツケ
#DIM SAVEDATA DAILY_EV29_NIKOMI			;煮込み雑炊
#DIM SAVEDATA FLAG_DAILY_EV29_PAID		;支払回数
#DIM SAVEDATA FLAG_DAILY_EV29_DEBT		;借金
#DIM SAVEDATA FLAG_DAILY_EV29_LASTTIME	;最後に発生した日


code:
@DAILY_CONDITION_EV0
SIF RAND:8 > 0
	RETURN -1
SIF DAY <= 30
	RETURN -1
SIF AVERAGE_YOGORE() > 800
	RETURN -1

CALL DAILY_EV0
RETURN 1

@DAILY_EV0
PRINTFORML %마스터가% 자는 사이에 %GET_MAPNAME(MAIN_MAP)%에 요정 대군이 몰려왔다!
IF RAND:10 < ABL:MASTER:전투능력 && AT_HOME(MASTER) == 1
	PRINTFORML 요정들은 %GET_MAPNAME(MAIN_MAP)% 안을 어지럽히고 있었지만
	PRINTFORML 도중에 눈치챈 %마스터가% 나오자마자, 쏜살같이 도망쳤다
	FOR LOCAL, MINROOM(), MAXROOM
		IF RAND:10 <= ABL:MASTER:전투능력
			YOGORE:LOCAL = MIN(YOGORE:LOCAL + 1000 + RAND:4000, 6000)
		ENDIF
	NEXT
;블랙썬더를 10개 이상 소지
ELSEIF ITEM:120 >= 10 && AT_HOME(MASTER) == 1
	;최대 9개(살). 당신의 몫은 남겨 둔다
	PRINTFORML %마스터가% 눈을 떠보니 요정들은 블랙썬더를 우물우물 하고 있었다
	PRINTFORML %마스터는% 요정들의 입 주위를 닦아주고 돌아가라고 타일렀다
	CALL LOST_ITEM("블랙썬더", RAND:9 + 1, 1)
ELSE
	PRINTFORML 요정들은 %GET_MAPNAME(MAIN_MAP)% 안을 어지럽혀 나갔다
	FOR LOCAL, MINROOM(), MAXROOM
		YOGORE:LOCAL = MIN(YOGORE:LOCAL + 1000 + RAND:4000, 6000)
	NEXT
ENDIF
PRINTW

path: ERB/이벤트관련/데일리이벤트/DAIRY_EV0 어질러짐_t.ERB


code:
@DAILY_CONDITION_EV10
#DIM YOME
#DIM CHILDNUM
SIF RAND:20 > 0
	RETURN -1
CALL PickRandGrownChild
YOME = RESULT
CHILDNUM = RESULT:1
VARSET RESULT

SIF YOME <= 0
	RETURN -1

CALL DAILY_EV10(YOME, CHILDNUM)
RETURN 1

@DAILY_EV10(ARG, ARG:1)
;PRINTFORMW %CALLNAME:ARG%과의 아이의 %CHILDNAME:ARG:(ARG:1)%로부터 편지가 왔다
PRINTFORMW %조사처리(CHILDNAME:ARG:(ARG:1),"로")%부터 편지가 왔다
LOCAL:1 = 0
FOR LOCAL, 1, CHARANUM
	SIF CFLAG:LOCAL:호감도 > CFLAG:ARG:호감도
		LOCAL:1 ++
NEXT

TARGET = ARG
;구상전용의 편지 이벤트 분기(없으면 통상 이벤트)
;口上専用の手紙イベント分岐（なければ通常イベント）,存在しないならRESULT = 2
IF !RAND:2
	CALL KOJO_MESSAGE_SEND("CHILD", 0, TARGET, ARG:1, 0, "DAILY_구상편지")
	SIF RESULT == 2
		GOTO TEGAMI
ELSE
	$TEGAMI
	IF LOCAL:1 >= 3
		CALL KOJO_MESSAGE_SEND("CHILD", 0, TARGET, ARG:1, 0, "DAILY_공통편지", "바람둥이")
		;存在しないならRESULT = 2（以下同様）
		IF RESULT == 2
			PRINTFORMW 좀 더 %ARG를% 소중히 하라고 적혀있었다……
		ENDIF
	ELSEIF !RAND:5
		CALL KOJO_MESSAGE_SEND("CHILD", 0, TARGET, ARG:1, 0, "DAILY_공통편지", "염치없는돈")
		;金銭を無遠慮にねだって得ようとすること、思慮なく金をせびることを意味する語。親に小遣いをたかることを指す場合が多い。
		;용돈 달라는 뜻
		IF RESULT == 2
			PRINTFORMW 용돈을 달라고 한다……
		ENDIF
		PRINTFORMW %마스터는% \@ MONEY > 30000? ￥30000을# 가진 돈 전부를\@ 보내기로 했다
		MONEY = MAX(MONEY - 30000,0)
	ELSE
		CALL KOJO_MESSAGE_SEND("CHILD", 0, TARGET, ARG:1, 0, "DAILY_공통편지", "송금")
		IF RESULT == 2
			PRINTFORMW %ARG와% %마스터를% 신경쓰는 말과 송금\\10000이 들어가 있었다
		ENDIF
		MONEY += 10000
	ENDIF
ENDIF
TARGET = TARGET:1
RETURN 1

path: ERB/이벤트관련/데일리이벤트/DAIRY_EV10 아이로부터편지_t.ERB


code:
@DAILY_CONDITION_EV11
IF !FLAG:자식기원
	RETURN -1
ELSEIF RAND:10 > 0
	RETURN -1
ELSE
	CALL DAILY_EV11(FLAG:자식기원)
	RETURN 1
ENDIF

@DAILY_EV11(ARG)
PRINTFORMW %ARG는% %마스터와%의 아이가 생기도록 기도하고 있다……
CFLAG:ARG:자식보정 ++

path: ERB/이벤트관련/데일리이벤트/DAIRY_EV11 자식기원_t.ERB


code:
@DAILY_CONDITION_EV12
VARSET LOCAL
IF RAND:15 > 0
	RETURN -1
ELSE
	CALL FISHER_YATES_SHAFFLE(CHARANUM - 1)
	FOR LOCAL:10, 0, CHARANUM - 1
		LOCAL = SHAFFLE_ARRAY:(LOCAL:10) + 1
		IF CFLAG:LOCAL:신뢰도 >= 100 && CFLAG:LOCAL:신사거주
			LOCAL:1 = LOCAL
			BREAK
		ENDIF
	NEXT
	IF !LOCAL:1
		RETURN -1
	ELSE
		CALL DAILY_EV12(LOCAL:1)
		RETURN 1
	ENDIF
ENDIF

@DAILY_EV12(ARG)
#DIM 정답
#DIM 경험치
#DIM 제한시간
#DIM 배율
#DIM FEINT
#DIM 가짜
VARSET LOCAL
가짜 = 0
제한시간 = 2000 - (ABL:ARG:전투능력 - ABL:MASTER:전투능력) * 100
CALL KOJO_MESSAGE_SEND("DAILY", 12, ARG, 1)
CALL ASK_YN("교제한다","싫다")
IF RESULT
	CALL KOJO_MESSAGE_SEND("DAILY", 12, ARG, 2)
ELSE
	정답 = RAND:4
	경험치 = MAX(ABL:ARG:전투능력 * 5, 5)
	SELECTCASE RAND:20
		CASE IS <= 11
		;보통으로
			LOCALS:0 = 위
			LOCALS:1 = 오른쪽
			LOCALS:2 = 왼쪽
			LOCALS:3 = 아래
			배율 = 100
		CASE 12 TO 15
		;로마자
			LOCALS:0 = WI
			LOCALS:1 = OREUNJJOK
			LOCALS:2 = OENJJOK
			LOCALS:3 = ARAE
			배율 = 160
		;영어
		CASE 16,17,18
			LOCALS:0 = UP
			LOCALS:1 = RIGHT
			LOCALS:2 = LEFT
			LOCALS:3 = DOWN
			배율 = 120
		;스페인어
		CASE 19
			LOCALS:0 = encima de
			LOCALS:1 = a la derecha
			LOCALS:2 = a la izquierda
			LOCALS:3 = abajo de
			배율 = 300
	ENDSELECT
	;전투능력이 높으면 페인트 걸쳐 온다
	IF ABL:ARG:전투능력 > 3 && !RAND:3
		FEINT = 1
		제한시간 += 400
		배율 += 80
	ELSE
		FEINT = 0
	ENDIF
	CALL FISHER_YATES_SHAFFLE(4)
	PRINTFORMW %CALLNAME:ARG%의 공격을 피하자 (클릭으로 개시)
	CALL KOJO_MESSAGE_SEND("DAILY", 12, ARG, 3)
	PRINTFORM %CALLNAME:ARG%의
	SETCOLOR C_YELLOW
	IF FEINT
		PRINTFORM %조사처리(LOCALS:(RAND:4),"로")%
		RESETCOLOR
		PRINTFORM 속이고 %조사처리(LOCALS:(3 - 정답),"로")%
	ELSE
		PRINTFORM %조사처리(LOCALS:(3 - 정답),"로")%
		RESETCOLOR
	ENDIF
	PRINTFORML 부터의 공격!
	$LOOP_IKASAMA
	FOR LOCAL:10, 0,4
		LOCAL = SHAFFLE_ARRAY:(LOCAL:10)
		PRINTFORML [{LOCAL}] - %조사처리(LOCALS:LOCAL,"로")% 피한다
	NEXT
	SIF !가짜 && BASE:MASTER:TSP > 500
		PRINTFORML [{4}] - 시간정지
	TINPUT 제한시간, -1, 1
	IF RESULT == 정답
		PRINTFORML %마스터는% 공격을 회피했다!
		IF !가짜
			PRINTFORMW 서로에게 좋은 경험이 되었다
		ELSE
			PRINTFORMW 미끄러졌기 떄문에 그다지 연습이 되지 않았다……
		ENDIF
		CALL KOJO_MESSAGE_SEND("DAILY", 12, ARG, 4, 가짜)
		경험치 = 경험치 * 배율 / 100
		EXP:MASTER:전투경험 += 경험치
		EXP:ARG:전투경험 += 경험치 * 2
		CFLAG:ARG:호감도 += 경험치 * 2
		CFLAG:ARG:신뢰도 += 경험치 * 2
		PRINTFORML 전투경험 ＋{경험치}(%CALLNAME:MASTER%)
		PRINTFORML 전투경험 ＋{경험치 * 2}(%CALLNAME:ARG%)
		PRINTFORML 호감도 ＋{경험치 * 2}(%CALLNAME:ARG%)
		PRINTFORML 신뢰도 ＋{경험치 * 2}(%CALLNAME:ARG%)
	ELSEIF RESULT == 4 && !가짜 && BASE:MASTER:TSP > 500
		가짜 = 1
		배율 /= 5
		PRINTFORMW %마스터는% 시점 정지 능력으로 공격을 판별했다!
		GOTO LOOP_IKASAMA
	ELSE
		PRINTFORML 피하지 못했다!
		PRINTFORM %마스터는%
		SELECTCASE ABL:ARG:전투능력
			CASE IS <= 1
				PRINTFORM 조금
			CASE 2,3
				PRINTFORM 상당히
			CASEELSE
				PRINTFORM 엉망진창
		ENDSELECT
		PRINTFORMW 아프다는 생각이 들었다
		CALL KOJO_MESSAGE_SEND("DAILY", 12, ARG, 5)
		BASE:MASTER:체력 -= 400 * (1 + ABL:ARG:전투능력)
		PRINTFORML 전투경험 ＋{경험치}(%CALLNAME:ARG%)
		EXP:ARG:전투경험 += 경험치
		CFLAG:ARG:호감도 += 경험치 / 2
		CFLAG:ARG:신뢰도 += 경험치 / 2
		PRINTFORML 호감도 ＋{경험치}(%CALLNAME:ARG%)
		PRINTFORML 신뢰도 ＋{경험치}(%CALLNAME:ARG%)
	ENDIF
ENDIF
PRINTW

@DAILY_MESSAGE_12(ARG, ARG:1, ARG:2, ARG:3, ARG:4, ARGS:1, ARGS:2)
SELECTCASE ARG:1
CASE 1
	PRINTFORMDL %ARG로%부터 신기술 훈련에 교제해달라는 말을 들었다
	SIF ABL:ARG:전투능력 >= 4
			PRINTFORML 상대가 상대이므로 자칫 잘못하면 크게 다칠지도 모른다……
CASE 2
	PRINTFORMDL %ARG는% 아쉬운 것 같다
ENDSELECT

path: ERB/이벤트관련/데일리이벤트/DAIRY_EV12 특훈_t.ERB


code:
@DAILY_CONDITION_EV13
#DIM 대상

SIF RAND:(10 - TALENT:MASTER:정욕감지)> 0
	RETURN -1
대상 = 0
CALL FISHER_YATES_SHAFFLE(CHARANUM - 1)
FOR LOCAL:10, 0, CHARANUM - 1
	LOCAL = SHAFFLE_ARRAY:(LOCAL:10) + 1
	SIF !CFLAG:LOCAL:신사거주
		CONTINUE
	IF CFLAG:LOCAL:욕구불만도 >  700
		대상 = LOCAL
		BREAK
	ENDIF
NEXT
SIF !대상
	RETURN -1

CALL DAILY_EV13(대상)
RETURN 1

@DAILY_EV13(ARG)
PRINTFORMW 한밤중 잠에서 깨버리고 말았다
PRINTFORMW 더 자려고 생각했지만 아무래도 잠들 수 있을 것 같지 않다……
CALL ASK_YN("돌아다닌다","방에 있는다")
IF RESULT
	PRINTFORM %마스터는% 졸려질 때까지
	PRINTDATA
		DATAFORM 콜렉션을 정리·감상하며 보냈다
		DATAFORM 미와 선, 팬티에 대해 생각을 깊이 했다
		DATAFORM 벽의 얼룩과 담소했다
	ENDDATA
	PRINTFORML ……
	CALL COLORMESSAGE("마음이 풍부해지면서 일시적으로 기력과 TSP가 좀 늘었다", C_YELLOW, 2,1)
	CALL BUFF_BASE(MASTER, BASE_기력, 100)
	CALL BUFF_BASE(MASTER, BASE_TSP, 100)
ELSEIF !RAND:4 || (!RAND:4 && ABL:ARG:친밀 >= 7)
	PRINTFORMW 적당히 돌아다니다가, \@ CFLAG:MASTER:311 == CFLAG:ARG:311 ? 자기 방으로 돌아왔더니 뭔가 소리가 들렸다# 이제 슬슬 돌아갈까 하다가 %CALLNAME:ARG%의 방에서 뭔가 들려서 들러봤다\@
	CALL ONABARE(ARG)
	CFLAG:MASTER:철야 = 1
ELSEIF ITEM:토부「개운대문」
		PRINTL 이런, 개운대문의 모습이…?
		CALL ASK_YN("꺼낸다","방치한다")
		IF !RESULT
			ITEM:토부「개운대문」 --
			SETCOLOR COLOR ("YELLOW")
			PRINTFORMW 토부「개운대문」이 발동!
			RESETCOLOR
			PRINTFORMW 적당히 돌아다니다가, \@ CFLAG:MASTER:311 == CFLAG:ARG:311 ? 자기 방으로 돌아왔더니 뭔가 소리가 들렸다# 이제 슬슬 돌아갈까 하다가 %CALLNAME:ARG%의 방에서 뭔가 들려서 들러봤다\@
			CALL ONABARE(ARG)
			CFLAG:MASTER:철야 = 1
		ELSEIF !RAND:2
			GOTO OYATU
		ELSE
			GOTO NANIMONAI
		ENDIF
ELSEIF !RAND:2
	$OYATU
	PRINTFORML 배가 조금 고프다 게다가 뭔가 먹으면 졸려질지도 모른다
	PRINTFORMW %마스터는% 부엌에 가서 찬장을 뒤지자, 맛있는 간식을 발견했다
	PRINTFORMW %마스터가% 산 것은 아니다. 그러나 누군가가 %마스터를% 위해 준비해줬을 가능성도 버리지 못한다
	PRINTFORML
	CALL COLORMESSAGE("어떠케, 이제 곧 유통기하니 지나! 누가 빨리 머거주면 안 될까!", C_YELLOW, 2,2)
	;ﾄﾞｳｼﾖｳ､ﾓｳｽｸﾞｼｮｳﾐｷｹﾞﾝｶﾞｷﾁｬｳﾜ!ﾀﾞﾚｶﾊﾔｸﾀﾍﾞﾃｸﾚﾅｲｶｼﾗ!
	;どうしょう､もうすぐしょうみきげんがきちゃうわ!だれかはやくたべてくれないかしら!
	;어떻게, 이제 곧 유통기한이 지나! 누가 빨리 먹어주면 안될까!
	PRINTFORML
	PRINTFORML 간식도 이렇게 말한다. 고맙게 받기로 하자
	CALL BUFF_BASE(MASTER, BASE_기력, 300)
	CALL COLORMESSAGE("맛있는 간식을 먹어 일시적으로 기력이 증가한다", C_YELLOW, 2,1)
	PRINTFORMW %마스터는% 이를 다시 닦고 기분 좋게 잠자리에 들었다
	CALL COLORMESSAGE(@"왠지 %CALLNAME:ARG%의 기분이 나빠진다……", C_RED, 2,1)
	TALENT:ARG:기분 = -1
ELSE
	$NANIMONAI
	PRINTFORMW 돌아다녀보긴 했지만 딱히 재미있는 일도 없고 얌전하게 다시 자기로 했다
	CFLAG:MASTER:취침시간 += 60
	CALL COLORMESSAGE("수면시간이 1시간 줄어들었다", C_YELLOW, 2,1)
ENDIF

path: ERB/이벤트관련/데일리이벤트/DAIRY_EV13 자택에서_딸딸이들킴_t.ERB


code:
@DAILY_CONDITION_EV15
SIF RAND:20 > 0 || FLAG:이상기상
	RETURN -1
CALL DAILY_EV15
RETURN 1

@DAILY_EV15
FLAG:이상기상 = 6 + RAND:2

TIME:5 = 0
WIND_VELOCITY = 0
SELECTCASE FLAG:이상기상
	CASE 6
		LOCALS = 치마가 넘어가 머리까지 닿으면 기분 좋아지게
	CASE 7
		LOCALS = 바람이 팬티를 싣고 오게
	;CASE 8
	;LOCALS = 분홍빛 바람이 불게
ENDSELECT

PRINTFORMW 지구온난화, 환경오염, 난민 문제, 대통령 선거, 팬더믹, 장시간 노동, 보행 중 폰질, Switch 품귀……
PRINTFORMW 바깥세상의 갖가지 문제는 환상향에도 영향을 미치지 않을 수는 없었다
CALL COLORMESSAGE(@"바깥세상의 영향으로 발생한 이상기상 때문에 %LOCALS% 되었다", C_PINK, 2,1)

path: ERB/이벤트관련/데일리이벤트/DAIRY_EV15 이상기상_t.ERB


code:
@DAILY_CONDITION_EV16
SIF !TFLAG:치마걷어올리기지나고
	RETURN -1
SIF RAND:(TFLAG:치마걷어올리기지나고) < RAND:300
	RETURN -1
CALL DAILY_EV16
RETURN 1

@DAILY_EV16
PRINTFORML 붕붕마루 신문의 호외가 나왔습니다!
PRINTFORMW 치마를 넘기는 %CALLNAME:MASTER%의 기사가 실려 있습니다!
FOR LOCAL, 1, CHARANUM
	CFLAG:LOCAL:호감도 = MAX(CFLAG:LOCAL:호감도 - 100, 0)
	CFLAG:LOCAL:신뢰도 = MAX(CFLAG:LOCAL:신뢰도 - 50, 0)
NEXT
CALL COLORMESSAGE("전원의 호감도가 100, 신뢰도가 50 떨어졌다……", C_YELLOW, 2,1)

path: ERB/이벤트관련/데일리이벤트/DAIRY_EV16 치마걷어올리기_t.ERB


code:
@DAILY_CONDITION_EV17
SIF RAND:20 > 0 || FLAG:이상기상
	RETURN -1
CALL DAILY_EV17
RETURN 1

@DAILY_EV17
FLAG:이상기상 = RAND:5 + 1
TIME:5 = 0
SELECTCASE FLAG:이상기상
	CASE 1
		LOCALS = 모두의 클리토리스가 민감
	CASE 2
		LOCALS = 모두의 여성기가 민감
	CASE 3
		LOCALS = 모두의 항문이 민감
	CASE 4
		LOCALS = 모두의 가슴이 민감
	CASE 5
		LOCALS = 모두의 입술이 민감
ENDSELECT

PRINTFORMW 영원정의 조약실이 폭발 사고를 냈습니다!
PRINTFORMW 기화한 약품은 캇파 공방의 매연과 반응해, 미지의 화합물이 되어 대기 중에 퍼져나간다……
CALL COLORMESSAGE(@"%LOCALS%해졌다", C_PINK, 2,1)

path: ERB/이벤트관련/데일리이벤트/DAIRY_EV17 폭발사고_t.ERB


code:
@DAILY_CONDITION_EV18
SIF RAND:15 > 0
	RETURN -1
CALL DAILY_EV18
RETURN 1

@DAILY_EV18
PRINTFORMW 달에서 요염한 빛이 쏟아져 내린다……

FOR LOCAL, 0, CHARANUM
	SIF ((TALENT:LOCAL:150 || TALENT:LOCAL:151) && !TALENT:LOCAL:초경) || !GETBIT(TALENT:LOCAL:성별, 0)
		CONTINUE
	SIF RAND:5 > 0
		CONTINUE
	CALL COLORMESSAGE(@"%CALLNAME:LOCAL%의 생리주기가 흐트러졌다", C_YELLOW, 1)
	CFLAG:LOCAL:900 = RAND:10 + 1
NEXT
PRINTW

path: ERB/이벤트관련/데일리이벤트/DAIRY_EV18 생리주기랜덤화_t.ERB


code:
;(天候パッチ)廃止
; @DAILY_CONDITION_EV19
; SIF RAND:10 > 0
; 	RETURN -1
; SIF FLAG:이상기상
; 	RETURN -1
; ;여름과 겨울만
; SIF !GROUPMATCH(DAY:2, 2,4)
; 	RETURN -1
; CALL DAILY_EV19
; RETURN 1

; @DAILY_EV19
; FLAG:이상기상 = RAND:2 + 9

; SELECTCASE DAY:2
; 	CASE 2
; 		FLAG:이상기상 = 9
; 		TIME:5 = 5
; 		PRINTFORML 바람이 엄청 분다……
; 	CASE 4
; 		FLAG:이상기상 = 10
; 		TIME:5 = 9
; 		PRINTFORML 오늘밤은 냉기가 심하다……
; ENDSELECT

; CALL COLORMESSAGE(@"내일은 %조사처리(GET_WEATHER(),"가")% 될 것 같다……", C_YELLOW, 2,1)

path: ERB/이벤트관련/데일리이벤트/DAIRY_EV19 계절한정날씨_t.ERB


code:
@DAILY_CONDITION_EV1
;영원정거주지라면1/10, 그 이외로1/30로 발생
SIF MAIN_MAP == 4 && RAND:10 > 0
	RETURN -1
SIF MAIN_MAP != 4 && RAND:30 > 0
	RETURN -1
CALL DAILY_EV1
RETURN 1

@DAILY_EV1
;여기에 이벤트 내용을 쓰자 최후는 RETURN 1해
IF MAIN_MAP != 4
	PRINTFORML 영원정에서 약 시공품이 도착했다
ELSE
	PRINTFORML 에이린에게 약 시공품을 받았다
ENDIF
CALL GET_ITEM("영양드링크", 1, 2)
RETURN 1


code:
@DAILY_CONDITION_EV20
SIF RAND:15 > 0
	RETURN -1
SIF CFLAG:[[반키]]:신사거주
	RETURN -1
SIF CFLAG:[[반키]]:출금
	RETURN -1
SIF !TALENT:[[반키]]:사모 && !TALENT:[[반키]]:연모
	RETURN -1
CALL DAILY_EV20
RETURN 1

@DAILY_EV20
#DIM 손님수

PRINTFORML 한밤중, 창가에서 목소리가 들렸다
PRINTFORMW %마스터가% 졸린 눈을 비비며 창문을 열자, 머리가 떠 있었다
PRINTFORML ……세키반키의 머리다
PRINTFORML 지나가는 길에 들러봤다고 한다
;通りがかったから寄ってみたという
CALL ASK_YN("들어오게 한다","졸리다")
IF RESULT
	PRINTFORMW 졸리기 때문에 되돌려 보냈다
	RETURN
ENDIF
PRINTFORML 들어오도록 했다

손님수 = 1
IF ABL:[[반키]]:친밀 >= 10 && !RAND:3
	PRINTFORML 창문으로 머리가 들어왔다… …하지만, 하나뿐이 아니었다
	FOR LOCAL, 2, ABL:73:친밀 - 5
		SIF !RAND:2
			CONTINUE
		손님수 ++
		PRINTFORMW {손님수}번째 머리가 들어왔다
		SIF 손님수 >= 9
			BREAK
	NEXT
	SIF 손님수 == 1
		PRINTFORML 라고 생각했지만, 역시 하나였다
	SIF 손님수 >= 4
		PRINTFORMW 이만큼 같은 얼굴에 둘러싸지면, 역시 지내기가 불편하다……
ENDIF
IF TIME:5 >= 4
	PRINTFORML 젖어있었기 때문에 목욕타올을 건네주었다
	PRINTFORML 타올 위에서 굴러다니는 머리는 꽤 재미있는 광경이다
ELSE
	SELECTCASE RAND:3
		CASE 0
			PRINTFORML 눈높이를 맞추기 위해 책상 위에 방석을 놓고 앉도록 했다
			SIF 손님수 > 1
				PRINTFORML 책상 위에 생목이 늘어서 있는 광경은 마치 베어진 적군의 머리 수를 확인하는 것 같았다
				;獄門か首実検だ
		CASE 1
			PRINTFORML 차를 끓였더니 마시기 힘들어해서 빨대를 꽂았다
			PRINTFORML 마신 것은 도대체 어디로 가는 것일까……
		CASE 2
			PRINTFORML 뒷머리에 잎사귀가 걸려 있어서 때주었다
	ENDSELECT
ENDIF
PRINTFORMW
PRINTFORMW …
PRINTFORMW ……
PRINTFORMW ………

IF RAND:100 < ABL:[[반키]]:친밀 * 3 + TALENT:[[반키]]:연모 * 15
	PRINTFORM 잠시 담소한 후, 머리\@ 손님수 > 1 ? 들은 # 는\@ 내일은 몸통째 온다며
	IF !PRIVATEROOM:(CFLAG:MASTER:초기위치 % 100) && TALENT:[[반키]]:연모
		PRINTFORML 돌아가려고 했지만 어차피 내일 올거면 자고 가라고 했다
		PRINTFORM %마스터는% 머리맡에 머리를 \@ 손님수 > 1 ? 나란히 # 두고 \@ 잠들었다……
	ELSE
		PRINTFORML 돌아갔다
	ENDIF
	CFLAG:[[반키]]:다음날내방 = 1
ELSE
	PRINTFORM 잠시 담소한 후, 머리\@ 손님수 > 1 ? 들은 # 는\@ 창문으로 돌아갔다
	SIF !RAND:2
		CFLAG:[[반키]]:다음날내방 = 1
ENDIF
;시스템 메시지 다하고 호감도 변화
;마지막 인수만 호감도가 변화
CALL CHANGE_CFLAG(2,[[반키]], 50)

;시스템 메시지 다하고 신뢰도 변화
;마지막 인수만 신뢰도가 변화
CALL CHANGE_CFLAG(4,[[반키]], 20)

;수면시간 감소
CFLAG:MASTER:취침시간 += 60
CALL COLORMESSAGE("수면시간이 1시간 줄어들었다", C_YELLOW, 2)

path: ERB/이벤트관련/데일리이벤트/DAIRY_EV20 생목님_어서오세요_t.ERB


code:
@DAILY_CONDITION_EV21
#DIM YOME
SIF !FLAG:자식기원
	RETURN -1
SIF FLAG:획득팬티총수 < 100
	RETURN -1
YOME = FLAG:자식기원

SIF RAND:30 > 0
	RETURN -1
SIF TALENT:YOME:임신
	RETURN -1
SIF !CHECK_CHARA(YOME, "유령") && !CHECK_CHARA(YOME, "인형") && !CHECK_CHARA(YOME, "봉래인") && !CHECK_CHARA(YOME, "신령")
	RETURN -1
SIF CFLAG:YOME:아이인원수 > 0
	RETURN -1
CALL DAILY_EV21(YOME)

RETURN 1

@DAILY_EV21(YOME)
#DIM YOME
PRINTFORMW 팬티님은 자식 없는 %조사처리(CALLNAME:YOME,"를")% 불쌍히 여겼다……
CALL BUFF_BASE(MASTER, BASE_체력, 1000)
CALL BUFF_BASE(MASTER, BASE_기력, 1000)
CALL COLORMESSAGE(@"%CALLNAME:YOME%의 체력과 기력이 일시적으로 크게 강화되었다", C_YELLOW, 1,1)
CALL BUFF_BASE(MASTER, BASE_정력, 500)
CALL COLORMESSAGE(@"%CALLNAME:MASTER%의 정력이 일시적으로 강화되었다", C_YELLOW, 1,1)
IF !TALENT:MASTER:농후정액
	TALENT:MASTER:농후정액 = 1
	CLEARBIT TFLAG:일일일회, 8
	CALL COLORMESSAGE(@"%마스터는%【농후정액】이 되었다", C_YELLOW, 1,1)
	BASE:MASTER:정력 = MAXBASE:MASTER:정력
ENDIF
PRINTW
CFLAG:YOME:다음날내방 = 1
CFLAG:YOME:자식보정 ++


code:
@DAILY_CONDITION_EV22
#DIM YOME
#DIM CHILDNUM
SIF CMATCH(TALENT:육아중, 0) == CHARANUM
	RETURN -1
SIF RAND:5 > 0
	RETURN -1
YOME = 0
CALL FISHER_YATES_SHAFFLE(CHARANUM - 1)
FOR LOCAL:10, 0, CHARANUM - 1
	LOCAL = SHAFFLE_ARRAY:(LOCAL:10) + 1
	IF CHK_HAVE_CHILD(LOCAL, CHILD_취학, CHILD_자립) > 0
		YOME = LOCAL
		CHILDNUM = CHK_HAVE_CHILD(LOCAL, CHILD_취학, CHILD_자립)
		BREAK
	ENDIF
NEXT
IF YOME
	CALL DAILY_EV22(YOME, CHILDNUM)
	RETURN 1
ELSE
	RETURN -1
ENDIF

@DAILY_EV22(YOME, CHILDNUM)
#DIM YOME
#DIM CHILDNUM
#DIM 위자료

PRINTFORM %조사처리(CALLNAME:YOME,"로")%부터, 서당에서 %조사처리(CHILDNAME:YOME:CHILDNUM,"가")%

IF !RAND:4
	PRINTFORMW 다른 아이에게 부상을 입혔다고 전해들었다……
	PRINTFORML 설령 악의가 없었더라도, 보통사람 상대에게 %조사처리(CHILDNAME:YOME:CHILDNUM,"가")% 힘을 준다면 쉽게 상처를 줄 것이다
	PRINTFORML 부모님께 사과한 후에 %CHILDNAME:YOME:CHILDNUM%에게도 잘 말하게 해야……
	PRINTFORML %마스터는% 마음이 무거워졌다
	PRINTFORML
	CALL KOJO_MESSAGE_SEND("EVENT", 0, YOME, CHILDNUM, 0, "CHILD_DAILY_서당", "다치게 했다")
	위자료 = MIN(10000, MONEY)
	MONEY -= 위자료
	CALL COLORMESSAGE(@"상대방에게 치료비로 {위자료} 지불했다", C_YELLOW, 1,1)
	CALL COLORMESSAGE(@"%CALLNAME:MASTER%의 기력이 일시적으로 줄었다", C_YELLOW, 1,1)
	CALL BUFF_BASE(MASTER, BASE_기력,-1000)
ELSE
	PRINTFORMW 칭찬받았다고 전해들었다……
	PRINTFORML 과연 우리 아이다!
	PRINTFORML
	CALL KOJO_MESSAGE_SEND("EVENT", 0, YOME, CHILDNUM, 0, "CHILD_DAILY_서당", "칭찬받았다")
	CALL COLORMESSAGE(@"%CALLNAME:MASTER%의 기력이 일시적으로 늘었다", C_YELLOW, 1,1)
	CALL BUFF_BASE(MASTER, BASE_기력, 1000)
ENDIF
PRINTW

path: ERB/이벤트관련/데일리이벤트/DAIRY_EV22 아이가_서당으로_t.ERB


code:
@DAILY_CONDITION_EV24

SIF RAND:15 > 0
	RETURN -1
SIF DAY:3 >= 30
	RETURN -1

CALL DAILY_EV24
RETURN 1

@DAILY_EV24
PRINTFORMW 계절 이변이다！
FOR LOCAL,1,CHARANUM
	IF TALENT:LOCAL:요정
		CALL BUFF_BASE(LOCAL,BASE_체력,300)
		CALL BUFF_BASE(LOCAL,BASE_기력,300)
	ENDIF
NEXT
GatheringSeason = DAY:2 + 2
SIF GatheringSeason > 4
	GatheringSeason -= 4
PRINTFORMW 요정들의 체력과 기력이 올라가고、또한 %GET_MONTH(GatheringSeason)%의 소재를 추가로 채집할 수 있게 되었다！

path: ERB/이벤트관련/데일리이벤트/DAIRY_EV24 계절이변.ERB


code:
@DAILY_CONDITION_EV25

IF !FLAG:패션 && !RAND:20
	CALL DAILY_EV25
	RETURN 1
ENDIF
RETURN -1

@DAILY_EV25

#DIM Type, 2
#DIMS Desc, 2

Type:0 = 62
Desc:0 = 멋진 T셔츠
Type:1 = 74
Desc:1 = 세라복
; Type:2 = 55
; Desc:2 = 바니 슈트

LOCAL = RAND:2

PRINTFORML 아무래도 향간에는 %조사처리(Desc:LOCAL,"가")% 유행하고 있는 것 같다
FLAG:패션 = Type:LOCAL
;62 대신 원하는 의상세트 번호를 입력하면 모두의 복장을 바꿀 수 있다
;의상세트.ERB를 참조
PRINTW
RETURN 1

path: ERB/이벤트관련/데일리이벤트/DAIRY_EV25 패션_붐.ERB


code:
@DAILY_CONDITION_EV26
#DIM 발생율
발생율 = 25
SIF FLAG:획득팬티총수 >= 4
	발생율 += SQRT(FLAG:획득팬티총수) / 2
발생율 += FLAG:독신 * 5
IF 발생율 < RAND:1000 || FLAG:이상기상
	RETURN -1
ENDIF
CALL DAILY_EV26
RETURN 1

@DAILY_EV26
FLAG:독신 = 0
PRINTFORMW %CALLNAME:MASTER%의 쌓아온 덕이 기적을 일으켰다……！
CALL COLORMESSAGE(@"분홍빛 바람이 불게 되었다",C_PINK,2,1)
FLAG:이상기상 = 8

path: ERB/이벤트관련/데일리이벤트/DAIRY_EV26 분홍바람.ERB


code:
@DAILY_CONDITION_EV27

IF !FLAG:기념품점염가대매출 && !RAND:30
	CALL DAILY_EV27
	RETURN 1
ELSE
	RETURN -1
ENDIF

@DAILY_EV27
;この後にフラグリセットが入るので目印として+1100しておく
FLAG:기념품점염가대매출 = RAND:11 + 1100
SIF FLAG:기념품점염가대매출 % 100 == MAIN_MAP
	RESTART
IF FLAG:신문구독
	PRINTFORML 붕붕마루 신문에 의하면, %STR:(5000 + (FLAG:기념품점염가대매출 % 100) )% 방면 기념품 가게가 대염가 판매를 시작했다고 한다
	PRINTFORMW 전품 반액 세일인 것 같다
ENDIF
RETURN 1

path: ERB/이벤트관련/데일리이벤트/DAIRY_EV27 반액세일.ERB